



# για αρχάριους

με έμφαση στη χρήση του στο εργαστήριο Φυσικών Επιστημών

```
/*  
  Bare minimum Arduino sketch  
*/  
  
void setup() {  
  // put your setup code here, to run once:  
  
}  
  
void loop() {  
  // put your main code here, to run repeatedly:  
  
}
```



# ARDUINO

## για αρχάριους

με έμφαση στη χρήση του στο εργαστήριο Φυσικών Επιστημών

Νούσης Βασίλης

ISBN : 978-618-83502-3-6



## Πίνακας περιεχομένων

Πρόλογος 1 <sup>ης</sup> έκδοσης.....	7
Πρόλογος 2 <sup>ης</sup> έκδοσης.....	9
Εισαγωγή.....	11
Το υλικό του Arduino .....	12
Άλλοι Arduinos.....	16
1. Arduino Nano.....	16
2. Arduino MKR .....	17
3. Κλασσικοί Arduino .....	17
4. Arduino Mega .....	18
Το λογισμικό του Arduino .....	19
1. Bootloader.....	19
2. Ολοκληρωμένο περιβάλλον ανάπτυξης (Arduino IDE) .....	19
3. Βιβλιοθήκες.....	20
Η κοινότητα .....	21
Στοιχεία προγραμματισμού του Arduino.....	22
Τα βασικά.....	22
Πριν το πρώτο sketch - προαπαιτούμενες ρυθμίσεις .....	22
Το πρώτο sketch - χρήση ψηφιακής ακίδας ως εξόδου .....	25
Σύνδεση LED σε άλλη ψηφιακή έξοδο - τροποποίηση του σχετικού sketch.....	27
Χρήση ψηφιακής ακίδας ως εισόδου .....	29
Σειριακή επικοινωνία μεταξύ Arduino και υπολογιστή ή άλλων συσκευών.....	32
Δομές λήψης απόφασης.....	35
α. Η δομή if...else .....	35
β. Η δομή switch...case .....	37
Βρόχοι επανάληψης .....	40
α. Ο βρόχος for (for loop) .....	41
β. Ο βρόχος while (while loop).....	42
Στοιχεία δομημένου προγραμματισμού .....	44
Ανάγνωση αναλογικής εισόδου .....	47
Χρήση προεγκατεστημένης βιβλιοθήκης.....	50
Εγκατάσταση και χρήση εξωτερικής βιβλιοθήκης .....	53
Χρήση του δισύρματου (I <sup>2</sup> C) σειριακού διαύλου επικοινωνίας.....	56
α. Ο αισθητήρας BMP280 .....	58
β. Ρολόι πραγματικού χρόνου (RTC) DS1307 .....	62
γ. Χρήση οθόνης υγρών κρυστάλλων (LCD) μέσω διαύλου I <sup>2</sup> C .....	67

Χρήση του μονοσύρματου διαύλου 1-Wire.....	69
Χρήση του διαύλου SPI.....	75
Διακοπές.....	81
α. Εξωτερικές διακοπές.....	81
β. Διακοπές χρονιστή (Timer Interrupts).....	83
Πηγαίνοντας λίγο μακρύτερα... ..	87
Ο ηλεκτρονόμος ή ρελέ (relay).....	87
Ένας απλός αυτοματισμός με ηλεκτρονόμο .....	89
Ο αισθητήρας υπερήχων HC-SR04 .....	93
Ένα απλό σύστημα καταγραφής δεδομένων (Data Logger).....	97
Ένα ρολόι πραγματικού χρόνου με οθόνη υγρών κρυστάλλων.....	100
Ηλεκτρικές μετρήσεις με τον Arduino και το INA219 – Νόμος του Ohm .....	102
Δικτύωση Bluetooth.....	108
Το πρότυπο Bluetooth Low Energy (BLE).....	116
Ένας απλός αυτοματισμός μέσω Bluetooth.....	117
Μια απλή φωτοπύλη .....	119
RGB LEDs – Μίξη χρωμάτων .....	124
Τι είναι η διαμόρφωση παλμών κατά πλάτος (PWM);.....	125
Μίξη χρωμάτων .....	125
Οι μικροελεγκτές ESP8266 και ESP32 της Espressif.....	128
Εγκατάσταση του ESP32 στο Arduino IDE .....	132
Ένα παράδειγμα αξιοποίησης του ESP32: Μίνι μετεωρολογικός σταθμός .....	133
Ασύρματη μετάδοση δεδομένων μέσω Bluetooth .....	135
Ο μετεωρολογικός σταθμός στην IoT πλατφόρμα ThingSpeak .....	138
Παράρτημα Α: Τύποι δεδομένων.....	147
Παράρτημα Β: Τελεστές .....	148
Παράρτημα Γ: AT εντολές για τις μονάδες Bluetooth .....	151
Παράρτημα Δ: Οι ADC ακίδες του ESP32.....	153
Βιβλιογραφία.....	154

## Πρόλογος 1<sup>ης</sup> έκδοσης

Το βιβλίο που κρατάτε στα χέρια σας δε φιλοδοξεί να αποτελέσει ένα πλήρη οδηγό προγραμματισμού και χρήσης της μικροϋπολογιστικής πλατφόρμας Arduino. Άλλωστε είμαι μόνο ένας αυτοδίδακτος χρήστης της πλατφόρμας, και συνεπώς το βιβλίο, που είναι αποτέλεσμα της προσωπικής μου ενασχόλησης με τον Arduino με στόχο την αξιοποίησή του στο εργαστήριο Φυσικών Επιστημών, απευθύνεται πρωτίστως στον αρχάριο χρήστη.

Τρεις ενότητες συναποτελούν το βιβλίο:

1. Στην εισαγωγική ενότητα σύντομα παρουσιάζονται τα χαρακτηριστικά (υλικό, λογισμικό και κοινότητα χρηστών) που αποτελούν τη βάση της επιτυχίας της πλατφόρμας.
2. Στη δεύτερη ενότητα αναπτύσσονται τα βασικά στοιχεία που αφορούν στον προγραμματισμό του Arduino, και παρουσιάζεται ένας σημαντικός αριθμός αισθητήρων και άλλων διατάξεων που μπορούν να συνδεθούν στον Arduino.
3. Στην τρίτη ενότητα παρουσιάζονται μερικά πιο απαιτητικά projects που συνδυάζουν αρκετά από τα χαρακτηριστικά που παρουσιάστηκαν στην προηγούμενη ενότητα.

Συνολικά 37 sketches (προγράμματα) παρουσιάζονται στο βιβλίο, τα οποία ως ένα συμπιεσμένο αρχείο τύπου "rar" είναι διαθέσιμα προς "μεταφόρτωση" στον υπολογιστή σας μέσω του συνδέσμου: <https://tinycloud.com/3dsvbt35> ή μέσω του επόμενου QR Code:



Κάθε παράδειγμα του βιβλίου συνοδεύεται και από το σχετικό κύκλωμα με τις απαραίτητες συνδέσεις, σχεδιασμένο με τη βοήθεια του εξαιρετικού ελεύθερου λογισμικού Fritzing.

Ελπίζω το βιβλίο να φανεί χρήσιμο στον αρχάριο που ενδιαφέρεται να ανακαλύψει τις δυνατότητες της πλατφόρμας Arduino.

Ηγουμενίτσα 1/3/2019

Βασίλης Νούσης



## Πρόλογος 2<sup>ης</sup> έκδοσης

Σχεδόν έξη χρόνια έχουν περάσει από την πρώτη έκδοση του βιβλίου μου για το μικροϋπολογιστικό σύστημα Arduino. Πολλά έχουν μεσολαβήσει από τότε, με κυριότερα:

- Την ανάπτυξη πλήθους νέων μικροϋπολογιστικών συστημάτων αυξημένων δυνατοτήτων π.χ. με μικροεπεξεργαστές 32bit, περισσότερες ψηφιακές εισόδους/εξόδους, μετατροπείς αναλογικού σε ψηφιακό ανάλυσης 12bit, κ.ά, που μπορούν να προγραμματιστούν με το ίδιο περιβάλλον ανάπτυξης (Arduino IDE).
- Τη μετατόπιση προς συστήματα λογικής 3,3 V σε αντίθεση με τη λογική 5 V του κλασσικού Arduino.
- Την ενσωμάτωση δυνατοτήτων δικτύωσης Bluetooth είτε με το κλασσικό είτε με το νεότερο Bluetooth Low Energy (BLE) πρωτόκολλο.
- Τη δυνατότητα WiFi δικτύωσης για την ανάπτυξη συστημάτων και εφαρμογών Internet of Things (IoT).
- Την ανάπτυξη της έκδοσης 2 του περιβάλλοντος εργασίας Arduino IDE, που αντιμετωπίζει το σημαντικότερο πρόβλημα της πρώτης του έκδοσης, αφού οι δυνατότητές του καλύπτουν πλέον και τον έμπειρο προγραμματιστή.

Θα προσπαθήσουμε σε αυτή τη δεύτερη έκδοση του βιβλίου να παρακολουθήσουμε αυτές τις νέες τάσεις, ακολουθώντας όμως και τις βασικές αρχές της πρώτης έκδοσης: θα απευθύνεται στον αρχάριο χρήστη και θα δίνουμε έμφαση στη χρήση των μικροϋπολογιστικών συστημάτων στο εργαστήριο Φυσικών Επιστημών.

Θα βρείτε τα επιπλέον παραδείγματα του βιβλίου εδώ: <https://tinyurl.com/3tuw4pc2> ή μέσω του επόμενου QR Code:



Ηγουμενίτσα 18/11/2024


Βασίλης Νούσης



# Arduino

## Εισαγωγή

Η πλατφόρμα Arduino είναι ένα ανοιχτού υλικού και ανοιχτού κώδικα σύστημα ανάπτυξης ηλεκτρονικών πρωτοτύπων, βασισμένο σε ευέλικτο και εύκολο στη χρήση υλικό και λογισμικό.




Εικόνα 1: Arduino - Ιστορική εξέλιξη

Προέκυψε ως το αποτέλεσμα της ανάγκης για ένα φθηνότερο, ισχυρότερο, περισσότερο ευέλικτο και εύκολο στη χρήση αναπτυξιακό σύστημα, σε σχέση με το Basic Stamp της Parallax που στις αρχές του 2000 χρησιμοποιούσαν οι φοιτητές του Massimo Banzi στο Interaction Design Institute (IDI) στην Ivrea της Ιταλίας. Το 2003 ο Hernando Barragan στα πλαίσια της μεταπτυχιακής διπλωματικής του εργασίας στο IDI, υπό την επίβλεψη του Massimo Banzi και του Ceasey Reas, ανέπτυξε το σύστημα Wiring, ένα συνδυασμό μικροϋπολογιστικού συστήματος και κατάλληλου προγραμματιστικού περιβάλλοντος βασισμένου στη γλώσσα προγραμματισμού Processing που λίγα χρόνια νωρίτερα είχαν παρουσιάσει ο Ben Fry και ο Ceasey Reas [1]. Το 2005 ο Massimo Banzi με μια ομάδα συνεργατών (στην οποία δε συμμετείχε ο Barragan) με βάση το λογισμικό του Wiring παρουσίασαν ένα σημαντικά φθηνότερο μικροϋπολογιστικό σύστημα με το όνομα "Arduino", δανεισμένο από το όνομα ενός μπαρ που υπήρξε το στέκι της ομάδας του Banzi εκείνη την εποχή [2].

Πολλές διαφορετικές εκδόσεις του συστήματος Arduino έχουν παρουσιαστεί μέχρι σήμερα με πιο επιτυχημένη την έκδοση Arduino Uno R3.

## Το υλικό του Arduino

Το υλικό του Arduino (μοντέλο Uno R3 και αντίστοιχοι κλώνοι, στο οποίο θα αναφερόμαστε εν συνεχεία) συναρμολογείται σε μια μικρή πλακέτα διαστάσεων 68mm x 53mm.



Εικόνα 2: Η πλακέτα του Arduino Uno R3

Στην καρδιά του συστήματος βρίσκεται ο οκτάμπιτος μικροελεγκτής ATmega 328 της Atmel χρονισμένος στα 16MHz, που διαθέτει ενσωματωμένη μνήμη τριών τύπων:

1. 2 kB στατικής μνήμης RAM στην οποία τα προγράμματα κατά την εκτέλεσή τους αποθηκεύουν μεταβλητές, πίνακες κ.λπ. Η μνήμη αυτή χάνει τα δεδομένα της όταν η παροχή ρεύματος σταματήσει ή αν γίνει επανεκκίνηση (reset) του συστήματος.
2. 1 kB μνήμης EEPROM στην οποία κατά την εκτέλεση των προγραμμάτων μπορούν να εγγραφούν/διαβαστούν δεδομένα byte προς byte. Η EEPROM δεν χάνει τα περιεχόμενά της με απώλεια τροφοδοσίας ή επανεκκίνηση του συστήματος.
3. 32 kB μνήμης Flash, από τα οποία τα 0,5 kB χρησιμοποιούνται από τον bootloader του Arduino, το λογισμικό δηλ. εκείνο που είναι απαραίτητο για την εγκατάσταση των δικών μας προγραμμάτων στον μικροελεγκτή μέσω της θύρας USB, χωρίς να χρειάζεται εξωτερικός προγραμματιστής. Το υπόλοιπο της μνήμης Flash χρησιμοποιείται για την αποθήκευση αυτών ακριβώς των προγραμμάτων. Και η μνήμη Flash δε χάνει τα περιεχόμενά της με απώλεια τροφοδοσίας ή επανεκκίνηση [3].

Στο υλικό του μικροελεγκτή ATmega 328 συμπεριλαμβάνονται επίσης:

- 32 καταχωρητές γενικής χρήσης.
- 23 προγραμματιζόμενες γραμμές εισόδου/εξόδου (I/O lines).

- Μια μονάδα USART (Universal Synchronous/Asynchronous Receiver Transmitter) μέσω της οποίας επιτυγχάνεται η σειριακή επικοινωνία με διάφορες συσκευές.
- Δύο μονάδες για τη σειριακή επικοινωνία μέσω των διαύλων SPI και I<sup>2</sup>C.
- Μία εξακάναλη μονάδα μετατροπέα αναλογικού σε ψηφιακό (A/D C), διακριτικής ικανότητας 10bit.
- Τρεις χρονιστές/απαριθμητές: timer0 και timer2 διακριτικής ικανότητας 8bit και ο timer1 των 16bit. Το λογισμικό του Arduino εκμεταλλεύεται τους χρονιστές για τη χρονομέτρηση ή την εισαγωγή καθυστέρησης κατά την εκτέλεση των εντολών κάποιου προγράμματος, την εκτέλεση συγκεκριμένων εντολών σε καθορισμένες χρονικές στιγμές ή ανά τακτά χρονικά διαστήματα, κ.ά.

Στη μια πλευρά της πλακέτας του Arduino (Uno R3) και σε αντίστοιχες θηλυκές ακίδες σύνδεσης καταλήγουν 14 από τις ψηφιακές γραμμές εισόδου εξόδου (I/O lines) του μικροελεγκτή AT-Mega 328, οι οποίες λειτουργούν με τάση 5V και μπορούν να δώσουν ή να “αντλήσουν” μέχρι 40mA ηλεκτρικού ρεύματος. Μπορούν για παράδειγμα ως έξοδοι να χρησιμοποιηθούν για να ανάψουν ή να σβήσουν ένα LED ή ως εισοδοί να καταγράψουν την κατάσταση ενός διακόπτη. Όλες οι ψηφιακές εισοδοί/έξοδοι έχουν εσωτερικές pull-up (πρόσδεσης στην τροφοδοσία) αντιστάσεις των 20-50 kΩ οι οποίες εξ ορισμού είναι απενεργοποιημένες [3]. Μερικές από αυτές τις ψηφιακές ακίδες (που αριθμούνται από 0 έως 13), έχουν και ειδικές χρήσεις:

- Οι ακίδες 0 και 1 χρησιμοποιούνται για τη λήψη και μετάδοση σειριακών (ένα bit τη φορά) δεδομένων.
- Οι ακίδες 3, 5, 6, 9, 10, 11 λειτουργούν ως “ψευδοαναλογικές” έξοδοι (PWM) των 8-bit.
- Οι ακίδες 2 και 3 μπορούν να ρυθμιστούν ώστε να προκαλέσουν εξωτερική διακοπή (interrupt) του εκτελούμενου κώδικα στον μικροελεγκτή.
- Οι ακίδες 10, 11, 12 και 13 υποστηρίζουν τη σειριακή επικοινωνία SPI.

Στην ίδια πλευρά της πλακέτας υπάρχει μία ακίδα γείωσης (GND) και επιπλέον οι ακίδες:

- **AREF** μέσω της οποίας υπάρχει η δυνατότητας παροχής εξωτερικής τάσης αναφοράς για τη λειτουργία του αναλογικο-ψηφιακού μετατροπέα (A/D C) του μικροελεγκτή.
- **SCL** και **SDA** που με το κατάλληλο λογισμικό χρησιμοποιούνται για την υλοποίηση του δισύρματου σειριακού πρωτοκόλλου επικοινωνίας I<sup>2</sup>C.

Στην άλλη πλευρά της πλακέτας, στις ακίδες που σημαίνονται από A0 - A5, καταλήγουν οι 6 αναλογικές εισοδοί του Atmega328 οι οποίες μέσω πολυπλέκτη συνδέονται στον εντός του μικροελεγκτή ευρισκόμενο μετατροπέα αναλογικού σε ψηφιακό, που έχει διακριτική ικανότητα 10bit δηλ. ψηφιοποιεί μια αναλογική τάση 0-5V σε 1024 διαφορετικά βήματα. Μεγάλος αριθμός διαφορετικών αισθητήρων (π.χ. θερμοκρασίας, πίεσης, δύναμης, μαγνητικού πεδίου, φωτοπύλες κ.ά.) μπορεί να συνδεθεί στις εισόδους αυτές δίνοντάς μας τη δυνατότητα να πραγματοποιήσουμε μετρήσεις των αντίστοιχων φυσικών μεγεθών. Και οι έξι αναλογικές εισοδοί μπορούν να χρησιμοποιηθούν ως ψηφιακές εισοδοί/έξοδοι (ψηφιακές ακίδες 14 μέχρι 19), ενώ οι αναλογικές εισοδοί A4 και A5 είναι εσωτερικά συνδεδεμένες με τις ακίδες SCL και SDA του δισύρματου σειριακού πρωτοκόλλου επικοινωνίας I<sup>2</sup>C.

Στην ίδια πλευρά της πλακέτας υπάρχει μια σειρά ακίδων σχετικών με την τροφοδοσία του συστήματος, ως εξής:


- **Vin**: Η τάση τροφοδοσίας του συστήματος όταν χρησιμοποιείται εξωτερική πηγή. Μπορεί να χρησιμοποιηθεί και ως είσοδος τροφοδοσίας του συστήματος.
- **5V**: Έξοδος σταθεροποιημένης τάσης 5V. Χρησιμεύει για την τροφοδοσία συσκευών ή αισθητήρων που πρόκειται να συνδεθούν στον Arduino.

- **3.3V:** Έξοδος σταθεροποιημένης τάσης 3,3V.
- **GND:** Γείωση.
- **Reset:** Ακίδα επανεκκίνησης. Το σύστημα επανεκκινεί όταν η ακίδα γειωθεί.
- **IOREF:** Παρέχει ένδειξη για την τάση αναφοράς με την οποία λειτουργεί ο μικροελεγκτής.

Η τροφοδοσία του Arduino μπορεί να γίνει είτε μέσω της θύρας USB είτε από εξωτερική τάση (7-12V). Η πηγή τροφοδοσίας επιλέγεται αυτόματα.


Στην πλακέτα του Arduino, υπάρχουν επίσης:

- Το κύκλωμα χρονισμού του μικροελεγκτή, αποτελούμενο είτε από ένα κρύσταλλο 16MHz, είτε από αντίστοιχης συχνότητας κεραμικό ταλαντωτή, και τα κατάλληλα παθητικά εξαρτήματα.
- Μια διπλή σειρά ακίδων ICSP (In Circuit Serial Programming) για τον προγραμματισμό του μικροελεγκτή μέσω εξωτερικού προγραμματιστή.
- Τα κατάλληλα ηλεκτρονικά εξαρτήματα για την παροχή σταθερής τάσης τροφοδοσίας στο κύκλωμα.
- Βύσμα τύπου jack για την τροφοδοσία του Arduino από εξωτερική πηγή.
- Βύσμα USB για τη σειριακή σύνδεση του Arduino με υπολογιστή.
- Ένας δεύτερος μικροελεγκτής (ATmega 16U2) με τον κρύσταλλο χρονισμού του, που κατάλληλα προγραμματισμένος λειτουργεί ως μετατροπέας σειριακού σε USB για την επικοινωνία του Arduino με ηλεκτρονικό υπολογιστή. Ουσιαστικά με τον τρόπο αυτό ο Arduino αναγνωρίζεται από τον ηλεκτρονικό υπολογιστή ως μια σειριακή θύρα (π.χ. COM5, κλπ). Για τον προγραμματισμό του Atmega 16U2 στην πλακέτα του αυθεντικού Arduino Uno R3 συμπεριλαμβάνεται και μία ακόμη διπλή ακιδοσειρά ICSP, που βρίσκεται δίπλα από τον ATmega 16U2. Σε νεότερους Arduino-κλώνους συναντάμε ένα αρκετά φθηνότερο και εξειδικευμένο ολοκληρωμένο κύκλωμα (CH 340), που έχει αναλάβει τη σειριακή επικοινωνία Arduino-υπολογιστή, μέσω της θύρας USB.
- Διακόπτης επανεκκίνησης (reset).
- 4 ενδεικτικά LED (Power, Tx, Rx και L συνδεδεμένο στην ψηφιακή έξοδο 13) καθώς και μικρός ακόμη αριθμός συνοδευτικών ηλεκτρονικών εξαρτημάτων.




Εικόνα 3: Arduino Uno R3 κλώνος

- AVR
- DIGITAL
- ANALOG
- POWER
- SERIAL
- SPI
- I2C
- PWM
- INTERRUPT



Εικόνα 4 : Οι ακίδες του Arduino Uno R3

Σε επίπεδο υλικού πρέπει να αναφέρουμε πως έχει αναπτυχθεί μεγάλος αριθμός πλακετών (shields) οι οποίες “κουμπώνουν” κατευθείαν πάνω στην πλακέτα του Arduino αυξάνοντας τις δυνατότητες και τη λειτουργικότητά του συστήματος.



Εικόνα 5: Arduino Shields


## Άλλοι Arduinos

Ο Arduino Uno R3 δεν ήταν ούτε το πρώτο ούτε το τελευταίο μοντέλο Arduino που έδωσε στην κυκλοφορία η εταιρεία σχεδίασης και ανάπτυξης. Βασίζονται σε μικροελεγκτές οκτάμπιτους (8 bit) ή τριανταδιάμπιτους (32 bit) της Atmel αλλά και άλλων κατασκευαστών όπως της Nordic Semiconductors, της Espressif ή της Raspberry Pi. Όλα όμως τα μοντέλα προγραμματίζονται με τον ίδιο τρόπο, δηλ. μέσω του ολοκληρωμένου περιβάλλοντος ανάπτυξης (IDE) με τις κατάλληλες προσθήκες/επεκτάσεις στη γλώσσα προγραμματισμού, τις βιβλιοθήκες και το περιβάλλον εργασίας, ώστε να αξιοποιούνται τα επιπλέον χαρακτηριστικά.

Σήμερα η σειρά των συστημάτων Arduino εκτείνεται σε τέσσερις οικογένειες:

### 1. Arduino Nano

Αποτελούν τη συνέχεια και επέκταση του αρχικού Arduino Nano με πλακέτες μικρού μεγέθους, αλλά με ισχυρότερους μικροεπεξεργαστές και πλήρεις επιπλέον χαρακτηριστικών.



Εικόνα 6: Η οικογένεια Arduino Nano.


Από αριστερά: Ο κλασικός Arduino Nano, Nano Every, Nano RP2040 Connect, Nano ESP32, Nano 33 BLE, Nano33 BLE Sense, Nano IoT

Ενδεικτικά αναφέρουμε:

- Το εισαγωγικό μοντέλο Arduino Nano Every με τον ισχυρότερο ATmega4809 επεξεργαστή, που “τρέχει” στα 20 MHz, είναι λογικής 5V και διαθέτει 48 kB μνήμη flash και 6 kB μνήμη RAM.
- Τον Arduino Nano 33 BLE Sense που ενσωματώνει τον 32μπιτο επεξεργαστή nRF52840 της Nordic Semiconductors που ακολουθεί λογική 3,3V, “τρέχει” στα 64 MHz, χρησιμοποιεί 1 MB μνήμης flash και 256 kB RAM, έχει 14 ψηφιακές εισόδους/εξόδους και 8 αναλογικές εισόδους των 12 bit, υποστηρίζει Bluetooth διασύνδεση (BLE πρότυπο), ενώ διαθέτει και πλήθος ενσωματωμένων αισθητήρων, όπως επιταχυνσιόμετρο, γυροσκόπιο, αισθητήρα μαγνητικού πεδίου, μικρόφωνο, αισθητήρα φωτός, χρώματος, ατμοσφαιρικής πίεσης, θερμοκρασίας και υγρασίας.
- Τον Arduino Nano RP2040 Connect που χρησιμοποιεί τον επεξεργαστή RP2040 της Raspberry Pi και μπορεί να προγραμματιστεί είτε στη γλώσσα προγραμματισμού του Arduino (με το ολοκληρωμένο περιβάλλον ανάπτυξης) είτε σε MicroPython.

## 2. Arduino MKR

Περιλαμβάνει σειρά πλακετών, που βασίζονται στον μικροελεγκτή SAMD21 της Atmel (τώρα Microchip) και, εκτός του εισαγωγικού μοντέλου MKR Zero, διαθέτουν δυνατότητες δικτύωσης Bluetooth και WiFi. Προσφέρεται επίσης μια σειρά πλακετών επέκτασης (shields), οι οποίες “κουμπώνουν” πάνω στην πλακέτα του μικροελεγκτή και μεταξύ τους (δημιουργώντας στοίβα) και οι οποίες περιλαμβάνουν διάφορους αισθητήρες, GPS, δικτύωση Ethernet, κ.ά.




Εικόνα 7: Πλακέτες της οικογένειας MKR.  
Από αριστερά: MKR Zero, MKR WiFi 1010, MKR IMU Shield

## 3. Κλασικοί Arduino

Εδώ εκτός από τον Arduino Uno R3 συμπεριλαμβάνονται:

- Ο Arduino Leonardo, βασισμένος στον ATmega32u4, που διαθέτει ενσωματωμένη υποστήριξη USB και συνεπώς δε χρειάζεται κάποιο εξειδικευμένο ολοκληρωμένο κύκλωμα για την υποστήριξη της USB επικοινωνίας.
- Ο Arduino Zero, βασισμένος στο 32μπιτο επεξεργαστή Atmel SAMD21, που αντίθετα από τα άλλα μοντέλα της οικογένειας ακολουθεί λογική 3,3V.
- Η νεότερη προσθήκη ο Arduino Uno R4 με τον επεξεργαστή RA4M1 της Renesas σε δύο εκδόσεις Minima και WiFi. Προσφέρει ισχυρότερο επεξεργαστή, περισσότερη μνήμη για τα προγράμματά μας και την αποθήκευση μεταβλητών, συνεχίζει τη λογική 5V του Uno R3, 14 ψηφιακές εισόδους/εξόδους, 6 αναλογικές εισόδους με ρυθμιζόμενη ανάλυση ( 10, 12 ή 14 bit) και μια έξοδο για μετατροπή ψηφιακού σε αναλογικό (DAC). Και όλα αυτά διατηρώντας την κλασική φόρμα του Arduino Uno R3.
- Στην ίδια οικογένεια συμπεριλαμβάνονται και κάποιες άλλες μικρού μεγέθους πλακέτες από διάφορους άλλους κατασκευαστές, όπως π.χ. Arduino Pro Mini, Arduino Micro, κ.ά.




Εικόνα 8: Από αριστερά: Arduino Leonardo, Arduino Uno R4 Minima και Sparkfun Arduino Micro

#### 4. Arduino Mega

Τα τρία μέλη της οικογένειας είναι:

- Arduino Mega 2560: Βασίζεται στον Atmega2560, διαθέτει 54 ψηφιακές εισόδους/εξόδους, 16 αναλογικές εισόδους των 10bit, 256 kB μνήμης flash και 8 kB RAM.
- Arduino Due: Βασίζεται στον 32μπιτο και λογικής 3.3V Atmel SAM3X8E, διαθέτει 54 ψηφιακές εισόδους/εξόδους, 12 αναλογικές εισόδους των 12bit, 2 εξόδους DAC (ψηφιακό σε αναλογικό), 512 kB μνήμης flash και 96 kB RAM.
- Arduino Giga R1 WiFi: Βασίζεται στον 32μπιτο επεξεργαστή STM32H747XI της STMicroelectronics διαθέτει 76 ψηφιακές εισόδους/εξόδους, 12 αναλογικές εισόδους των 16bit, 2 εξόδους DAC (ψηφιακό σε αναλογικό), 2 MB μνήμης flash και 1 MB RAM.



Εικόνα 9: Από αριστερά: Arduino Mega 2560, Arduino Due


# Το λογισμικό του Arduino

## 1. Bootloader

Ο *bootloader* είναι ένα μικρού μεγέθους πρόγραμμα (0,5 kB στην περίπτωση του Arduino Uno) προεγκατεστημένο από τον κατασκευαστή στη μνήμη flash του μικροελεγκτή, που εκτελείται κατά την τροφοδοσία ή την επανεκκίνηση του Arduino. Ο ρόλος του είναι να επιτρέπει τη μεταφόρτωση των προγραμμάτων μας στη μνήμη του Arduino από τον υπολογιστή μέσω του καλωδίου USB, χωρίς να είναι απαραίτητη για το σκοπό αυτό η χρήση ειδικών συσκευών προγραμματισμού.

## 2. Ολοκληρωμένο περιβάλλον ανάπτυξης (Arduino IDE)

Το ολοκληρωμένο περιβάλλον ανάπτυξης (IDE) του Arduino είναι μια εφαρμογή που εκτελείται στον υπολογιστή, στον οποίο μέσω της θύρας USB συνδέεται ο Arduino. Είναι γραμμένη σε Java, και προέρχεται από το IDE της γλώσσας προγραμματισμού Processing. Η σχεδίαση του IDE είναι μινιμαλιστική. Περιλαμβάνει ένα επεξεργαστή κειμένου για τη συγγραφή των προγραμμάτων (που στη γλώσσα του Arduino αποκαλούνται sketches), ενώ με τη χρήση εξωτερικών προγραμμάτων εκτελεί τη μεταγλώττιση του sketch σε κώδικα κατανοητό από τον μικροελεγκτή (“γλώσσα μηχανής”), και μεταφορτώνει με ένα “κλικ” τον κώδικα στον Arduino.




Εικόνα 10: Το παράθυρο του ολοκληρωμένου περιβάλλοντος εργασίας του Arduino

Στο κύριο παράθυρο του Arduino IDE περιλαμβάνονται :

- Η γραμμή μενού.
- Η γραμμή εργαλείων (Επικύρωση, Ανέβασμα, Δημιουργία, Άνοιγμα, Αποθήκευση και Παρακολούθηση σειριακής).
- Η περιοχή με τις καρτέλες του επεξεργαστή κειμένου (για τη συγγραφή των sketches).
- Η περιοχή εμφάνισης διαφόρων μηνυμάτων (π.χ. μηνύματα λάθους του μεταγλωττιστή, μηνύματα κατά το ανέβασμα του κώδικα στον Arduino, κ.λ.π.).
- Η γραμμή κατάστασης με τις πληροφορίες για το είδος της χρησιμοποιούμενης πλακέτας, και τη σειριακή θύρα μέσω της οποίας γίνεται η επικοινωνία υπολογιστή - Arduino.

Το ολοκληρωμένο περιβάλλον ανάπτυξης περιλαμβάνει και μια **σειριακή κονσόλα** (τερματικό), μέσω της οποίας ο χρήστης μπορεί να αλληλεπιδρά με τον Arduino. Ο Arduino μπορεί προγραμματιστικά να αποστέλλει πληροφορίες, οι οποίες θα εμφανίζονται στο τερματικό, ή να λαμβάνει πληροφορίες που του αποστέλλονται μέσω του τερματικού. Θα πούμε περισσότερα για τη σειριακή κονσόλα στη συνέχεια.

Τέλος με τον **διαχειριστή πλακετών** και το **διαχειριστή βιβλιοθηκών** το λογισμικό του Arduino μπορεί να αναβαθμίζεται, είτε για τη βελτίωση της λειτουργίας των υποστηριζόμενων πλακετών και δυνατοτήτων, είτε για την υποστήριξη και άλλων νεότερων πλακετών μικροελεγκτών ή την εγκατάσταση επιπλέον δυνατοτήτων.



Εικόνα 11: Το παράθυρο του διαχειριστή πλακετών

Παρότι ο έμπειρος προγραμματιστής (στον οποίο έτσι κι αλλιώς δεν αναφερόμαστε εδώ) θα βρει την έκδοση 1 του λογισμικού αρκετά περιοριστική, το ολοκληρωμένο περιβάλλον ανάπτυξης του Arduino παρέχει στον αρχάριο όλα όσα χρειάζεται για τη διαχείριση του Arduino από τον υπολογιστή του.

### 3. Βιβλιοθήκες

Οι τυπικές βιβλιοθήκες που συνοδεύουν τον Arduino είναι αρχεία κώδικα γραμμένα σε C ή C++, που παρέχουν επιπλέον λειτουργικότητα στη διαχείριση του υλικού και των δεδομένων και βελτιώνουν την αναγνωσιμότητα του κώδικα. Η βιβλιοθήκη πυρήνα (core library) συμπυκνώνει

χαμηλού επιπέδου πτυχές του προγραμματισμού του μικροελεγκτή (π.χ. διαχείριση καταχωρητών), επιτρέποντας στους χρήστες να επικεντρωθούν στο κάθε φορά ιδιαίτερο έργο τους, και όχι στην χρονοβόρα και επίπονη διαδικασία εκμάθησης του προγραμματισμού σε χαμηλό επίπεδο (γλώσσα μηχανής). Η βιβλιοθήκη πυρήνα εξ ορισμού συμπεριλαμβάνεται σε οποιοδήποτε πρόγραμμά μας (χωρίς αυτό να απαιτεί κάποια ιδιαίτερη ενέργεια από εμάς). Λόγω της περιορισμένης μνήμης του μικροελεγκτή, μέρος του κώδικα έχει διαχωριστεί σε επιμέρους βιβλιοθήκες, οι οποίες μπορεί κατά περίπτωση να συμπεριληφθούν όταν απαιτείται σε κάποιο πρόγραμμα. Μεταξύ αυτών περιλαμβάνονται:

- **EEPROM** - διαχείριση της μνήμης EEPROM του μικροελεγκτή
- **Ethernet** - Για σύνδεση στο διαδίκτυο μέσω του Ethernet Shield
- **Firmata** - Για την επικοινωνία με εφαρμογές στον υπολογιστή χρησιμοποιώντας το τυπικό σειριακό πρωτόκολλο.
- **GSM** - για τη σύνδεση σε ένα δίκτυο GSM/GRPS με το GSM shield.
- **LiquidCrystal** - για τον έλεγχο οθόνης υγρών κρυστάλλων (LCD).
- **SD** - διαχείριση καρτών μνήμης SD.
- **Servo** - για τον έλεγχο σερβοκινητήρων.
- **SPI** - για την επικοινωνία με συσκευές μέσω του Serial Peripheral Interface (SPI) Bus
- **SoftwareSerial** - για τη σειριακή επικοινωνία με χρήση οποιωνδήποτε ψηφιακών ακίδων.
- **Stepper** - για τον έλεγχο βηματικών κινητήρων.
- **TFT** - για τον έλεγχο οθόνης TFT.
- **WiFi** - Για ασύρματη σύνδεση στο διαδίκτυο μέσω του Arduino WiFi shield.
- **Wire** - για την επικοινωνία με συσκευές μέσω του Two Wire Interface (TWI/I<sup>2</sup>C).

Ιδιαίτερη προσπάθεια έχει γίνει, ώστε οι βιβλιοθήκες πυρήνα να απλοποιούν τη διαδικασία συγγραφής του κώδικα χωρίς να περιορίζουν υπερβολικά την ευελιξία του χρήστη. Παρ' όλα αυτά διαδικασίες όπως η ανάγνωση μιας ψηφιακής εισόδου ή η εγγραφή μιας τιμής σε κάποια ψηφιακή έξοδο (`digitalRead` ή `digitalWrite`) ή η ψηφιακή μετατροπή και ανάγνωση της επιστρεφόμενης τιμής σε μια αναλογική είσοδο (`analogRead`) -διαδικασίες βασικές για ένα σύστημα μετρήσεων- μπορούν εναλλακτικά να υλοποιηθούν, ώστε να τρέχουν πολλές φορές γρηγορότερα.

## Η κοινότητα

Αποτελεί την τρίτη συνιστώσα της επιτυχίας του Arduino. Το ανοιχτό υλικό, ο ανοιχτός κώδικας και η ευκολία διαχείρισης του συστήματος συνέβαλλαν καθοριστικά στη δημιουργία μιας κοινότητας περί το project. Ίσως η σημαντικότερη συνεισφορά της είναι το πλήθος των επιπλέον βιβλιοθηκών που έχει αναπτυχθεί από τα μέλη της κοινότητας των χρηστών του Arduino αυξάνοντας τόσο τη λειτουργικότητα όσο και την απόδοση της πλατφόρμας.

# Στοιχεία προγραμματισμού του Arduino

## Τα βασικά

Η δύναμη της γλώσσας προγραμματισμού C/C++ κρύβεται πίσω από ένα πρόγραμμα (sketch) του Arduino, παρότι η δομή του sketch δεν είναι η τυπική ενός προγράμματος C/C++. Μόνο δύο συναρτήσεις είναι απαραίτητες για τη δημιουργία ενός απλού προγράμματος κυκλικής εκτέλεσης στον Arduino:


- Η συνάρτηση **setup()**: Εκτελείται μόνο μια φορά στην αρχή του προγράμματος, και καθορίζει τις αρχικές ρυθμίσεις του συστήματος, π.χ. ορισμός εισόδων/εξόδων, ενεργοποίηση σειριακής επικοινωνίας, κλπ.
- Η συνάρτηση **loop()**: Περιέχει τον κυρίως κώδικα του προγράμματος μας, και εκτελείται κυκλικά μέχρι την απενεργοποίηση ή την επανεκκίνηση του Arduino.

Και οι δύο συναρτήσεις είναι τύπου **void**, που σημαίνει πως κατά την εκτέλεσή τους δεν επιστρέφουν κάποια τιμή (αποτέλεσμα) στον Arduino.

## Πριν το πρώτο sketch - προαπαιτούμενες ρυθμίσεις


Για την ολοκλήρωση της διαδικασίας μεταγλώττισης και “ανεβάσματος” του κώδικά μας στον Arduino, στο ολοκληρωμένο περιβάλλον εργασίας (Arduino IDE) απαιτούνται οι εξής ρυθμίσεις:

1. Να επιλέξουμε μέσω του μενού “Εργαλεία/Πλακέτα” το μοντέλο του Arduino (ή άλλης συμβατής πλακέτας) που χρησιμοποιούμε, π.χ. Arduino/Genuino Uno, Arduino Leonardo, Arduino Mini, κ.λ.π.



Εικόνα 12: Επιλογή πλακέτας


- Επιπλέον να επιλέξουμε, μέσω του μενού “*Εργαλεία/Θύρα*”, τη σειριακή θύρα (π.χ. COM5) μέσω της οποίας η πλακέτα του Arduino επικοινωνεί με τον υπολογιστή στον οποίο συνδέεται.



Εικόνα 13 : Επιλογή σειριακής θύρας

Στην περίπτωση που η σειριακή θύρα, στην οποία ο υπολογιστής “συνομιλεί” με τον Arduino, δεν είναι γνωστή, μπορούμε να την εντοπίσουμε ως εξής: Αφού αποσυνδέσουμε το USB καλώδιο από τον Arduino, μέσω του μενού “*Εργαλεία/Θύρα*” του Arduino IDE ελέγχουμε τη λίστα με τις διαθέσιμες σειριακές θύρες. Στη συνέχεια και αφού προηγουμένως κλείσουμε το μενού (κάνοντας κλικ με το ποντίκι σε κάποιο σημείο του Arduino IDE έξω από το μενού), συνδέουμε τον Arduino με το καλώδιο USB, και τέλος ελέγχουμε και πάλι τις διαθέσιμες σειριακές θύρες. Η νέα εγγραφή, που εμφανίζεται στη λίστα με τις διαθέσιμες σειριακές θύρες, είναι η θύρα στην οποία ο υπολογιστής “συνομιλεί” με τον Arduino.

Εναλλακτικά, σε υπολογιστές με κάποια από τις νεότερες εκδόσεις του λειτουργικού συστήματος Windows, μπορούμε να αναζητήσουμε τη σχετική σειριακή θύρα ως εξής: Αφού αποσυνδέσουμε το USB καλώδιο από τον Arduino, ανοίγουμε τον “*Πίνακα Ελέγχου*” των Windows, επιλέγουμε την εμφάνιση της καρτέλας “*Υλικό και Ήχος*”, και στην περιοχή εφαρμογών “*Συσκευές και εκτυπωτές*” επιλέγουμε την εφαρμογή “*Διαχείριση συσκευών*”. Στην εμφανιζόμενη λίστα με τις κατηγορίες των συνδεδεμένων συσκευών επιλέγουμε την εμφάνιση (“*Ξεδίπλωμα*” της λίστας) των συσκευών τύπου “*Θύρες (COM & LPT)*”. Μετά συνδέουμε και πάλι με το USB καλώδιο τον Arduino στον υπολογιστή. Μια νέα σειριακή θύρα αναγνωρίζεται από τα Windows και εμφανίζεται στον πίνακα των συνδεδεμένων συσκευών. Αυτή είναι η σειριακή θύρα μέσω της οποίας επιτυγχάνεται η επικοινωνία Arduino - υπολογιστή.



Εικόνα 14: Διαχείριση συσκευών - στη θύρα COM5 ο υπολογιστής "συνομιλεί" με τον Arduino (CH-340 κλώνο)

Τόσο η επιλεγμένη πλακέτα, όσο και η επιλεγμένη σειριακή θύρα αναγράφονται στη γραμμή κατάστασης του Arduino IDE (Εικόνα 6).

## Το πρώτο sketch - χρήση ψηφιακής ακίδας ως εξόδου

Στην πλακέτα του Arduino ένα LED, μέσω κατάλληλης αντίστασης, συνδέεται στην ψηφιακή ακίδα 13. Το πρώτο απλό πρόγραμμα κάνει χρήση αυτού του χαρακτηριστικού:

```
/*
   Ex.1 : Αναβοσβήνοντας το ενσωματωμένο στον Arduino LED.
*/
void setup ()
{
    pinMode (13, OUTPUT);           // Η ακίδα 13 καθορίζεται ως ψηφιακή έξοδος
}

void loop ()
{
    digitalWrite (13, HIGH);       // Αναψε το LED
    delay (1000);                  // Αναμονή 1sec (1000 milliseconds)
    digitalWrite (13, LOW);        // Σβήσε το LED
    delay (1000);                  // Αναμονή 1sec
}
```

**Σημείωση:** Σε μια γραμμή του sketch ότι υπάρχει μετά τις δύο γραμμές “//” θεωρείται επεξηγηματικό σχόλιο. Σχόλια που εκτείνονται σε περισσότερες γραμμές περικλείονται μεταξύ των “/\*” και “\*/” (χωρίς τα εισαγωγικά εννοείται). Το σώμα των εντολών μιας οποιασδήποτε συνάρτησης πρέπει να περικλείεται μεταξύ δύο αγκιστρών {...}. Γενικά χρησιμοποιούμε τα δύο άγκιστρα για να δημιουργήσουμε ένα μπλοκ εντολών, οι οποίες εκτελούνται υπό συγκεκριμένες συνθήκες στο πρόγραμμά μας.

Παρότι το όνομα των χρησιμοποιούμενων εντολών χαρακτηρίζει με ικανοποιητική σαφήνεια και τη λειτουργία τους, ας σχολιάσουμε κάποια σημεία:

- Η εντολή *pinMode(pin, mode)* καθορίζει αν μια ψηφιακή ακίδα (pin) θα χρησιμοποιείται από το πρόγραμμά μας ως έξοδος (OUTPUT) ή ως είσοδος (INPUT). Μια επιπλέον ειδική επιλογή μιας ακίδας ως εισόδου είναι η INPUT\_PULLUP με την οποία ενεργοποιείται μια εσωτερική αντίσταση και η αντίστοιχη είσοδος προσδένεται στην τάση τροφοδοσίας [4].
- Με την εντολή *digitalWrite(pin, value)* “γράφουμε” την τιμή HIGH (1) ή LOW (0) σε μια ψηφιακή έξοδο (παράμετρος pin) [4]. Στην περίπτωση του Arduino Uno λογική τιμή HIGH σε μια ψηφιακή ακίδα σημαίνει ότι θέτουμε την τάση στην ακίδα αυτή στην τιμή 5 V. Αντίστοιχα λογική τιμή LOW σημαίνει πρόσδεση της ακίδας αυτής στη γείωση (τάση 0 V).
- Η εντολή *delay(value)* διακόπτει την εκτέλεση του κώδικα για χρόνο (σε ms) ίσο με την τιμή της παραμέτρου value [4]. Γενικά η χρήση της πρέπει να αποφεύγεται για μεγάλες χρονικές διάρκειες παύσης (εκτός βέβαια από την περίπτωση πολύ απλών προγραμμάτων), αφού κατά τη διάρκεια της λειτουργίας της, και πέρα από κάποιες εξειδικευμένες λειτουργίες (π.χ. διακοπές υλικού ή χρονιστή), παύει κάθε δραστηριότητα του μικροελεγκτή.

Αφού αποθηκεύσουμε το sketch μέσω του μενού “Αρχείο/Αποθήκευση ως...” με κάποιο όνομα της αρεσκείας μας, κάνουμε στη συνέχεια κλικ στο κουμπί “Ανέβασμα” (ή “Upload to I/O Board”) στη γραμμή εργαλείων του IDE του Arduino, και το λογισμικό αναλαμβάνει:

- Να πραγματοποιήσει ένα συντακτικό προέλεγχο του προγράμματος.
- Να τροποποιήσει το πρόγραμμα ώστε να το μετατρέψει σε έγκυρο C++ πρόγραμμα.

- Να μεταγλωττίσει το πρόγραμμα με χρήση του εξωτερικού προγράμματος *avr-gcc*, δημιουργώντας ένα αρχείο που περιέχει τον κώδικα σε μορφή κατανοητή από τον μικροελεγκτή (γλώσσα μηχανής σε δεκαεξαδικό κώδικα).
- Να ανεβάσει το αρχείο με τον κώδικα στη μνήμη του Arduino, με χρήση του εξωτερικού προγράμματος *avrdude*, και αφού προηγουμένως έχει εξαναγκάσει τον Arduino σε επανεκκίνηση - reset ((ώστε να "αναλάβει δράση" ο bootloader του Arduino).

Μετά το ανέβασμα του κώδικα στον Arduino αρχίζει αυτόματα να εκτελείται, μέχρι να αποσυνδέσουμε τον Arduino από την τροφοδοσία ή να εξαναγκάσουμε το σύστημα σε επανεκκίνηση.

**Ας τονίσουμε για μια ακόμη φορά τις προαπαιτούμενες για κάθε νέο sketch ρυθμίσεις:**

1. Μέσω του μενού “*Εργαλεία/Πλακέτα*”, επιλέγουμε το μοντέλο του Arduino (ή άλλης συμβατής πλακέτας) που χρησιμοποιούμε, π.χ. Arduino/Genuino Uno, Arduino Leonardo, Arduino Mini, κ.λ.π..
2. Μέσω του μενού “*Εργαλεία/Θύρα*”, επιλέγουμε τη σειριακή θύρα (π.χ. COM5) μέσω της οποίας η πλακέτα του Arduino επικοινωνεί με τον υπολογιστή στον οποίο συνδέεται.

## Παρατήρηση

Στο περιβάλλον ανάπτυξης του Arduino περιλαμβάνεται η σταθερά *LED\_BUILTIN* που αντιστοιχεί στην ψηφιακή ακίδα της πλακέτας στην οποία είναι συνδεδεμένο το LED. Αυτό απλοποιεί τη διαδικασία προγραμματισμού, καθώς δεν είναι απαραίτητο θα θυμόμαστε για κάθε πλακέτα τη συγκεκριμένη ακίδα στην οποία συνδέεται το LED.

Έτσι για παράδειγμα γράφουμε:


*pinMode (LED\_BUILTIN, OUTPUT)* ή *digitalWrite (LED\_BUILTIN, HIGH)* κλπ

και ο μεταγλωττιστής που κάθε φορά χρησιμοποιείται αντικαθιστά το συμβολικό όνομα με την τιμή που αντιστοιχεί στην ακίδα της πλακέτας (13 στην περίπτωση του Uno R3) στην οποία είναι συνδεδεμένο το LED.

## Σύνδεση LED σε άλλη ψηφιακή έξοδο - τροποποίηση του σχετικού sketch


Η δίοδος εκπομπής φωτός (Light Emitting Diode ή απλά LED) είναι ένα σύστημα αποτελούμενο από ένα ημιαγωγό τύπου  $p$  και ένα ημιαγωγό τύπου  $n$  σε άμεση επαφή που, όταν συνδεθεί σε ηλεκτρική τάση κατά την ορθή φορά πόλωσης (δηλ. ο  $p$ -ημιαγωγός σε υψηλότερο δυναμικό σε σχέση με τον  $n$ -ημιαγωγό), εκπέμπει φως στενού φασματικού εύρους. Η διάταξη περικλείεται μέσα σε πλαστικό έγχρωμο ή διαφανές περιβλήμα. Το χρώμα του φωτός που εκπέμπεται μπορεί να είναι υπέρυθρο, ορατό, ή υπεριώδες, και εξαρτάται από τη χημική σύσταση του ημιαγωγικού υλικού που χρησιμοποιείται.

Τα απλά LED διαθέτουν δυο ακροδέκτες για τη σύνδεσή τους σε οποιοδήποτε κύκλωμα: Το θετικό άκρο, που είναι ο μεγαλύτερου μήκους ακροδέκτης και ονομάζεται “άνοδος”, και το αρνητικό άκρο που ονομάζεται “κάθοδος”. Σε περίπτωση που οι δύο ακίδες έχουν το ίδιο μήκος, η κάθοδος μπορεί να αναγνωριστεί, καθώς είναι η ακίδα που βρίσκεται πλησιέστερα στην επίπεδη πλευρά του πλαστικού περιβλήματος του LED. Το LED συνδέεται στην ηλεκτρική τάση σε σειρά με μια αντίσταση για τον περιορισμό του ηλεκτρικού ρεύματος, και κατά τέτοιο τρόπο ώστε η άνοδος να βρίσκεται σε υψηλότερο δυναμικό σε σχέση με την κάθοδο (ορθή φορά πόλωσης).



Εικόνα 15: Η δίοδος εκπομπής φωτός (LED)

Θα χρησιμοποιήσουμε ένα LED, το οποίο -μέσω κατάλληλης αντίστασης (μας κάνουν αντιστάσεις από  $220\Omega$  μέχρι  $1k\Omega$ ) για τον περιορισμό του ρεύματος- θα συνδέσουμε σε κάποια ψηφιακή ακίδα του Arduino (π.χ. στην ακίδα 7). Οι σχετικές συνδέσεις, που γίνονται με τη βοήθεια κατάλληλων καλωδίων και breadboard, φαίνονται στο ακόλουθο σχήμα:



Εικόνα 16: Σύνδεση LED στον Arduino

Το πρόγραμμα (sketch) του Arduino για το διαδοχικό άναμμα και σβήσιμο του LED διαμορφώνεται ως εξής:

```

/*
   Ex.2 : Αναβοσβήνουμε ένα LED που συνδέεται στην ψηφιακή ακίδα 7 του Arduino
*/

void setup ()
{
    pinMode (7, OUTPUT);           // Η ακίδα 7 καθορίζεται ως ψηφιακή έξοδος
}

void loop ()
{
    digitalWrite (7, HIGH);        // Αναψε το LED
    delay (1000);                  // Αναμονή 1sec (1000 milliseconds)
    digitalWrite (7, LOW);         // Σβήσε το LED
    delay (1000);                  // Αναμονή 1sec
}

```

Αν θέλουμε να αλλάξουμε την ψηφιακή ακίδα σύνδεσης του LED στον Arduino, οι αλλαγές στον κώδικα διευκολύνονται με χρήση ενός ορισμού **#define**. Για παράδειγμα:

```

/*
   Ex.3 : Αναβοσβήνουμε ένα LED που συνδέεται σε κάποια ψηφιακή ακίδα του Arduino
*/

#define LEDPIN      8

void setup ()
{
    pinMode (LEDPIN, OUTPUT);      // Η ακίδα LEDPIN καθορίζεται ως ψηφιακή έξοδος
}


void loop ()
{
    digitalWrite (LEDPIN, HIGH);   // Αναψε το LED
    delay (1000);                  // Αναμονή 1sec (1000 milliseconds)
    digitalWrite (LEDPIN, LOW);    // Σβήσε το LED
    delay (1000);                  // Αναμονή 1sec
}

```

Ας παρατηρήσουμε ότι στο sketch όλες οι αριθμητικές αναφορές στην ακίδα που συνδέεται το LED έχουν αντικατασταθεί με το λεκτικό LEDPIN. Στην περίπτωση αυτή, πριν το τελικό στάδιο της μεταγλώττισης, κάθε εμφάνιση του ονόματος LEDPIN στο sketch αυτόματα αντικαθίσταται από το λογισμικό με τη σταθερή τιμή 8. Έτσι αν θέλουμε να αλλάξουμε την ακίδα σύνδεσης του LED στον Arduino (και εκτός από τις όποιες αλλαγές στο κύκλωμα), στο sketch χρειάζεται μόνο να αλλάξουμε την αριθμητική τιμή που εμφανίζεται στον ορισμό LEDPIN.

## Χρήση ψηφιακής ακίδας ως εισόδου


Στα προηγούμενα sketch είδαμε πώς μπορούμε να χρησιμοποιήσουμε μια ψηφιακή ακίδα του Arduino ως έξοδο, και συνεπώς πώς γράφουμε -με την εντολή `digitalWrite()`- στην ακίδα αυτή μια ψηφιακή πληροφορία, δηλ. λογικό 1 ή 0 (HIGH ή LOW). Αυτό έχει σαν αποτέλεσμα την εμφάνιση τάσης +5V ή 0V αντίστοιχα στην ακίδα, και άρα μπορεί να χρησιμεύσει για να ανάψει ή να σβήσει ένα LED, που συνδέεται στην ίδια ακίδα. Θα εξετάσουμε τώρα τη χρήση μιας ψηφιακής ακίδας ως εισόδου, από την οποία θα ανασύρουμε την ψηφιακή πληροφορία που κάποιος εξωτερικός παράγοντας (ένα κατάλληλο ηλεκτρικό κύκλωμα) έχει “εγγράψει”. Θα διαβάσουμε δηλαδή την τιμή της τάσης στην ακίδα αυτή με την εντολή `digitalRead()`. Το απλούστερο κύκλωμα που μπορεί να “εγγράψει” ψηφιακή πληροφορία (δηλ. τάση 0 ή 5V) σε μια ψηφιακή ακίδα μπορεί απλά να είναι ένας στιγμιαίος διακόπτης (momentary push button) σε σειρά με μία αντίσταση. Το σύστημα τροφοδοτείται από τάση 5V, και το κοινό σημείο της αντίστασης και του διακόπτη οδηγείται σε κάποια ψηφιακή ακίδα (π.χ. την ακίδα 7) του Arduino.



Εικόνα 17: Σχηματικό διάγραμμα σύνδεσης διακόπτη στον Arduino

Όταν ο διακόπτης είναι ανοικτός η ψηφιακή ακίδα μέσω της (pull down) αντίστασης των 10 kΩ οδηγείται στη γη (γειώνεται), άρα λογικό 0 εγγράφεται στην ακίδα. Όταν ο διακόπτης κλείσει η ακίδα οδηγείται σε υψηλό δυναμικό +5V (λογικό 1).

Οι αντίστοιχες συνδέσεις στον Arduino φαίνονται στο επόμενο σχήμα:



Εικόνα 18: Σύνδεση διακόπτη push button στον Arduino

Ο (ημιτελής προς το παρόν) κώδικας του Arduino, μπορεί να έχει τη μορφή:

```

/*
  Ανάγνωση ψηφιακής ακίδας του Arduino
*/

#define BUTTONPIN      7

void setup ()
{
  pinMode (BUTTONPIN, INPUT);      // Η ακίδα 7 καθορίζεται ως ψηφιακή είσοδος
}

void loop ()
{
  int value = digitalRead (BUTTONPIN);    // Διάβασε την κατάσταση του διακόπτη

  // Εδώ πρέπει να κάνουμε κάτι, ανάλογα με την τιμή της ακίδας 7
}

```

Το αποτέλεσμα της ανάγνωσης της ψηφιακής ακίδας BUTTONPIN (7), το αποθηκεύουμε σε μια μεταβλητή με το περιγραφικό όνομα *value*. Μπορούμε να σκεφτόμαστε μια μεταβλητή ως μια περιοχή της μνήμης RAM του Arduino, στην οποία μπορούμε να γράφουμε κάποια τιμή, να την τροποποιούμε, και να τη διαβάζουμε κατά την εκτέλεση ενός προγράμματος. Επιπλέον ενημερώνουμε τον μεταγλωττιστή πως η μεταβλητή *value* είναι ακέραίου (*int*) τύπου, ώστε να δεσμευτεί κατάλληλου μεγέθους περιοχή της μνήμης (περισσότερα για τους τύπους δεδομένων που υποστηρίζει ο Arduino στο Παράρτημα Α). Στο sketch η τιμή της μεταβλητής *value* γίνεται 1 (HIGH) όταν ο διακόπτης είναι πατημένος, και 0 (LOW) όταν ο διακόπτης απελευθερωθεί.

**Παρατήρηση :** Ο ορισμός μιας μεταβλητής (*variable*) ακολουθεί το γενικό σχήμα:

**Επιστρεφόμενος\_τύπος** Όνομα\_μεταβλητής = αρχική τιμή;

Η απόδοση κάποιας αρχικής τιμής στη μεταβλητή είναι προαιρετική. Μεταβλητές που ορίζονται μέσα στο σώμα μιας συνάρτησης έχουν ισχύ (δηλ. είναι προσβάσιμες για εγγραφή ή ανάγνωση) μόνο εντός της συνάρτησης και χαρακτηρίζονται ως **τοπικές** (*local*), ενώ μεταβλητές που ορίζονται έξω από οποιαδήποτε συνάρτηση έχουν καθολική ισχύ και χαρακτηρίζονται ως **καθολικές** (*global*). Εκτός από μεταβλητές μπορούμε στα προγράμματά μας να ορίσουμε και σταθερές (*constants*), δηλ. τιμές που δε μπορούν να μεταβληθούν κατά την εκτέλεση του προγράμματος, σύμφωνα με το γενικό σχήμα:

**const Επιστρεφόμενος\_τύπος** Όνομα\_μεταβλητής = αρχική τιμή;

Για παράδειγμα με τη δήλωση:

**const float** pi = 3.14159;

ορίζουμε μια σταθερά τύπου κινητής υποδιαστολής, με το όνομα *pi* και τιμή 3,14159.

Βέβαια δεν είναι αρκετό να διαβάσουμε την τιμή μιας ψηφιακής ακίδας, αλλά χρειάζεται το πρόγραμμά μας να παίρνει αποφάσεις, και να εκτελεί τις αντίστοιχες εντολές ανάλογα με την κατάσταση της ακίδας. Για το λόγο αυτό στο παραπάνω ημιτελές sketch υπάρχει το σχόλιο:

*// Εδώ πρέπει να κάνουμε κάτι, ανάλογα με την τιμή της ακίδας 7*

Ας δεχτούμε στο σημείο αυτό πως θέλουμε, όταν ο διακόπτης είναι κλειστός, να ανάβει το LED που συνδέεται στην ακίδα 13 του Arduino (και είναι, όπως έχουμε ήδη πει, ενσωματωμένο στην πλακέτα του Arduino Uno), και όταν ο διακόπτης είναι ανοιχτός το LED να είναι σβηστό. Μια πρώτη προσέγγιση στο πρόβλημα είναι η εξής: Η ψηφιακή ακίδα 13, που πρέπει να έχει χαρακτηριστεί ως έξοδος (OUTPUT), θα πρέπει να βρίσκεται πάντα στην ίδια κατάσταση (HIGH ή LOW)

με την ακίδα 7, στην οποία είναι συνδεδεμένος ο διακόπτης.

Το αντίστοιχο sketch παίρνει τη μορφή:

```
/*
   Ex.4 : Ανάγνωση ψηφιακής ακίδας του Arduino
*/

#define BUTTONPIN      7
#define LEDPIN         13

void setup ()
{
    pinMode(BUTTONPIN, INPUT);           // Η ακίδα 7 καθορίζεται ως ψηφιακή είσοδος
    pinMode(LEDPIN, OUTPUT);             // Η ακίδα 13 καθορίζεται ως ψηφιακή έξοδος
}
void loop ()
{
    int value = digitalRead(BUTTONPIN);   // Διάβασε την κατάσταση του διακόπτη

    digitalWrite(LEDPIN, value);         // Στην ακίδα 13 γράφουμε την τιμή της ακίδας 7
}
```

Οι δύο εντολές στη συνάρτηση *loop()*, θα μπορούσαν να αντικατασταθούν με μία, ως εξής:

*digitalWrite(LEDPIN, digitalRead(BUTTONPIN));*

Ο κατ' αυτό τον τρόπο συνδυασμός των δύο εντολών έχει το πλεονέκτημα πως δεν απαιτείται δέσμευση της περιορισμένης μνήμης RAM για την αποθήκευση της κατάστασης της ακίδας 7 (BUTTONPIN). Φυσικά, στο συγκεκριμένο απλό sketch η περιορισμένη μνήμη RAM του Arduino δεν αποτελεί πρόβλημα.

Ας σημειώσουμε πως με το τελευταίο sketch ο Arduino είναι σε θέση αυτόματα να λαμβάνει μια απλή απόφαση: Πότε να ανάβει και πότε να σβήνει ένα LED. Θα συζητήσουμε σε επόμενη παράγραφο πώς μπορούμε να υλοποιούμε πολυπλοκότερες διαδικασίες λήψης αποφάσεων στον Arduino. Νωρίτερα όμως θα συζητήσουμε το θέμα της σειριακής επικοινωνίας μεταξύ Arduino και άλλων συσκευών, π.χ. του υπολογιστή με τον οποίο ο Arduino συνδέεται μέσω καλωδίου USB.

**Παρατήρηση :** Ο Arduino αναγνωρίζει ως λογικό 1 (HIGH) σε μια είσοδό του οποιαδήποτε τάση μεγαλύτερη από 3V, και ως λογικό 0 (LOW) οποιαδήποτε τάση μικρότερη από 1,5V.


## Σειριακή επικοινωνία μεταξύ Arduino και υπολογιστή ή άλλων συσκευών

Η σειριακή επικοινωνία μεταξύ δύο συσκευών είναι μια διαδικασία αποστολής δεδομένων διαδοχικά -ένα bit τη φορά- μέσω ενός διαύλου. Στην απλούστερη υλοποίησή του ένας τέτοιος δίαυλος αποτελείται από δύο αγωγούς: ένα για την αποστολή και ένα για τη λήψη δεδομένων. Αντίστοιχα οι συνδεδεμένες συσκευές πρέπει να διαθέτουν δύο σειριακές ακίδες:

- RX για τη λήψη δεδομένων
- TX για την αποστολή δεδομένων

Η επιτυχής σειριακή επικοινωνία δύο συσκευών προϋποθέτει:

1. Σταυρωτή σύνδεση των ακίδων RX και TX των δύο συσκευών, δηλ. η ακίδα RX της μιας συσκευής στην ακίδα TX της άλλης και αντίστροφα (Εικόνα 15).
2. Ίδιο ρυθμό επικοινωνίας (baud rate σε bits per second ή bps) στις δύο συσκευές.
3. Ίδια επίπεδα τάσης στις σειριακές ακίδες των δύο συσκευών, δηλ. δε μπορούμε να συνδέσουμε απευθείας τις σειριακές ακίδες του Arduino (επίπεδα τάσης 5V TTL) στις αντίστοιχες ακίδες μιας σειριακής θύρας υπολογιστή RS232 (που έχει επίπεδα τάσης 12V TTL).



Εικόνα 19: Σειριακή σύνδεση δύο συσκευών

Έχουμε ήδη χρησιμοποιήσει -χωρίς να το αναφέρουμε- τις δυνατότητες σειριακής επικοινωνίας του Arduino με τον υπολογιστή: Με τον τρόπο αυτό το Arduino IDE “ανεβάζει” τα μεταγλωττισμένα sketch στον Arduino. Μάλιστα κάποιος προσεκτικός πιθανόν να έχει ήδη παρατηρήσει πως, κατά τη διάρκεια του ανεβάσματος του sketch, τα LED Rx και Tx της πλακέτας του Arduino αναβοσβήνουν.

Στο υλικό του μικροελεγκτή του Arduino Uno συμπεριλαμβάνεται μια μονάδα (USART) που εγγενώς υποστηρίζει τη σειριακή επικοινωνία (Hardware Serial), και υλοποιείται μέσω των ψηφιακών ακίδων 0 και 1: Τα δεδομένα λαμβάνονται από τον Arduino μέσω της ακίδας 0 (RX) και στέλνονται απ' αυτόν μέσω της ακίδας 1 (TX), και για το λόγο αυτό, όταν το sketch μας χρησιμοποιεί τη δυνατότητα σειριακής επικοινωνίας, οι ακίδες 0 και 1 δεν είναι διαθέσιμες (δηλ. δε μπορούν να χρησιμοποιηθούν από το πρόγραμμά μας) ως ψηφιακές είσοδοι ή έξοδοι.

Αν θέλουμε να χρησιμοποιήσουμε στο sketch μας τη δυνατότητα σειριακής επικοινωνίας του Arduino με τον υπολογιστή (και ουσιαστικά να χρησιμοποιήσουμε τη σειριακή κονσόλα του Arduino IDE ως τη “διαδραστική” οθόνη που δε διαθέτει ο Arduino), θα πρέπει πρώτα να εκκινήσουμε την επικοινωνία και να καθορίσουμε το ρυθμό επικοινωνίας.

Για παράδειγμα η εντολή:


```
Serial.begin(9600);
```

ξεκινάει τη σειριακή επικοινωνία με ρυθμό 9600 bps, και πρέπει να συμπεριλαμβάνεται στη συνάρτηση *setup()* του sketch μας. Για να επιτευχθεί σωστή επικοινωνία πρέπει απαραίτητα και οι δύο πλευρές (ο Arduino από τη μια, και ο υπολογιστής από την άλλη) να χρησιμοποιούν τον ίδιο ρυθμό επικοινωνίας (baud rate) π.χ. 9600 bps (bits per second).

Η αποστολή δεδομένων από τον Arduino στον υπολογιστή γίνεται συνήθως (αλλά όχι μόνο) μέσω των εντολών (συναρτήσεων):

- **Serial.print()**
- **Serial.println()**

Πρόκειται για δύο αρκετά ευέλικτες εντολές, καθώς μπορούμε με αυτές να τυπώσουμε δεδομένα στη σειριακή κονσόλα ως συνδυασμό αλφαριθμητικών χαρακτήρων. Εξ ορισμού οι δεκαδικοί αριθμοί τυπώνονται με δύο δεκαδικά ψηφία. Η διαφορά μεταξύ των δύο εντολών είναι πως με τη δεύτερη, αφού τυπωθούν τα δεδομένα στη σειριακή κονσόλα, ο δρομέας μετακινείται στην επόμενη γραμμή της σειριακής κονσόλας.



Εικόνα 20: Το παράθυρο της σειριακής κονσόλας του Arduino IDE

### Παραδείγματα:

- `Serial.print("Hello!!!");` // Τυπώνει *Hello!!!* στη σειριακή κονσόλα
- `Serial.print(value);` // Τυπώνει στη σειρ. κονσόλα την τιμή της μεταβλητής *value*
- `Serial.print(value,1);` // Τυπώνει στη σειρ. κονσόλα την τιμή της μεταβλητής *value* με ακρίβεια ενός δεκαδικού ψηφίου
- `Serial.print(78, HEX)` // Τυπώνει τον αριθμό 78 σε δεκαεξαδική μορφή ("*4E*")
- `Serial.print("\t")` // Στέλνει στη σειρ. κονσόλα το χαρακτήρα ελέγχου *tab*

Μπορούμε να διαβάσουμε ένα χαρακτήρα από τη σειριακή θύρα χρησιμοποιώντας την εντολή **Serial.read()**, ενώ με τη **Serial.available()** ελέγχουμε αν και πόσα σειριακά δεδομένα υπάρχουν έτοιμα προς λήψη.

### Παραδείγματα:

- `Serial.print(Serial.available());` // Τυπώνει στη σειρ. κονσόλα τον αριθμό των *bytes* που είναι διαθέσιμα για λήψη
- `int c = Serial.read();` // Διαβάζει ένα *byte* που έχει αποσταλεί στη σειριακή θύρα και το αποθηκεύει σε μια ακέραιου τύπου (*int*) μεταβλητή

Ας σημειωθεί στο σημείο αυτό πως, τους μέσω σειριακής θύρας εισερχόμενους χαρακτήρες, ο Arduino τους αποθηκεύει σε μια προσωρινή περιοχή μνήμης (serial buffer), με εξ' ορισμού χωρητικότητα 64 bytes.

**Σημείωση:** Οι συναρτήσεις που σχετίζονται με τη σειριακή επικοινωνία Arduino-υπολογιστή περιέχονται σε μια βιβλιοθήκη κώδικα (library) του Arduino με το όνομα *Serial*. Η συμπερίληψη αυτής της βιβλιοθήκης στα προγράμματά μας γίνεται αυτόματα, χωρίς δηλαδή να απαιτείται οποιαδήποτε πρόσθετη ενέργεια από εμάς. Μπορούμε να πάρουμε περισσότερες πληροφορίες για τις σχετικές συναρτήσεις στην ιστοσελίδα [Arduino Language Reference](https://www.arduino.cc/reference/en/) (<https://www.arduino.cc/reference/en/>). Γενικά οι κλήσεις από τα προγράμματά μας συναρτήσεων που περιέχονται σε βιβλιοθήκες γίνονται με το σχήμα: (όνομα βιβλιοθήκης).(όνομα συνάρτησης) και ακολουθούν οι όποιες παράμετροι δέχεται η συνάρτηση.

Στο παράδειγμα που ακολουθεί, τυπώνεται στη σειριακή κονσόλα (οθόνη) ένας τυχαίος αριθμός που υπολογίζεται με βάση την ενσωματωμένη στο λογισμικό του Arduino γεννήτρια ψευδοτυχαίων αριθμών:

```
/*  
   Ex.5 : Παράδειγμα χρήσης σειριακής κονσόλας  
*/  
  
void setup()  
{  
    Serial.begin(9600);           // Ενεργοποίηση σειριακής επικοινωνίας στα 9600 bps  
}  
  
void loop()  
{  
    int value = random(1,10);     // Παραγωγή ψευδοτυχαίου αριθμού μεταξύ 1 και 10  
    Serial.print("A random number (1-10): "); // Εκτύπωση μηνύματος στη σειριακή κονσόλα  
    Serial.println(value);       // Εκτύπωση του ψευδοτυχαίου αριθμού  
    delay (1000);                // Αναμονή 1 sec  
}
```

**Σημείωση:** Οι αριθμοί που παράγει η συνάρτηση *random()* χαρακτηρίζονται ως ψευδοτυχαίοι, αφού ο χρησιμοποιούμενος αλγόριθμος παράγει μια σειρά φαινομενικά τυχαίων αλλά στην πραγματικότητα πλήρως καθορισμένων αριθμών. Δηλαδή κάθε φορά που “τρέχει” το πρόγραμμα, παράγει την ίδια ακριβώς σειρά “τυχαίων” αριθμών. Χρησιμοποιώντας στη συνάρτηση *setup()* του sketch μας τη συνάρτηση *randomSeed(seed)* μπορούμε να παράγουμε μια διαφορετική σειρά ψευδοτυχαίων αριθμών. Αν όμως η τιμή της μεταβλητής *seed* είναι σταθερή, τότε και αυτή η σειρά αριθμών θα είναι πλήρως καθορισμένη. Η λύση για να παίρνουμε πραγματικά τυχαίους αριθμούς κάθε φορά που εκτελείται το λογισμικό είναι να χρησιμοποιούμε κάθε φορά τυχαία τιμή για τη μεταβλητή *seed*.

Όπως έχουμε ήδη αναφέρει ο μικροελεγκτής ATmega 328 διαθέτει στο υλικό του μία μόνο μονάδα USART και συνεπώς μπορεί να επικοινωνεί με μία μόνο σειριακή συσκευή. Για την περίπτωση που θέλουμε ο Arduino Uno να επικοινωνεί σειριακά με περισσότερες από μία συσκευές, μπορούμε να χρησιμοποιούμε την εξωτερική βιβλιοθήκη *SoftwareSerial*, η οποία αναπαράγει μέσω λογισμικού τη λειτουργία μιας σειριακής θύρας σε οποιοσδήποτε άλλες ψηφιακές ακίδες του Arduino.

## Δομές λήψης απόφασης

Όταν θέλουμε στο sketch μας να ληφθούν αποφάσεις ανάλογα με κάποιες συνθήκες (και συνεπώς να εκτελεστούν οι αντίστοιχες εντολές) μπορούμε να χρησιμοποιήσουμε τις δομές *if...else* ή *switch...case*.

### α. Η δομή *if...else*

Στην πιο απλή μορφή της δομής *if...else*, απουσιάζει εντελώς το τμήμα *else* της δομής, έχει δηλαδή τη μορφή:

```
if (συνθήκη)
{
    // εντολές που εκτελούνται αν η συνθήκη είναι αληθής
}
```

Τα δύο άγκιστρα {...} καθορίζουν το μπλοκ των εντολών που θα εκτελεστούν αν η συνθήκη που ακολουθεί την εντολή *if* είναι αληθής. Στο επόμενο παράδειγμα:

```
if (value == 5)
{
    Serial.println("value is equal to 5");
}
```

το σύμβολο "==" (διπλό =) που χρησιμοποιήσαμε είναι ένας *τελεστής σύγκρισης* (βλέπε και Πίνακα 1) και χρησιμοποιείται εδώ για τη σύγκριση της μεταβλητής *value* με την τιμή 5: Δηλαδή η συνθήκη (*value == 5*) είναι αληθής αν η μεταβλητή *value* έχει τιμή ίση με 5, και τότε εκτελείται η εντολή *Serial.println("value is equal to 5")*, ενώ είναι ψευδής σε οποιαδήποτε άλλη περίπτωση, και το πρόγραμμα συνεχίζει με την πρώτη εντολή αμέσως μετά το μπλοκ εντολών του *if*.

**Σημείωση:** Παρότι στην περίπτωση του παραδείγματός μας (όπου μόνο μία εντολή εκτελείται όταν η συνθήκη της εντολής *if* είναι αληθής) τα δύο άγκιστρα θα μπορούσαν να παραληφθούν, είναι καλή τακτική (τουλάχιστον στη φάση αυτή) να μην το κάνουμε.

Πίνακας 1: Τελεστές σύγκρισης

!=	Δεν είναι ίσο με
<	Μικρότερο από
<=	Μικρότερο από ή ίσο με
>	Μεγαλύτερο από
>=	Μεγαλύτερο από ή ίσο με
==	Ίσο με

Η πλήρης μορφή της δομής αναπτύσσεται ως εξής:

```
if (συνθήκη)
{
    // εντολές που εκτελούνται αν η συνθήκη είναι αληθής
}
else
{
    // εντολές που εκτελούνται αν η συνθήκη είναι ψευδής
}
```

Ένα απλό παράδειγμα αποτελεί ο συνδυασμός εντολών:

```

if (value == HIGH)
{
    Serial.println(" value is HIGH");    // Εκτελείται όταν η μεταβλητή value είναι HIGH
}
else
{
    Serial.println(" value is LOW");    // Εκτελείται όταν η μεταβλητή value ΔΕΝ είναι HIGH
}

```

Πολυπλοκότερες συνθήκες ελέγχου μπορούν να υλοποιηθούν με χρήση των λογικών τελεστών (βλέπε και Πίνακα 2). Για παράδειγμα με την εντολή:

```
if ((val1 == HIGH) && (val2 == LOW))
```

γίνεται έλεγχος αν η μεταβλητή *val1* έχει τιμή 1 (HIGH) και ταυτόχρονα η μεταβλητή *val2* έχει τιμή 0 (LOW).

Πίνακας 2: Λογικοί τελεστές

!	Λογική άρνηση (NOT)
&&	Λογική σύζευξη (AND)
	Λογική διάζευξη (OR)

Επιπλέον **if** μπορούν να συνδυαστούν με την **else**, δίνοντας τη δυνατότητα λήψης πολυπλοκότερων αποφάσεων, όπως στο ακόλουθο παράδειγμα:

```

if (value < 5)                // Αν η μεταβλητή value έχει τιμή μικρότερη από 5
{
    Serial.println("value is less than 5");
}
else if (value > 5)          // Αλλιώς, αν η μεταβλητή value έχει τιμή μεγαλύτερη από 5
{
    Serial.println("value is greater than 5");
}
else                          // Σε άλλη περίπτωση, δηλ. αν value = 5
{
    Serial.println("value is equal to 5");
}

```

Επανερχόμενοι στο παράδειγμα όπου ένας διακόπτης συνδέεται σε μια αναλογική είσοδο του Arduino (Εικόνες 17 & 18), ας δεχτούμε πως θέλουμε, όταν πατάμε το διακόπτη, να τυπώνει στη σειριακή κονσόλα το μήνυμα "HIGH", ενώ όταν ο διακόπτης έχει απελευθερωθεί, να τυπώνει το μήνυμα "LOW". Το σχετικό sketch μπορεί να πάρει τη μορφή:

```

/*
    Ex.6 : Ανάγνωση ψηφιακής ακίδας του Arduino και λήψη απόφασης
*/

#define BUTTONPIN 7

void setup ()

```

```

{
    Serial.begin(9600);           // Εκκίνηση σειρ. επικοινωνίας στα 9600 bps
    pinMode(BUTTONPIN, INPUT);  // Η ακίδα 7 καθορίζεται ως ψηφιακή είσοδος
}

void loop ()
{
    int value = digitalRead(BUTTONPIN); // Διάβασε την κατάσταση του διακόπτη
    if (value == HIGH)                // Ελέγχει αν ο διακόπτης είναι κλειστός
    {
        Serial.println("Switch is closed");
    }
    else                                // Αν ο διακόπτης δεν είναι κλειστός
    {
        Serial.println("Switch is opened");
    }

    delay(1000);                    // Αναμονή 1sec
}

```

### β. Η δομή *switch...case*

Η δομή *if...else* επιτρέπει τη λήψη απόφασης με βάση δύο διακριτές επιλογές (δηλ. ανάλογα αν μια συνθήκη είναι *true* ή *false*). Όταν υπάρχουν περισσότερες από δύο επιλογές, μπορούμε να χρησιμοποιήσουμε πολλαπλές συνδυασμένες εντολές *if*, όμως τόσο η αναγνωσιμότητα του κώδικα όσο και η ταχύτητα εκτέλεσης βελτιώνονται αν χρησιμοποιήσουμε τη δομή *switch...case*, στην οποία οι διάφορες επιλογές καθορίζονται με βάση τις τιμές μιας μεταβλητής. Αυτό αποτελεί και το βασικό μειονέκτημα της δομής καθώς δε μπορούν να χρησιμοποιηθούν πολυπλοκότερες συνθήκες, όπως στη δομή *if...else*.

Η σύνταξη της δομής *switch...case* έχει τη μορφή:

```

switch (var)
{
    case label1:
        // Εντολές για την 1η επιλογή
        break;
    case label2:
        // Εντολές για τη 2η επιλογή
        break;
    .....
    default:
        // Αν καμία από τις άλλες επιλογές δεν ταιριάζει
        break;
}

```

Η μεταβλητή *var* μπορεί να είναι ένας ακέραιος αριθμός ή ένας χαρακτήρας, ενώ οι *label1*, *label2*, κ.λ.π., πρέπει να είναι σταθερές του ίδιου τύπου. Με την εντολή *break* πραγματοποιείται έξοδος από τη δομή *switch...case*.

Στο επόμενο παράδειγμα το -ενσωματωμένο στον Arduino- LED που συνδέεται στην ψηφιακή ακίδα 13 ανάβει, σβήνει ή αναβοσβήνει ανάλογα με την εντολή που δέχεται ο Arduino μέσω της σειριακής κονσόλας.

```
/*
   Ex. 6.1 : Ανάγνωση ψηφιακής ακίδας του Arduino και λήψη απόφασης
*/

#define LEDPIN      13
char inChar = 'a';

void setup ()
{
    Serial.begin(9600);           // Εκκίνηση σειρ. επικοινωνίας στα 9600 bps
    pinMode(LEDPIN, OUTPUT);     // Η ακίδα 13 καθορίζεται ως ψηφιακή έξοδος

    // Με τις επόμενες εντολές τυπώνεται στη σειριακή κονσόλα ένα απλό μενού
    Serial.println("Send :");
    Serial.println("a - To turn the LED on");
    Serial.println("b - To turn the LED off");
    Serial.println("c - To turn the LED on and off continuously");
}


void loop ()
{
    // Αν υπάρχουν διαθέσιμοι για διάβασμα χαρακτήρες στη σειριακή προσωρινή μνήμη
    if (Serial.available() != 0)
    {
        inChar = Serial.read();   // Διάβασε τον πρώτο διαθέσιμο χαρακτήρα
        while (Serial.available()) // Καθάρισε την προσωρινή μνήμη
        {
            char tmp = Serial.read();
        }
    }

    // Λήψη απόφασης ανάλογα με τον ληφθέντα χαρακτήρα
    switch(inChar)
    {
        case 'a' :
            digitalWrite(LEDPIN, HIGH);
            break;
        case 'b' :
            digitalWrite(LEDPIN, LOW);
            break;
        case 'c' :
            digitalWrite(LEDPIN, HIGH);
    }
}
```

```
        delay(200);
        digitalWrite(LEDPIN, LOW);
        delay(200);
        break;
    default :
        inChar = 'b';
}
}
```

## Βρόχοι επανάληψης

Ας υποθέσουμε πως έχοντας συνδέσει ένα διακόπτη σε μια ψηφιακή είσοδο του Arduino θέλουμε κάθε φορά που κλείνουμε το διακόπτη να αναβοσβήνει πέντε φορές, και με συγκεκριμένο ρυθμό, ένα LED που είναι συνδεδεμένο σε μια άλλη ψηφιακή ακίδα (έξοδο) του Arduino. Οι συνδέσεις του Arduino με το διακόπτη και το LED φαίνονται στο επόμενο σχήμα:



Εικόνα 21: LED και διακόπτης push button στον Arduino

Έχουμε ήδη δει σε προηγούμενα παραδείγματα:

- Πώς να ορίσουμε μια ακίδα του Arduino ως ψηφιακή είσοδο ή έξοδο:

```
pinMode (7, INPUT);  
pinMode (8, OUTPUT);
```

- Πώς να “διαβάζουμε” την κατάσταση ενός διακόπτη:

```
int value = digitalRead (7);
```

- Πώς να παίρνουμε αποφάσεις με βάση μια συνθήκη:

```
if (value == HIGH)  
{  
    // Εντολές που εκτελούνται αν η συνθήκη είναι αληθής  
}
```

- Πώς να αναβοσβήσουμε μια φορά το LED:

```
digitalWrite (8, HIGH);    // Αναψε το LED  
delay (200);              // Αναμονή 200 milliseconds  
digitalWrite (8, LOW);   // Σβήσε το LED  
delay (200);             // Αναμονή 200 milliseconds
```

Είναι φανερό πως, αφού θέλουμε το LED να αναβοσβήσει πέντε φορές, οι τέσσερις γραμμές κώδικα με τις οποίες το λαμπάκι αναβοσβήνει πρέπει να εκτελεστούν συνολικά πέντε φορές. Μια απλοϊκή προσέγγιση είναι να επαναλάβουμε αυτές τις τέσσερις γραμμές συνολικά πέντε φορές στο

σώμα της συνάρτησης *loop()* του sketch μας. Όμως η γλώσσα προγραμματισμού του Arduino διαθέτει εξειδικευμένες εντολές για τη δημιουργία βρόχων επανάληψης, δηλ. όταν θέλουμε ένα μέρος του κώδικα να επαναληφθεί κάποιες φορές.

### **a. Ο βρόχος *for* (*for loop*)**

Όταν είναι γνωστός εκ των προτέρων ο αριθμός των επαναλήψεων του κώδικα, μπορούμε να χρησιμοποιήσουμε την εντολή *for*, η σύνταξη της οποίας έχει την ακόλουθη μορφή:

```
for (εντολή αρχικοποίησης; συνθήκη; εντολή επανάληψης)
{
    // Εντολές που πρέπει να εκτελεστούν
}
```

Ένα απλό παράδειγμα θα διευκρινίσει τη χρήση της εντολής:

```
for (int i = 0; i < 5; i++)
{
    Serial.println(i);
}
```

Κατά την εκτέλεση του βρόχου:

1. Με την εντολή αρχικοποίησης (*int i = 0*) ορίζεται μια τοπική ακέραια μεταβλητή απαριθμησης με το όνομα “*i*” της δίνεται η αρχική τιμή 0. Η εντολή αρχικοποίησης εκτελείται μόνο μια φορά στην αρχή του βρόχου.
2. Μετά συγκρίνεται η τιμή της μεταβλητής αυτής με την τιμή 5, και αν ικανοποιείται η συνθήκη “*i < 5*” τυπώνεται σε μια γραμμή της σειριακής κονσόλας η τιμή της μεταβλητής “*i*” (εντολή *Serial.println...*). Όταν όμως η μεταβλητή “*i*” γίνει μεγαλύτερη ή ίση με 5, η συνθήκη “*i < 5*” γίνεται ψευδής, οπότε το sketch εξέρχεται από το βρόχο επανάληψης και συνεχίζει την εκτέλεση με την πρώτη εντολή αμέσως μετά το μπλοκ εντολών του βρόχου.
3. Εν συνεχεία η τιμή της μεταβλητής “*i*” αυξάνεται κατά ένα (αυτό είναι το νόημα της εντολής επανάληψης *i++*).
4. Ο βρόχος επαναλαμβάνεται από το βήμα (2) και κάτω.

Το αποτέλεσμα είναι η επανάληψη όλων των εντολών του βρόχου για συνολικά πέντε φορές (δηλ. από *i = 0* μέχρι και *i = 4*). Αν τώρα στο προηγούμενο παράδειγμα την εντολή *Serial.println(i)* την αντικαταστήσουμε με τις τέσσερις γραμμές κώδικα που κάνουν το λαμπάκι να αναβοσβήνει, έχουμε πετύχει το στόχο μας.

Το σχετικό sketch διαμορφώνεται ως εξής:

```
/*
   Ex.7 : Παράδειγμα βρόχου επανάληψης με την εντολή for
*/

#define BUTTONPIN      8
#define LEDPIN         10

void setup ()
{
    pinMode(LEDPIN,    OUTPUT);           // Η ακίδα 10 καθορίζεται ως ψηφιακή έξοδος
    pinMode(BUTTONPIN, INPUT);           // Η ακίδα 8 καθορίζεται ως ψηφιακή είσοδος
}
```

```

void loop ()
{
    int value = digitalRead(BUTTONPIN);           // Διάβασε την κατάσταση του διακόπτη

    if (value == HIGH)                            // Ελέγχει αν ο διακόπτης είναι κλειστός
    {
        for (int i = 0; i < 5; i++)              // Βρόχος επανάληψης
        {
            digitalWrite(LEDPIN, HIGH);         // Αναψε το LED
            delay(200);                          // Αναμονή 200 milliseconds
            digitalWrite(LEDPIN, LOW);          // Σβήσε το LED
            delay(200);                          // Αναμονή 200 milliseconds
        }
    }
}

```

Η σύνταξη του βρόχου *for* είναι αρκετά ευέλικτη. Στο επόμενο παράδειγμα:

```

    for (int i = 0; i < 5; i+=2)
    {
        Serial.println(i);
    }

```

η εντολή *Serial.println(i)* επαναλαμβάνεται από  $i = 0$  μέχρι και  $i = 4$  με βήμα αύξησης 2, δηλ. στη σειριακή κονσόλα τυπώνονται οι αριθμοί 0, 2 και 4. Και στο παράδειγμα:

```

    for (int i = 10; i >= 0; i-=2)
    {
        Serial.println(i);
    }

```

η εντολή *Serial.println(i)* επαναλαμβάνεται από  $i = 10$  μέχρι και  $i = 0$  με βήμα μείωσης 2, δηλ. στη σειριακή κονσόλα τυπώνονται κατά σειρά οι αριθμοί 10, 8, 6, 4, 2 και 0

### **β. Ο βρόχος while (while loop)**

Με χρήση του βρόχου *while* επιτυγχάνουμε την εκτέλεση ενός συγκεκριμένου μπλοκ εντολών ενόσω μια λογική συνθήκη είναι αληθής. Για παράδειγμα το προηγούμενο sketch μπορεί να διαμορφωθεί ως εξής:

```
/*
```

*Ex.8 : Παράδειγμα βρόχου επανάληψης με την εντολή while*

```
*/
```

```

#define BUTTONPIN 8
#define LEDPIN    10

```

```

void setup ()
{
    pinMode(LEDPIN,OUTPUT);           // Η ακίδα 10 καθορίζεται ως ψηφιακή έξοδος
    pinMode(BUTTONPIN, INPUT);       // Η ακίδα 8 καθορίζεται ως ψηφιακή είσοδος
}

```

```

void loop ()
{

```

```

int cnt = 0; // Ορισμός μεταβλητής απαριθμητή
int value = digitalRead(BUTTONPIN); // Διάβασε την κατάσταση του διακόπτη

if (value == HIGH) // Ελέγχει αν ο διακόπτης είναι κλειστός
{
    while (cnt < 5) // Βρόχος επανάληψης while
    {
        // Το μπλοκ των εντολών εκτελείται ενόσω
        // ο απαριθμητής έχει τιμή μικρότερη από 5
        digitalWrite(LEDPIN, HIGH); // Αναψε το LED
        delay(200); // Αναμονή 200 milliseconds
        digitalWrite(LEDPIN, LOW); // Σβήσε το LED
        delay(200); // Αναμονή 200 milliseconds
        cnt++; // Αύξηση κατά 1 του απαριθμητή
    }
}
}

```

Και σ' αυτή την περίπτωση ορίσαμε μια ακέραια μεταβλητή απαρίθμησης με το όνομα *cnt*, της δώσαμε την αρχική τιμή μηδέν, ενώ μέσα στο σώμα εντολών του βρόχου *while*, η τιμή της αυξάνεται κατά 1 (εντολή *cnt++*). Η μεταβλητή αυτή είναι τοπική στη συνάρτηση *loop()*, δηλ. έχει ισχύ μόνο εντός της συνάρτησης, τόσο εντός όσο και εκτός του βρόχου *while {...}*.

Η χρήση του βρόχου *while* ενδείκνυται όταν δεν είναι εκ των προτέρων γνωστός ο αριθμός των επαναλήψεων. Για παράδειγμα με την εντολή:

```

while (! Serial.available());

```

που έχει το ίδιο αποτέλεσμα με την:

```

while (Serial.available() == 0) ;

```

δημιουργείται ένας βρόχος αναμονής, που συνεχίζει δηλ. να εκτελείται ενόσω δεν υπάρχουν δεδομένα για λήψη μέσω της σειριακής θύρας.

Ιδιαίτερου ενδιαφέροντος είναι και η δημιουργία ατέρμονα βρόχου (infinite or endless loop) με χρήση των εντολών [5]:

```

while (true) ; ή while (1) ;

```

Μια παραλλαγή του βρόχου *while* έχει τη μορφή:

```

do
{
    // Εντολές του βρόχου
}
while (συνθήκη)

```

Η διαφορά μεταξύ των δύο μορφών του βρόχου *while* είναι πως με την πρώτη μορφή υπάρχει η περίπτωση (όταν εξ' αρχής η συνθήκη είναι αληθής) οι εντολές του βρόχου να μην εκτελεστούν ούτε μία φορά, ενώ με τη δεύτερη μορφή θα εκτελεστούν οπωσδήποτε μία φορά.

## Στοιχεία δομημένου προγραμματισμού

Η λειτουργικότητα και αναγνωσιμότητα του κώδικα μπορεί να βελτιωθεί με χρήση τεχνικών δομημένου προγραμματισμού: το πρόβλημα χωρίζεται σε ανεξάρτητα υποπροβλήματα, τα οποία προγραμματιστικά επιλύονται με ανεξάρτητα τμήματα κώδικα, και εν συνεχεία ακολουθεί η σύνθεση των επιμέρους λύσεων. Η επίλυση των επιμέρους υποπροβλημάτων υλοποιείται από συναρτήσεις οριζόμενες από το χρήστη, στις οποίες το πρόγραμμα μέσω κατάλληλων μεταβλητών (ορίσματα) μπορεί να μεταφέρει κάποιες τιμές απαραίτητες για τη λειτουργία τους. Επιπλέον οι συναρτήσεις μπορεί να επιστρέφουν κάποια τιμή στο πρόγραμμα από το οποίο καλούνται.

Γενικά ο ορισμός μιας συνάρτησης έχει τη μορφή:

```
Επιστρεφόμενος_τύπος Όνομα_συνάρτησης ( δηλώσεις ορισμάτων )
{
    // Σώμα της συνάρτησης
    .....
    return τιμή; // Μη απαραίτητη
}
```

Επανερχόμενοι στο προηγούμενο πρόβλημα (δηλ. το αναβόσβημα 5 φορές ενός LED με το πάτημα ενός διακόπτη, βλέπε και Εικόνα 21), μια τέτοια νέα συνάρτηση θα μπορούσε να αναλάβει την υλοποίηση του αναβοσβήσιματος του LED, ως εξής:

```
void blink() // Ορισμός συνάρτησης blink()
{
    digitalWrite (LEDPIN, HIGH); // Αναψε το LED
    delay (200); // Αναμονή 200 milliseconds
    digitalWrite (LEDPIN, LOW); // Σβήσε το LED
    delay (200); // Αναμονή 200 milliseconds
}
```

Στη νέα συνάρτηση δόθηκε το περιγραφικό όνομα *blink*, και εφόσον δεν επιστρέφει κάποια τιμή στον Arduino ο τύπος ορίστηκε της ως *void* (= χωρίς τύπο).

Το αντίστοιχο sketch διαμορφώνεται ως:

```
/*
   Ex.9 : Παράδειγμα δομημένου προγραμματισμού
*/

#define BUTTONPIN      8
#define LEDPIN        10

void setup ()
{
    pinMode(LEDPIN, OUTPUT); // Η ακίδα 10 καθορίζεται ως ψηφιακή έξοδος
    pinMode(BUTTONPIN, INPUT); // Η ακίδα 8 καθορίζεται ως ψηφιακή είσοδος
}
void loop ()
{
    int value = digitalRead(BUTTONPIN); // Διάβασε την κατάσταση του διακόπτη
    if (value == HIGH) // Ελέγχει αν ο διακόπτης είναι κλειστός
    {
```

```

        for (int i = 0; i <5; i++)           // Επανάληψη 5 φορές
        {
            blink();                         // Αναβοσβήνει το Led
        }
    }

void blink()                               // Συνάρτηση για αναβόσβημα του led
{
    digitalWrite(LEDPIN, HIGH);           // Αναψε το LED
    delay(200);                           // Αναμονή 200 milliseconds
    digitalWrite(LEDPIN, LOW);           // Σβήσε το LED
    delay(200);                           // Αναμονή 200 milliseconds
}

```

Μια ακόμη πιο ευέλικτη εκδοχή της συνάρτησης *blink* και του αντίστοιχου sketch, δίνεται στο επόμενο παράδειγμα:

```

/*
   Ex.10 : Δεύτερο παράδειγμα δομημένου προγραμματισμού
*/

#define BUTTONPIN 8
#define LEDPIN    10

void setup ()
{
    pinMode(LEDPIN, OUTPUT);           // Η ακίδα 10 καθορίζεται ως ψηφιακή έξοδος
    pinMode(BUTTONPIN, INPUT);        // Η ακίδα 8 καθορίζεται ως ψηφιακή είσοδος
}

void loop ()
{
    int value = digitalRead(BUTTONPIN); // Διάβασε την κατάσταση του διακόπτη
    if (value == HIGH)                 // Ελέγχει αν ο διακόπτης είναι κλειστός
    {
        blink(LEDPIN, 200, 5);        // Αναβοσβήνει το Led
    }
}

void blink(int ledpin, int dtime, int count) // Συνάρτηση για αναβόσβημα του led
{
    for (int i = 0; i < count; i++)
    {
        digitalWrite(ledpin, HIGH);   // Αναψε το LED
        delay(dtime);                 // Αναμονή
        digitalWrite(ledpin, LOW);    // Σβήσε το LED
        delay(dtime);                 // Αναμονή
    }
}

```

Στη νέα εκδοχή της συνάρτησης *blink* ο βρόχος για συγκεκριμένο αριθμό επαναλήψεων έχει μεταφερθεί εντός της συνάρτησης, η οποία δέχεται τρεις ακέραιου τύπου (*int*) παραμέτρους, ως εξής:

- *ledpin* : Είναι η ακίδα του Arduino στην οποία έχει συνδεθεί το LED
- *dtime* : Το χρονικό διάστημα σε msec που το LED παραμένει αναμμένο ή σβηστό
- *count* : Ο αριθμός επαναλήψεων του αναβοσβησίματος

Μια συνάρτηση οριζόμενη από το χρήστη μπορεί επιπλέον να επιστρέφει κάποια τιμή στο πρόγραμμα από το οποίο καλείται. Ο επιστρεφόμενος τύπος μιας τέτοιας συνάρτησης κατά τον ορισμό της πρέπει να είναι του ίδιου τύπου με τη μεταβλητή που επιστρέφει. Για παράδειγμα η επόμενη συνάρτηση δέχεται ως παράμετρο μια ακέραια μεταβλητή και επιστρέφει το (ακέραιου τύπου) τετράγωνό της:

```
int square(int x)
{
    return x*x;
}
```

**Παρατήρηση :** Σε ένα απλό πρόγραμμα C/C++ οι οριζόμενες από το χρήστη συναρτήσεις πρέπει να δηλώνονται στην αρχή του προγράμματος, ώστε στη συνέχεια να μπορούν να καλούνται από το κύριο σώμα του προγράμματος. Κάτι τέτοιο δεν είναι απαραίτητο στη γλώσσα προγραμματισμού του Arduino.

Επιπλέον η γλώσσα προγραμματισμού του Arduino υποστηρίζει τεχνικές:

- **Αρθρωτού κώδικα**, όπου όλες οι συναρτήσεις σχετικά με την επίλυση του ίδιου υποπρόβληματος αποθηκεύονται σε ξεχωριστά αρχεία κώδικα, τα οποία μαζί με το αντίστοιχο αρχείο επικεφαλίδων (header file), που περιέχει τις δηλώσεις των συναρτήσεων και των μεταβλητών, χαρακτηρίζονται ως βιβλιοθήκες (library).
- **Αντικειμενοστραφούς προγραμματισμού**: Βασικό δομικό στοιχείο στον αντικειμενοστραφή προγραμματισμό είναι η κλάση: μπορούμε να τη θεωρούμε ως ένα (προγραμματιστικό) χώρο όπου ένα σύνολο συναρτήσεων και μεταβλητών (τα μέλη της κλάσης) διατηρούνται μαζί. Τα μέλη της κλάσης μπορεί να είναι δημόσια (public) δηλαδή να είναι προσπελάσιμα από όλους όσους χρησιμοποιούν την κλάση, ή ιδιωτικά (private) δηλαδή είναι προσπελάσιμα μόνο από άλλα μέλη της ίδιας κλάσης. Κάθε κλάση διαθέτει μια ειδική συνάρτηση (constructor) που χρησιμεύει για τη δημιουργία ενός αντικειμένου που ανήκει στην κλάση αυτή.

## Ανάγνωση αναλογικής εισόδου


Στον μικροελεγκτή ATmega 328 του Arduino Uno έξι διαφορετικές αναλογικές εισοδοί, μέσω της μεθόδου της πολυπλεξίας, οδηγούνται στην είσοδο ενός μετατροπέα αναλογικού σε ψηφιακό (A/D C) διακριτικής ικανότητας 10bit [3]. Αυτό σημαίνει ότι με τις εξ ορισμού ρυθμίσεις ο μετατροπέας αντιστοιχίζει αναλογικές τάσεις μεταξύ 0 και 5V σε ακέραιες τιμές μεταξύ 0 και 1023, το οποίο οδηγεί σε διακριτική ικανότητα: 5V/1024 βήματα, ή περίπου 4.9 mV/βήμα.

Η περιοχή της τάσης που μπορούν να διαχειριστούν οι αναλογικές εισοδοί, και συνεπώς και η αντίστοιχη διακριτική ικανότητα, μπορούν, εφόσον είναι επιθυμητό, να ρυθμιστούν μέσω της εντολής *analogReference()*. Απαιτείται η σύνδεση της επιθυμητής τάσης αναφοράς ( $V_{REF}$ ) στην είσοδο *AREF* του Arduino, η οποία όμως σε καμία περίπτωση δε μπορεί να υπερβεί τα 5V.

Με τις εξ ορισμού ρυθμίσεις του λογισμικού του ο Arduino χρειάζεται λίγο περισσότερο από 100 μs για να ολοκληρώσει μια μετατροπή αναλογικού σε ψηφιακό, γεγονός που περιορίζει σε κάτι λιγότερο από 10.000 το (θεωρητικό) πλήθος των μετρήσεων αναλογικών μεγεθών που μπορούν να ληφθούν κάθε δευτερόλεπτο.

Το αποτέλεσμα μιας αναλογικο-ψηφιακής μετατροπής σε κάποια αναλογική είσοδο του Arduino μπορούμε να “διαβάσουμε” με την εντολή *analogRead()* [4], η οποία δέχεται ως παράμετρο την αναλογική είσοδο (A0 μέχρι A5) στην οποία η προς μέτρηση τάση έχει συνδεθεί.

Για τις ανάγκες της επόμενης άσκησης θα χρησιμοποιήσουμε μια φωτοαντίσταση (LDR Light Dependent Resistor) σε συνδεσμολογία διαιρέτη τάσης με μια αντίσταση σταθερής τιμής 10kΩ. Το σύστημα των δύο αντιστάσεων τροφοδοτείται από σταθερή τάση 5V, ενώ η αναλογική είσοδος A0 του Arduino τροφοδοτείται με την τάση που αναπτύσσεται στα άκρα της σταθερής αντίστασης.



Εικόνα 22: Διαιρέτης τάσης με φωτοαντίσταση


Αν συμβολίσουμε με  $R_x$  την αντίσταση της φωτοαντίστασης, εύκολα μπορούμε να αποδείξουμε πως ο λόγος της τάσης στα άκρα της φωτοαντίστασης προς την τάση στα άκρα της σταθερής αντίστασης  $R = 10 \text{ k}\Omega$  είναι:

$$\frac{V_x}{V_R} = \frac{R_x}{R}$$

Δηλαδή, χαρακτηριστικό του διαιρέτη τάσης είναι ότι διαιρεί την τάση τροφοδοσίας του σε επιμέρους τάσεις με τιμές ανάλογες των αντίστοιχων αντιστάσεων.

Χαρακτηριστικό της φωτοαντίστασης είναι πως η αντίστασή της μεταβάλλεται σε σχέση με την ένταση του φωτός που πέφτει πάνω της: Έχει πολύ μεγάλη αντίσταση (της τάξης των MΩ) στο σκοτάδι, η οποία μειώνεται σημαντικά καθώς αυξάνεται η ένταση του φωτός [6]. Συνεπώς όταν μεταβάλλεται η ένταση του φωτός που πέφτει στη φωτοαντίσταση, μεταβάλλεται αντίστοιχα και η τιμή της τάσης στα άκρα της σταθερής αντίστασης των 10 kΩ. Μετρώντας λοιπόν την τάση αυτή παίρνουμε μια ένδειξη της έντασης του φωτός που πέφτει στη φωτοαντίσταση.

Η σύνδεση του διαιρέτη τάσης στον Arduino (από τον οποίο και τροφοδοτείται) φαίνεται στην επόμενη εικόνα:



Εικόνα 23: Σύνδεση φωτοαντίστασης στον Arduino

Το σχετικό sketch διαμορφώνεται ως εξής:

```

/*
   Ex.11 : Μετατροπή αναλογικού σε ψηφιακό
*/

#define ANALOGPIN    A0

void setup ()
{
    Serial.begin(9600);           // Εκκίνηση σειρ. επικοινωνίας στα 9600 bps
    //pinMode(ANALOGPIN, INPUT); // Καλή τακτική, όχι όμως απαραίτητη
}

void loop ()
{
    int value = analogRead(ANALOGPIN); // Μετατροπή αναλογικού σε ψηφιακό, και
                                        // αποθήκευση αποτελέσματος στην
                                        // ακέραια μεταβλητή value

    float voltage = getVoltage(value); // Μετατροπή σε τάση
    Serial.print("Input Voltage is: "); // Αποστολή του αποτελέσματος στη
    Serial.print(voltage);              // σειριακή κονσόλα
    Serial.println(" V");

    delay (1000);                      // Αναμονή 1sec
}

float getVoltage(int value)
{
    return value*5.0 / 1024.0;
}

```

Η οριζόμενη από το χρήστη συνάρτηση `getVoltage()` δέχεται ως παράμετρο την ακέραια τιμή μιας

αναλογικοψηφιακής μετατροπής (*ADC value*) και επιστρέφει την αναλογική τάση στην οποία η τιμή αυτή αντιστοιχεί. Η μετατροπή γίνεται με βάση το γεγονός πως ο αναλογικοψηφιακός μετατροπέας του Arduino έχει διακριτική ικανότητα (5/1024) V/βήμα. Δηλαδή είναι:

$$V = (ADC\ value) \times \left( \frac{5}{1024} \right) \text{ ή γενικότερα } V = (ADC\ value) \times \left( \frac{V_{REF}}{1024} \right)$$

όπου  $V_{REF}$  είναι η τάση αναφοράς του αναλογικοψηφιακού μετατροπέα.

**Προσοχή** πρέπει να δοθεί στο γεγονός ότι το πηλίκο 5/1024 στη γλώσσα προγραμματισμού του Arduino επιστρέφει ακέραια τιμή (διαίρεση ακεραίων). Για να πάρουμε αποτέλεσμα τύπου κινητής υποδιαστολής (*float*) το πηλίκο πρέπει να γραφεί ως 5.0/1024.0.

**Παρατήρηση:** Αν και το σωστό είναι η μετατροπή μιας αναλογικοψηφιακής ένδειξης του Arduino σε τάση να γίνεται με τη σχέση:

$$V = (ADC\ value) \times \left( \frac{5}{1024} \right)$$


θα δείτε πολύ συχνά να χρησιμοποιείται η σχέση:

$$V = (ADC\ value) \times \left( \frac{5}{1023} \right)$$

Έχει επικρατήσει η δεύτερη σχέση επειδή για τιμή  $ADC = 0$  επιστρέφει τάση μηδέν και για τιμή  $ADC = 1023$  επιστρέφει τάση 5 V, ενώ η πρώτη σχέση δεν επιστρέφει ποτέ 5 V αφού η τιμή  $ADC$  έχει μέγιστο το 1023.

## Χρήση προεγκατεστημένης βιβλιοθήκης

Όπως έχουμε ήδη αναφέρει μαζί με το ολοκληρωμένο περιβάλλον εργασίας του Arduino εγκαθίστανται και κάποιες βιβλιοθήκες, οι οποίες δε συμπεριλαμβάνονται εξ' ορισμού στα sketches που δημιουργούμε. Τέτοια είναι και η βιβλιοθήκη **Servo**, που μας επιτρέπει να εκμεταλλευτούμε τις δυνατότητες των σερβοκινητήρων.




Εικόνα 24: Σερβοκινητήρας

Ένας σερβοκινητήρας αποτελείται από ένα μικρό ηλεκτρικό κινητήρα, ένα συνδυασμό γραναζιών ώστε να μπορεί να περιστρέφεται με μικρή ταχύτητα αλλά με μεγάλη ροπή, και ένα σύστημα για τον ακριβή έλεγχο της θέσης του άξονά του. Συνήθως το εύρος περιστροφής του άξονα περιορίζεται μεταξύ 0 και 180 μοιρών (μισή περιστροφή).

Η βιβλιοθήκη *Servo* χρησιμοποιώντας τεχνικές αντικειμενοστραφούς προγραμματισμού μας επιτρέπει τη δημιουργία αντικειμένων κλάσης *Servo* για το χειρισμό μέχρι και δώδεκα (12) σερβοκινητήρων συνδεδεμένων στις ακίδες του Arduino Uno. Οι βασικές συναρτήσεις της βιβλιοθήκης, είναι:

- *attach(pin)* : Δημιουργεί μια σύνδεση μεταξύ ενός σερβοκινητήρα και μιας I/O ακίδας του Arduino
- *write(pos)* : Περιστρέφει τον άξονα του σερβοκινητήρα σε θέση που καθορίζεται σε μοίρες από την παράμετρο *pos*.
- *read()* : Επιστρέφει την τρέχουσα γωνία του άξονα του σερβοκινητήρα (δηλ. την τιμή της παραμέτρου της τελευταίας κλήσης της συνάρτησης *write()*).

Ένας σερβοκινητήρας συνδέεται στον Arduino, όπως φαίνεται στην επόμενη εικόνα:



Εικόνα 25 : Σύνδεση σερβοκινητήρα στον Arduino

Για να αξιοποιήσουμε τις δυνατότητές του σερβοκινητήρα πρέπει να συμπεριλάβουμε στο sketch μας τη σχετική βιβλιοθήκη. Αυτό μπορεί να γίνει μέσω του μενού “Σχέδιο/Συμπερίληψη βι-

βλιοθήκης”, και επιλογή της βιβλιοθήκης “*Servo*”. Το λογισμικό ανταποκρίνεται εισάγοντας στην αρχή του sketch τη σχετική εντολή συμπερίληψης:

```
#include <Servo.h>
```

Εννοείται πως μπορούμε να γράψουμε και χειροκίνητα τη σχετική εντολή συμπερίληψης στην αρχή του sketch. Το επόμενο βήμα είναι να δημιουργήσουμε ένα αντικείμενο κλάσης *Servo*, για να αποκτήσουμε πρόσβαση στις διαθέσιμες εντολές διαχείρισης του σερβοκινητήρα.

Ας σημειώσουμε στο σημείο αυτό πως οι βιβλιοθήκες συνήθως συνοδεύονται από έτοιμα παραδείγματα, με τα οποία διασαφηνίζεται ο τρόπος χρήσης της βιβλιοθήκης. Στο επόμενο sketch, που αποτελεί παραλλαγή ενός από τα έτοιμα παραδείγματα της βιβλιοθήκης *Servo*, κάνοντας χρήση των επιλογών και δυνατοτήτων που η βιβλιοθήκη προσφέρει, περιστρέφουμε τον άξονα του κινητήρα παλινδρομικά μεταξύ 60° και 120° με βήμα περιστροφής 1 μοίρα.

```
/*
```

```
Ex.12 : Έλεγχος σερβοκινητήρα
```

```
*/
```

```
#include <Servo.h>
```

```
Servo myservo; // Δημιουργία αντικειμένου κλάσης Servo
```

```
int pos = 0; // Μεταβλητή για την αποθήκευση της θέσης του σερβοκινητήρα
```

```
void setup()
```

```
{
```

```
myservo.attach(9); // Ο σερβοκινητήρας συνδέεται στην ακίδα 9 του Arduino
```

```
myservo.write(60); // Η αρχική θέση του σερβοκινητήρα καθορίζεται στις 60°
```

```
}
```

```
void loop()
```

```
{
```

```
// Περιστροφή από τις 60 μέχρι τις 120 μοίρες με βήμα 1 μοίρα
```

```
for (pos = 60; pos <= 120; pos += 1)
```

```
{
```

```
myservo.write(pos); // Ενημερώνει τον κινητήρα να μεταβεί σε θέση "pos"
```

```
delay(10); // Αναμονή για να ολοκληρωθεί η περιστροφή
```

```
}
```

```
// Περιστροφή από τις 120 μέχρι τις 60 μοίρες με βήμα 1 μοίρα
```

```
for (pos = 120; pos >= 60; pos -= 1)
```

```
{
```

```
myservo.write(pos); // Ενημερώνει τον κινητήρα να μεταβεί σε θέση "pos"
```

```
delay(10); // Αναμονή για να ολοκληρωθεί η περιστροφή
```

```
}
```

```
}
```

Ας σημειώσουμε μια ενδιαφέρουσα παραλλαγή του βρόχου επανάληψης *for* στο συγκεκριμένο sketch:

```
for (pos = 120; pos >= 60; pos -= 1)
```

```
{
```

```
// Εντολές που πρέπει να εκτελεστούν
```

```
}
```

Ο βρόχος εκτελείται με αρχική τιμή  $120^\circ$  στη μεταβλητή *pos*, η οποία μετά κάθε εκτέλεση των εντολών του βρόχου μειώνεται με βήμα 1 (αυτό είναι το νόημα της εντολής *pos -= 1*, δες και το Παράρτημα Β), και μέχρι η τιμή της μεταβλητής *pos* να γίνει μικρότερη από  $60^\circ$ . Ο βρόχος :


```
for (pos = 120; pos >= 60; pos -= 2)
{
    // Εντολές που πρέπει να εκτελεστούν
}
```

έχει το ίδιο αποτέλεσμα αλλά με βήμα μείωσης 2, δηλ. οδηγεί σε μικρότερο αριθμό επαναλήψεων και τελικά στην περίπτωση του σερβοκινητήρα σε γρηγορότερη περιστροφή του άξονά του.

## Εγκατάσταση και χρήση εξωτερικής βιβλιοθήκης

Πολλές φορές, για παράδειγμα όταν θέλουμε να συνδέσουμε στον Arduino ένα αισθητήρα ή μια οθόνη υγρών κρυστάλλων, κ.λ.π., είναι απαραίτητη η επέκταση των δυνατοτήτων του Arduino με τη χρήση επιπλέον βιβλιοθηκών, οι οποίες δεν είναι εξ ορισμού εγκατεστημένες.

Ο απλούστερος τρόπος να εγκαταστήσουμε μια βιβλιοθήκη είναι μέσω του “*Διαχειριστή βιβλιοθήκης*” του Arduino IDE. Το παράθυρο “*Διαχειριστής βιβλιοθήκης*” ανοίγει μέσω του μενού “*Εργαλεία/Διαχείριση βιβλιοθηκών...*” ή “*Σχέδιο/Συμπερίληψη βιβλιοθήκης/Διαχείριση βιβλιοθηκών...*”. Με βάση τις διάφορες επιλογές που διαθέτει ο διαχειριστής βιβλιοθήκης μπορούμε να αναζητήσουμε και να εγκαταστήσουμε όποια από τις διαθέσιμες βιβλιοθήκες θέλουμε, π.χ. τη βιβλιοθήκη του Steven Wilmouth για τον αναλογικού τύπου αισθητήρα θερμοκρασίας LM35.



Εικόνα 26 : Διαχειριστής βιβλιοθήκης

Σε έναν υπολογιστή με Windows λειτουργικό σύστημα τα αρχεία που συναποτελούν μια βιβλιοθήκη εγκαθίστανται από το διαχειριστή βιβλιοθηκών σε ιδιαίτερο φάκελο μέσα στο φάκελο “*Εγγραφα\Arduino\libraries*”.

Οι βιβλιοθήκες στην απλούστερη μορφή τους αποτελούνται από ένα αρχείο επικεφαλίδων και ορισμών (με κατάληξη “.h”), και συνήθως ακόμη ένα αρχείο κώδικα γραμμένο σε C ή C++. Στην περίπτωση της βιβλιοθήκης LM35\_Sensor το αρχείο επικεφαλίδων (LM35.h) έχει τη μορφή:

```
#ifndef LM35_h
#define LM35_h

#include "Arduino.h"

enum Unity
{
    CELCIUS,
    KELVIN,
    FAHRENHEIT
};
```



επιστρέφει ο αισθητήρας. Το σχετικό sketch τελικά διαμορφώνεται ως εξής:

```
/*
   Ex.13 : Αισθητήρας θερμοκρασίας LM35 και χρήση της σχετικής βιβλιοθήκης
*/

#include <LM35.h>                // Συμπερίληψη βιβλιοθήκης LM35

LM35 lm35temp(A0);              // Ορισμός και αρχικοποίηση αντικειμένου τύπου LM35

void setup()
{
    Serial.begin(9600);          // Εκκίνηση σειρ. επικοινωνίας στα 9600 bps
}

void loop()
{
    Serial.println(lm35temp.getTemp());    // Θερμοκρασία σε βαθμούς Κελσίου
    delay(1000);
}
```

Στο αρχείο επικεφαλίδων της κλάσης LM35 διαπιστώνουμε την ύπαρξη μιας δεύτερης μορφής της συνάρτησης *getTemp*, που στο sketch μας θα μπορούσε να κληθεί π.χ. ως εξής:

```
float tempF = lmd35temp.getTemp(FAHRENHEIT);
```

επιστρέφοντας τη θερμοκρασία σε βαθμούς Fahrenheit (ή σε Kelvin ή βαθμούς Celsius ανάλογα με την τιμή της παραμέτρου που θα χρησιμοποιήσουμε).


**Παρατήρηση :** Υπάρχουν δύο ακόμη τρόποι εγκατάστασης μιας εξωτερικής βιβλιοθήκης στο *Arduino IDE*, και αφορούν την περίπτωση που από το διαδίκτυο έχουμε “κατεβάσει” τον κώδικα στον υπολογιστή μας, συνήθως με τη μορφή ενός συμπιεσμένου αρχείου τύπου “zip”:

1. Αυτόματα μέσω του μενού “Σχέδιο/Συμπερίληψη βιβλιοθήκης/Προσθήκη βιβλιοθήκης ZIP...”, και επιλογή του συμπιεσμένου αρχείου που περιέχει τα αρχεία της βιβλιοθήκης.
2. Χειροκίνητα αποσυμπιέζοντας στο φάκελο “*Εγγραφα\Arduino\libraries\*” το συμπιεσμένο αρχείο.

Αν η βιβλιοθήκη δεν είναι διαθέσιμη για συμπερίληψη μέσω του μενού “Σχέδιο/Συμπερίληψη βιβλιοθήκης” θα χρειαστεί να κάνουμε επανεκκίνηση του *Arduino IDE*.

## Χρήση του διαύλου (I<sup>2</sup>C) σειριακού διαύλου επικοινωνίας

Ο δίαυλος I<sup>2</sup>C είναι ένας σειριακός δίαυλος επικοινωνίας μεταξύ διαφόρων συσκευών, ο οποίος για τη μεταφορά των δεδομένων χρησιμοποιεί μόνο δύο καλώδια (αγωγίμες γραμμές): Τη γραμμή SCL που είναι η γραμμή χρονισμού, και τη γραμμή SDA που είναι η γραμμή δεδομένων. Στις γραμμές αυτές συνδέονται όλες οι συσκευές, που υπάρχουν στο δίαυλο I<sup>2</sup>C. Προφανώς εκτός από τους παραπάνω αγωγούς που μεταφέρουν δεδομένα, απαιτούνται και οι κατάλληλοι αγωγοί τροφοδοσίας των συσκευών. Για τη σωστή λειτουργία του διαύλου επιβάλλεται οι δύο γραμμές δεδομένων να προσδεθούν στη θετική τροφοδοσία μέσω κατάλληλων (pull up) αντιστάσεων [7]. Στην υλοποίηση του διαύλου στον Arduino χρησιμοποιούνται οι εσωτερικές pull up αντιστάσεις (της τάξης των 50 kΩ) που υπάρχουν στις ακίδες εισόδου/εξόδου.



Εικόνα 28: Ο δίαυλος I<sup>2</sup>C

Στο δίαυλο I<sup>2</sup>C κάθε συνδεδεμένη συσκευή αναγνωρίζεται με βάση ένα μοναδικό κωδικό (διεύθυνση) εύρους συνήθως 7 bit (ή και 10bit). Μ' αυτό τον τρόπο ο μέγιστος θεωρητικά αριθμός συσκευών που μπορούν να συνδεθούν στο δίαυλο είναι 128 (για διευθυνσιοδότηση 7bit) ή 1024 (για διευθυνσιοδότηση 10bit). Βέβαια άλλοι παράγοντες, όπως π.χ. η συνολική χωρητικότητα του διαύλου, περιορίζουν σημαντικά τον αριθμό αυτό. Οι συσκευές στον δίαυλο I<sup>2</sup>C μπορεί να είναι είτε κύριες (*masters*), είτε υποτελείς (*slaves*). Και οι κύριες και οι υποτελείς συσκευές μπορούν να μεταφέρουν δεδομένα στον δίαυλο, αλλά μόνο οι κύριες ελέγχουν την μεταφορά, δημιουργώντας τους κατάλληλους παλμούς χρονισμού. Εξ ορισμού στον Arduino ο δίαυλος I<sup>2</sup>C χρονίζεται στα 100 kHz, υπάρχει όμως η δυνατότητα χρονισμού και σε υψηλότερες συχνότητες (π.χ. 400kHz).

Δε θα επεκταθούμε σε περισσότερες λεπτομέρειες σχετικά με το δίαυλο I<sup>2</sup>C. Στον Arduino το σχετικό πρωτόκολλο επικοινωνίας υλοποιείται μέσω της βιβλιοθήκης **Wire**. Για τη χρήση της βιβλιοθήκης πρέπει:

1. Πρώτα να συμπεριληφθεί ο κώδικας που περιέχει με τη δήλωση συμπερίληψης:

```
#include <Wire.h>
```

2. Να ενεργοποιηθεί η σύνδεση στο δίαυλο I<sup>2</sup>C με την εντολή:

```
Wire.begin();
```

Η συνάρτηση *Wire.begin()* δέχεται ως παράμετρο τη διεύθυνση της συσκευής που όμως δεν είναι απαραίτητη προκειμένου για master συσκευή.

Η διαχείριση της επικοινωνίας με κάποια slave συσκευή γίνεται μέσω των εντολών [4]:

1. *Wire.beginTransmission(address)*: Έναρξη επικοινωνίας με τη συσκευή στη διεύθυνση *address* στο δίαυλο.
2. *Wire.write(data)*: Στέλνει δεδομένα (*data*) από τη master συσκευή στη slave, και επιστρέφει τον αριθμό των bytes που στάλθηκαν. Τα δεδομένα μπορεί να είναι ένα byte, μια συμβολοσειρά, ή ένας πίνακας από bytes.

3. *Wire.requestFrom(address, quantity, stop)*: Χρησιμοποιείται από τη master συσκευή για να ζητήσει ορισμένο αριθμό bytes (*quantity*) από τη slave συσκευή. Η παράμετρος *stop* καθορίζει αν μετά τη λήψη των bytes ο διάυλος θα απελευθερωθεί ή θα συνεχίσει να παραμένει ενεργός.
4. *Wire.available()*: Επιστρέφει τον αριθμό των bytes που είναι διαθέσιμα για ανάγνωση. Καλείται από μια master συσκευή μετά από κλήση της *requestFrom()*.
5. *Wire.read()*: Διαβάζει ένα byte που έχει αποσταλεί από μια slave συσκευή στη master μετά από κλήση της *requestFrom()*.
6. *Wire.endTransmission(stop)*: Λήξη της επικοινωνίας. Με αληθή (true) τιμή της παραμέτρου *stop* ο διάυλος απενεργοποιείται μετά τη λήξη της επικοινωνίας, ενώ με ψευδή (false) ο διάυλος παραμένει ενεργός. Η παράμετρος μπορεί ακόμη και να παραληφθεί. Η τιμή που επιστρέφει η συνάρτηση είναι μηδέν (0) αν η επικοινωνία master - slave ήταν επιτυχής, και μεγαλύτερη του 0 στην περίπτωση που κάποιο λάθος παρουσιάστηκε κατά τη διάρκεια της επικοινωνίας.

Το επόμενο sketch (που αποτελεί μια ελαφρά τροποποιημένη εκδοχή του αντίστοιχου sketch από την επίσημη ιστοσελίδα του Arduino) ελέγχει την ύπαρξη συνδεδεμένων συσκευών στο διάυλο, και αν βρεθούν επιστρέφει τη διεύθυνσή τους.

```

/*
   Ex.14 : Ανίχνευση συνδεδεμένων συσκευών στο διάυλο I2C
*/

#include <Wire.h>                                // Συμπερίληψη της βιβλιοθήκης Wire
int nDevices = 0;                                // Αριθμός συνδεδεμένων συσκευών στο διάυλο

void setup()
{
    Wire.begin();                                // Ενεργοποίηση διαύλου I2C
    Serial.begin(9600);                           // Ενεργοποίηση σειριακής επικοινωνίας
    Serial.println("\nI2C Scanner");              // Τυπώνει τον τίτλο του sketch
}

void loop()
{
    byte error, address;
    Serial.println("Scanning...");
    for (address = 1; address < 127; address++)
    {
        Wire.beginTransmission(address);         // Έναρξη επικοινωνίας με τη slave συσκευή
                                                    // στη διεύθυνση address (από 1 ως 126)
        error = Wire.endTransmission();          // Λήξη της επικοινωνίας και λήψη της
                                                    // επιστρεφόμενης τιμής λάθους
        nDevices += checkErrorValue(address, error);
    }
    if (nDevices == 0)                            // Αν δε βρεθούν I2C συσκευές στο διάυλο

```

```

        Serial.println("No I2C devices found\n");
    else // Αλλιώς αν έχουν βρεθεί I2C συσκευές
        Serial.println("done\n");
    nDevices = 0;
    delay(5000); // Αναμονή 5 seconds
}


int checkErrorValue(byte address, int error)
{
    int ret = 0;
    // Αν η τιμή που επιστρέφει η Wire.endTransmission() είναι μηδέν
    // υπάρχει συνδεδεμένη συσκευή στη συγκεκριμένη διεύθυνση του διαύλου
    if (error == 0)
    {
        Serial.print("I2C device found at address 0x");
        if (address < 16)
            Serial.print("0");
        Serial.print(address, HEX);
        Serial.println(" !");
        ret = 1;
    }
    else if (error == 4) // Άγνωστο λάθος στη συγκεκριμένη διεύθυνση
    {
        Serial.print("Unknown error at address 0x");
        if (address < 16)
            Serial.print("0");
        Serial.println(address, HEX);
    }
    return ret;
}

```

#### α. Ο αισθητήρας BMP280


Για να ελέγξουμε το sketch θα συνδέσουμε στον Arduino μια μονάδα που ενσωματώνει τον αισθητήρα ατμοσφαιρικής πίεσης και θερμοκρασίας BMP280 (Εικόνα 29). Σύμφωνα με το φυλλάδιο δεδομένων του κατασκευαστή (Bosch) ο αισθητήρας BMP280 μπορεί να λειτουργήσει με τάση τροφοδοσίας μεταξύ 1,71V και 3,6 V, και συνεπώς **ΔΕΝ είναι ασφαλές** να τον τροφοδοτήσουμε με τάση 5V. Θα μπορούσαμε βέβαια να τροφοδοτήσουμε τη μονάδα με την τάση των 3,3V από την αντίστοιχη ακίδα της πλακέτας του Arduino. Αλλά ακόμη και έτσι υπάρχει κίνδυνος καταστροφής του αισθητήρα, αφού ο αμφίδρομος δίαυλος I<sup>2</sup>C στον Arduino λειτουργεί στα 5V. Δηλαδή, όταν ο Arduino “διαβάζει” δεδομένα στο δίαυλο (που έχουν αποσταλεί από τον αισθητήρα), η τάση στη γραμμή δεδομένων δεν υπερβαίνει τα 3,3V (όση είναι και η τάση τροφοδοσίας του αισθητήρα), ενώ όταν ο Arduino “γράφει” δεδομένα στο δίαυλο η τάση στη γραμμή δεδομένων φτάνει μέχρι και τα 5V. Ο Arduino όμως, σύμφωνα με το φυλλάδιο δεδομένων του ATmega 328), απαιτεί τουλάχιστον 3V για ένα αξιόπιστο υψηλό (HIGH) επίπεδο στις γραμμές SDA και SCL, ενώ τάση μεγαλύτερη των 3,6V στις ίδιες γραμμές συνιστά κίνδυνο καταστροφής του αισθητήρα BMP280. Υπάρχει δη-

λαδή πρόβλημα προσαρμογής των τάσεων μεταξύ της master και της slave συσκευής στο διάλογο I<sup>2</sup>C.




Εικόνα 29: Η μονάδα με τον αισθητήρα BMP280

Η οικονομική λύση που επέλεξε ο κατασκευαστής της μονάδας του αισθητήρα για την αντιμετώπιση του προβλήματος είναι η χρήση αντιστάσεων 10 k $\Omega$ , μέσω των οποίων οι γραμμές SCL και SDA στη μονάδα προσδένονται στην τροφοδοσία των 3,3V (αντιστάσεις pull up).



Εικόνα 30: Το σχηματικό διάγραμμα του αισθητήρα BMP280

Έτσι μέσω του συνδυασμού αυτών των αντιστάσεων με τις αντίστοιχες εσωτερικές (pull up) αντιστάσεις του Arduino, οι οποίες προσδένουν τις γραμμές δεδομένων στα 5V, δημιουργείται ουσιαστικά ένας διαιρέτης τάσης, με αποτέλεσμα η τάση στις γραμμές SDA και SCL να μην υπερβαίνει τα 3,6V. Η λύση είναι επισφαλής, αφού η σύνδεση μιας ακόμη συσκευής στο διάλογο μπορεί να μεταβάλλει προς το χειρότερο την κατάσταση. Η απολύτως ασφαλής λύση είναι η χρήση μιας μονάδας μετατόπισης του επιπέδου της τάσης (level shifter) από τα 5V στα 3,3V, όπως αυτή που φαίνεται στην Εικόνα 31:



Εικόνα 31: Μονάδα αμφίδρομης μετατόπισης επιπέδου τάσης

Κατά την εκτέλεση του sketch για την ανίχνευση συνδεδεμένων συσκευών στο διάλογο I<sup>2</sup>C (ex.14.ino) με τη μονάδα BMP280 συνδεδεμένη στον Arduino, στη σειριακή κονσόλα παίρνουμε την πληροφορία:

*I2C device found at address 0x76 !*

Θέτοντας την ακίδα SDO της μονάδας σε υψηλή λογική κατάσταση (HIGH) μπορούμε να αλλάξουμε τη διεύθυνση με την οποία αναγνωρίζεται η μονάδα στο δίαυλο I<sup>2</sup>C σε 0x77.

Ο αισθητήρας BMP280 είναι μια πολύπλοκη ηλεκτρονική διάταξη, που μπορεί με ρυθμό μέχρι 157 Hz, να μετρήσει:


1. Πίεση μεταξύ 300 και 1100 hPa και με σχετική ακρίβεια  $\pm 0.12$  hPa.
2. Θερμοκρασία μεταξύ  $-40$  και  $85$  °C και με ακρίβεια  $\pm 1$ °C.

Μπορεί επιπλέον μέσω της μέτρησης της ατμοσφαιρικής πίεσης να χρησιμοποιηθεί και για τη μέτρηση υψομετρικών διαφορών με ακρίβεια  $\pm 1$ m.

Ο αισθητήρας υποστηρίζει δύο τρόπους λειτουργίας:

- α. **Εξαναγκασμένη λειτουργία:** Εκτελείται μία μέτρηση με βάση τις τρέχουσες ρυθμίσεις. Με το τέλος του κύκλου μέτρησης ο αισθητήρας μπαίνει σε κατάσταση αναμονής, και τα αποτελέσματα της μέτρησης μπορούν να διαβαστούν.
- β. **Κανονική λειτουργία:** Συνεχής λειτουργία εναλλασσόμενη μεταξύ περιόδων μέτρησης και περιόδων αναμονής. Ο χρόνος αναμονής μπορεί να ρυθμιστεί μεταξύ 0,5ms και 4 s.

Στο υλικό του αισθητήρα περιλαμβάνεται και ένας μετατροπέας αναλογικού σε ψηφιακό διακριτικής ικανότητας 16-bit. Τεχνικές βελτίωσης της διακριτικής ικανότητας και μείωσης του ψηφιακού θορύβου (oversampling) του αισθητήρα μπορούν να χρησιμοποιηθούν τόσο για την πίεση, όσο και για τη θερμοκρασία, ενώ υπάρχει και η δυνατότητα χρήσης εσωτερικού (υλοποιημένου στο υλικό του αισθητήρα) φίλτρου για την εξομάλυνση των απότομων κορυφών, που προκύπτουν λόγω ανεπιθύμητων διαταραχών κατά τη διάρκεια των μετρήσεων. Επιπλέον τόσο η μέτρηση της πίεσης όσο και η μέτρηση της θερμοκρασίας μπορούν να παρακαμφθούν, με αποτέλεσμα την αύξηση του ρυθμού μέτρησης του μεγέθους που παραμένει προς μέτρηση.



Εικόνα 32: Σύνδεση της μονάδας BMP280 στον Arduino

Τις δυνατότητες του αισθητήρα μπορούμε να εκμεταλλευτούμε με τη χρήση της κατάλληλης βιβλιοθήκης. Θα χρησιμοποιήσουμε στη συνέχεια τη βιβλιοθήκη BME280 του Tyler Glenn, η οποία μπορεί να εγκατασταθεί μέσω του “*Διαχειριστή βιβλιοθήκης*” του Arduino IDE. Η ίδια βιβλιοθήκη υποστηρίζει και τον αισθητήρα BME280 της ίδιας εταιρείας ο οποίος διαθέτει επιπλέον δυνατότητες μέτρησης της σχετικής υγρασίας με χρόνο απόκρισης 1s και ακρίβεια περί το 3%.

Για τις ανάγκες των δοκιμών μας θα συνδέσουμε τη μονάδα του αισθητήρα BMP280 απευθείας (χωρίς χρήση μονάδας μετατόπισης του επιπέδου τάσης) στον Arduino, όπως φαίνεται και στην

Εικόνα 32. Αυτή η επιλογή δε δημιουργεί κίνδυνο καταστροφής του αισθητήρα, αφού -προς το παρόν- δε πρόκειται να συνδέσουμε άλλον αισθητήρα στο διάλυο I<sup>2</sup>C.

Όπως έχουμε ήδη αναφέρει πρέπει ιδιαίτερα να προσέξουμε ότι **ο αισθητήρας τροφοδοτείται από την ακίδα των 3,3V της πλακέτας του Arduino.**

Με το επόμενο sketch, αφού γίνει πρώτα μια μέτρηση με τον αισθητήρα BMP280 και με τις εξ ορισμού ρυθμίσεις, τυπώνονται στη σειριακή κονσόλα οι τιμές της ατμοσφαιρικής πίεσης και της θερμοκρασίας περιβάλλοντος:

```
/*
   Ex.15 : BMP280 I2C Test
*/

#include <Wire.h>           // Συμπερίληψη της βιβλιοθήκης Wire
#include <BME280I2C.h>     // Συμπερίληψη της βιβλιοθήκης BME280I2C

BME280I2C bme;           // Εξ ορισμού ρυθμίσεις : forced mode, standby time = 1000 ms
                        // Oversampling = pressure ×1, temperature ×1, humidity ×1, filter off,

void setup()
{
  Serial.begin(9600);    // Ενεργοποίηση σειριακής επικοινωνίας
  Wire.begin();         // Ενεργοποίηση διαύλου I2C

  while(!bme.begin())  // Αναμονή για ανίχνευση συμβατής συσκευής
  {
    Serial.println("Could not find BMP280 sensor!");
    delay(1000);
  }
}

void loop()
{
  printBME280Data();   // Λήψη και τύπωση των αποτελεσμάτων της μέτρησης
  delay(1000);
}

void printBME280Data()
{
  float temp(NAN), pres(NAN), hum(NAN);

  bme.read(pres, temp, hum);
  Serial.print("Temperature: ");
  Serial.print(temp);
  Serial.print("°C");
  Serial.print("\t\tPressure: ");
  Serial.print(pres);
  Serial.println(" Pa");
}
```


## β. Ρολόι πραγματικού χρόνου (RTC) DS1307

Ένα ρολόι πραγματικού χρόνου παρέχει στον Arduino πληροφορίες για την τρέχουσα ημερομηνία και ώρα. Η πιο απλή και οικονομική λύση για χρήση ενός ρολογιού πραγματικού χρόνου στον Arduino είναι η μονάδα RTC DS1307 (Tiny RTC), που χρησιμοποιεί το σχετικό ολοκληρωμένο κύκλωμα DS1307 της Dallas Semiconducors, στα χαρακτηριστικά του οποίου περιλαμβάνονται:

- Ρολόι πραγματικού χρόνου ακριβές μέχρι το έτος 2100.
- 56 bytes μη πτητικής μνήμης RAM διαθέσιμη στο χρήστη.
- Προγραμματιζόμενη έξοδο τετραγωνικού σήματος.
- Κύκλωμα ανίχνευσης διακοπής τροφοδοσίας, κ.ά..

Μια λίγο πιο ακριβή, αλλά και ακριβέστερη λύση, είναι η μονάδα με το ολοκληρωμένο κύκλωμα DS3231, και πάλι από τη Dallas Semiconductors.

Η μονάδα Tiny RTC επικοινωνεί με τον Arduino μέσω του σειριακού διαύλου I<sup>2</sup>C -κάτι που απλοποιεί κατά πολύ τη σύνδεση- και αναγνωρίζεται στο δίαυλο στη δεξιαεξασδική διεύθυνση 0x68.



Εικόνα 33: Η μονάδα RTC DS1307 (Tiny RTC)

Εκτός από το ρολόι πραγματικού χρόνου και τα απαραίτητα για τη λειτουργία του παθητικά και ενεργά εξαρτήματα, η μονάδα Tiny RTC (Εικόνα 33) διαθέτει επιπλέον μια I<sup>2</sup>C μνήμη EEPROM 32 kbits ή 4 kBytes (ATMEL 24C32), η οποία αναγνωρίζεται στη διεύθυνση 0x50 στο δίαυλο I<sup>2</sup>C, ενώ έχει προβλεφθεί και η επέκταση της μονάδας με το ψηφιακό θερμόμετρο DS18B20 (περισσότερα για αυτό σε επόμενη παράγραφο). Επιπλέον διαθέτει μια μπαταριοθήκη για μπαταρίες λιθίου (τύπου CR1225 ή CR2032), για τη λειτουργία και διατήρηση των δεδομένων όταν η μονάδα δεν τροφοδοτείται από εξωτερική τάση. Η ακιδοσειρά P1 παρέχει όλες τις απαραίτητες ακίδες για την πλήρη εκμετάλλευση των δυνατοτήτων της μονάδας [8]:

- VCC : Σύνδεση της θετικής τροφοδοσίας (3,0 – 5,5V)
- GND : Γείωση
- SCL : I<sup>2</sup>C Clock
- SDA : I<sup>2</sup>C Data
- BAT : Για μέτρηση της τάσης της μπαταρίας
- DS : Για σύνδεση του ψηφιακού θερμομέτρου στο μονοσύρματο δίαυλο 1-Wire
- SQ : Έξοδος τετραγωνικών παλμών ρυθμιζόμενης συχνότητας

Οι ακίδες P2 της μονάδας Tiny RTC αποτελούν αντίγραφα των αντίστοιχων ακίδων της ακιδοσειράς P1. Ένα από τα μειονεκτήματα της διάταξης είναι πως η συχνότητα του ταλαντωτή που διαθέτει επηρεάζεται από εξωτερικούς παράγοντες, όπως π.χ. η θερμοκρασία με αποτέλεσμα να “χάνει” περί τα πέντε δευτερόλεπτα ανά μήνα.



```

Serial.print(" ");

Serial.print(now.hour(), DEC);
Serial.print(':');
Serial.print(now.minute(), DEC);
Serial.print(':');
Serial.println(now.second(), DEC);

delay(1000);           // Αναμονή 1 sec
}

```

Πριν την πρώτη χρήση της μονάδας Tiny RTC, και αφού έχουμε τοποθετήσει την κατάλληλη μπαταρία στη μπαταριοθήκη, πρέπει να ορίσουμε τη σωστή ημερομηνία και το σωστό χρόνο για την έναρξη λειτουργίας της μονάδας. Τον τρόπο που μπορούμε να το κάνουμε, δείχνουμε στο επόμενο sketch:

```

/*
   Ex.17 : Ρύθμιση ημερομηνίας και ώρας στη μονάδα Tiny RTC
*/

#include <Wire.h>           // Συμπερίληψη της βιβλιοθήκης Wire
#include <RTClib.h>        // Συμπερίληψη της βιβλιοθήκης RTClib

RTC_DS1307 rtc;           // Δημιουργία αντικειμένου τύπου RTC_DS1307

void setup ()
{
  Serial.begin(9600);      // Ενεργοποίηση σειριακής επικοινωνίας
  rtc.begin();            // Εκκίνηση επικοινωνίας με RTC μέσω I2C

  // Η επόμενη γραμμή θέτει το ρολόι στην ημερομηνία και ώρα μεταγλώττισης του sketch
  rtc.adjust(DateTime(__DATE__, __TIME__));
}

void loop()
{
  // Τίποτα δε χρειάζεται να γίνει εδώ
}

```

Οι μεταβλητές `__DATE__` και `__TIME__` του Arduino IDE κρατούν την ημερομηνία και την ώρα που έγινε η μεταγλώττιση του sketch. Μεταξύ μεταγλώττισης και εκτέλεσης του κώδικα μεσολαβεί το ανέβασμα του κώδικα στον Arduino, και συνεπώς με αυτό το sketch το ρολόι πραγματικού χρόνου βρίσκεται εκτός συγχρονισμού κατά μισό περίπου λεπτό. Η συνάρτηση `rtc.adjust()` δέχεται μια παράμετρο τύπου `DateTime` (που επίσης ορίζεται στη βιβλιοθήκη `RTClib`) και ρυθμίζει την ημερομηνία και την ώρα στο ρολόι πραγματικού χρόνου. Το αντικείμενο τύπου `DateTime` δέχεται μια ημερομηνία και ώρα με πολλούς διαφορετικούς τρόπους και επιστρέφει μέσω κατάλληλων συναρτήσεων επτά επιμέρους στοιχεία: έτος, μήνας, ημέρα του μήνα, ημέρα της εβδομάδας, ώρα, λεπτά και δευτερόλεπτα. Έτσι αν θέλουμε να ορίσουμε στο RTC την ημερομηνία ως 1/1/2019 και την ώρα ως 14:00:00, μπορούμε να χρησιμοποιήσουμε την εντολή:

```

rtc.adjust(DateTime(2019,1,1,14,0,0)); // Format: YYYY,MM,DD,HH,MM,SS

```

Το επόμενο sketch αποτελεί έναν τρόπο για ακριβέστερο ορισμό της ημερομηνίας και της ώρας στο

ρολόι πραγματικού χρόνου. Τα επιμέρους στοιχεία (έτος, μήνας, ημέρα, κ.λ.π.) εισάγονται κατά την εκτέλεση του κώδικα μέσω της σειριακής κονσόλας.

```

/*
   Ex.18 : Ακριβής ρύθμιση ημερομηνίας και ώρας στο DS1307
*/

#include <Wire.h>           // Συμπερίληψη της βιβλιοθήκης Wire
#include <RTClib.h>        // Συμπερίληψη της βιβλιοθήκης RTClib

RTC_DS1307 rtc;           // Δημιουργία αντικειμένου τύπου RTC_DS1307

void setup ()
{
  Serial.begin(115200);    // Ενεργοποίηση σειριακής επικοινωνίας
  Serial.setTimeout(30000); // Ορισμός χρόνου αναμονής για εισαγωγή
                           // χαρακτήρων από τη σειριακή κονσόλα

  rtc.begin();            // Εκκίνηση επικοινωνίας με RTC μέσω I2C

  // Με το επόμενο μπλοκ εντολών γίνεται λήψη δεδομένων ημερομηνίας και ώρας
  // που έχουν αποσταλεί μέσω της σειριακής κονσόλας
  int yy = input("Enter the year.... ", 2000, 2050, 2000);
  int mm = input("Enter the month... ", 1,12,1);
  int dd = input("Enter the day..... ", 1,31,1);
  int hh = input("Enter hour..... ", 0,23,0);
  int mn = input("Enter mins..... ", 0,59,0);
  int ss = input("Enter secs..... ", 0,59,0);

  // Με το επόμενο μπλοκ εντολών γίνεται εκκαθάριση της σειριακής buffer
  while (Serial.available() > 0) // Αν υπάρχουν διαθέσιμα σειριακά δεδομένα
  {
    char t = Serial.read();      // διάβασέ τα, ώστε η buffer να αδειάσει
  }

  Serial.println();              // Τυπώνει μια κενή γραμμή

  /*
  Με το επόμενο μπλοκ εντολών τυπώνεται ένα μήνυμα οδηγιών στη σειριακή κονσόλα
  και δημιουργείται ένας βρόχος αναμονής μέχρι να αποσταλεί σειριακά ένας χαρακτήρας.
  Στη συνέχεια ο χαρακτήρας διαβάζεται και αν είναι ο 'T' ρυθμίζεται η ώρα
  και η ημερομηνία στο RTC με βάση τα δεδομένα που έχουν ήδη εισαχθεί.
  */
  Serial.println("Send T to set Datetime, or any other key to Cancel...");

  while (Serial.available() == 0); // Αναμονή μέχρι να αποσταλεί ένας χαρακτήρας
  char inChar = Serial.read();     // Διάβασμα του χαρακτήρα από τη buffer
  if (inChar == 'T')               // Αν ο χαρακτήρας είναι ο 'T'
  {
    rtc.adjust(DateTime(yy,mm,dd,hh,mn,ss)); // Ρύθμιση του RTC
  }
  Serial.println();                // Τυπώνει μια κενή γραμμή
}

```

```

void loop()                // Τυπώνει την τρέχουσα ημερομηνία και ώρα στη σειριακή κονσόλα
{
    DateTime now = rtc.now();    // Διάβασμα ημερομηνίας - ώρας από το RTC

    // Με το επόμενο μπλοκ εντολών τυπώνεται η ημερομηνία και ώρα στη σειριακή κονσόλα
    Serial.print(now.year(), DEC);
    Serial.print('/');
    Serial.print(now.month(), DEC);
    Serial.print('/');
    Serial.print(now.day(), DEC);
    Serial.print(" ");
    Serial.print(now.hour(), DEC);
    Serial.print(':');
    Serial.print(now.minute(), DEC);
    Serial.print(':');
    Serial.println(now.second(), DEC);

    delay(1000);            // Αναμονή 1 sec
}

// Η συνάρτηση input() επιστρέφει έναν ακέραιο αριθμό μεταξύ των τιμών vmin και vmax,
// με βάση τους χαρακτήρες που αποστέλλονται από τη σειριακή κονσόλα
int input(char * msg, int vmin, int vmax, int vdefault)
{
    Serial.print(msg);      // Τυπώνει το μήνυμα της παραμέτρου msg
    int val = Serial.parseInt(); // Διαβάζει τους χαρακτήρες που αποστέλλονται
                                // μέσω σειριακής κονσόλας, και κωδικοποιεί
                                // το αποτέλεσμα ως ακέραιο αριθμό

    if ((val < vmin) || (val > vmax)) // Αν το αποτέλεσμα είναι εκτός των ορίων
    {                                  // που ορίζουν οι παράμετροι vmin και vmax
        val = vdefault;              // Το αποτέλεσμα παίρνει την εξ' ορισμού τιμή
    }

    Serial.println(val);      // Τυπώνει το αποτέλεσμα
    return val;              // Επιστρέφει το αποτέλεσμα
}


```

Για τη σωστή λειτουργία του κώδικα πρέπει στη γραμμή κατάστασης της σειριακής κονσόλας να ενεργοποιηθεί η επιλογή “Αλλαγή γραμμής”. Τα δεδομένα που γράφουμε στη γραμμή κειμένου της σειριακής κονσόλας αποστέλλονται στον Arduino είτε με κλικ στο κουμπί “Αποστολή”, είτε πατώντας το πλήκτρο Enter.

**Σημείωση :** Η βιβλιοθήκη *RTCLib* συνοδεύεται από αρκετά έτοιμα παραδείγματα κώδικα, μεταξύ των οποίων επισημαίνουμε την ύπαρξη ενός που δείχνει πώς γράφουμε ή διαβάζουμε μερικά bytes στην ή από την εσωτερική μνήμη RAM του ολοκληρωμένου *DS1307*, καθώς και ένα που δείχνει πως μπορούμε να ορίσουμε τη συχνότητα του τετραγωνικού παλμού που το ολοκληρωμένο παράγει.


### γ. Χρήση οθόνης υγρών κρυστάλλων (LCD) μέσω διαύλου I<sup>2</sup>C

Μια από τις πιο συνηθισμένες συσκευές που συνδέεται σε ένα μικροελεγκτή είναι η οθόνη υγρών κρυστάλλων (LCD). Πιο κοινές είναι οι οθόνες που μπορούν να απεικονίσουν 2 γραμμές των 16 χαρακτήρων (16x2), ή 4 γραμμές των 20 χαρακτήρων (20x4). Τον έλεγχο της οθόνης αναλαμβάνει εξειδικευμένο ολοκληρωμένο κύκλωμα (συνήθως το HD44780), το οποίο συνδέεται στο μικροελεγκτή μέσω 14 ακίδων (παράλληλη διεπαφή).




Εικόνα 35: LCD παράλληλης διεπαφής

Με ένα μικροελεγκτή περιορισμένων ακίδων, όπως ο Arduino, η χρήση LCD παράλληλης διεπαφής εύκολα μπορεί να αφήσει τα πρότζεκτ μας χωρίς πόρους. Η λύση στο πρόβλημα αυτό είναι η χρήση μιας οθόνης LCD εφοδιασμένης με ειδικό μετατροπέα της παράλληλης σύνδεσης σε σειριακή για το δίαυλο I<sup>2</sup>C. Τις δυνατότητες της I<sup>2</sup>C οθόνης LCD μπορούμε να εκμεταλλευτούμε πλήρως με τη χρήση της βιβλιοθήκης *LiquidCrystal\_I2C* του Frank de Brabander, η οποία μπορεί να εγκατασταθεί μέσω του “Διαχειριστή βιβλιοθήκης” του Arduino IDE.



Εικόνα 36: Οθόνη LCD με I<sup>2</sup>C διεπαφή (πίσω όψη)

Μια οθόνη LCD χρησιμοποιεί στο δίαυλο I<sup>2</sup>C μια διεύθυνση μεταξύ 0x20 ως 0x27. Η ακριβής διεύθυνση της μονάδας που διαθέτετε μπορεί να βρεθεί μέσω του sketch ανίχνευσης συνδεδεμένων συσκευών στο δίαυλο I<sup>2</sup>C (ex.14.ino), που στην αρχή αυτής της ενότητας συζητήσαμε. Στο επόμενο δοκιμαστικό sketch θεωρούμε πως η χρησιμοποιούμενη οθόνη διαθέτει 2 γραμμές των 16 χαρακτήρων και αναγνωρίζεται στη διεύθυνση 0x27 στο δίαυλο I<sup>2</sup>C. Αφού λοιπόν συμπεριλάβουμε στο sketch τη σχετική βιβλιοθήκη και δημιουργήσουμε ένα αντικείμενο τύπου *LiquidCrystal\_I2C*, είμαστε έτοιμοι να χρησιμοποιήσουμε την οθόνη για την εμφάνιση πληροφοριών και αποτελεσμάτων από το sketch μας. Ακολουθούν η εικόνα της σύνδεσης της οθόνης στον Arduino, και ο κώδικας του sketch:



fritzing

Εικόνα 37: Σύνδεση του LCD στον Arduino

```

/*
   Ex.19 : Δοκιμή της I2C οθόνης υγρών κρυστάλλων
*/

#include <Wire.h> // Συμπερίληψη βιβλιοθήκης Wire
#include <LiquidCrystal_I2C.h> // Συμπερίληψη βιβλιοθήκης LiquidCrystal_I2C
LiquidCrystal_I2C lcd(0x27,16,2); // Δημιουργία αντικειμένου LiquidCrystal_I2C
                                   // για οθόνη τύπου 16x2 στην I2C διεύθυνση 0x27
void setup()
{
    lcd.init(); // Ενεργοποίηση οθόνης


    lcd.backlight(); // Ενεργοποίηση οπίσθιου φωτισμού
    lcd.clear(); // Εκκαθάριση οθόνης
    lcd.setCursor(2,0); // Ο δρομέας στο 3ο χαρακτήρα της 1ης γραμμής
    lcd.print("Hello world!"); // Τυπώνει ένα μήνυμα
    lcd.setCursor(1,1); // Ο δρομέας στο 2ο χαρακτήρα της 2ης γραμμής
    lcd.print("I2C LC Display"); // Τυπώνει ένα μήνυμα
}

void loop()
{
    // Τίποτα δε γίνεται εδώ. Όλα έγιναν στη συνάρτηση setup()
}

```

## Χρήση του μονοσύρματου διαύλου 1-Wire

Το σειριακό πρωτόκολλο 1-Wire σχεδιάστηκε από τη Dallas Semiconductors (τώρα Maxim) για τη χαμηλού ρυθμού επικοινωνία (16,3 kbps) μεταξύ των συνδεδεμένων συσκευών, χρησιμοποιώντας μια απλή γραμμή για τη μεταφορά και τον έλεγχο της ροής των δεδομένων. Βεβαίως απαιτούνται και οι κατάλληλες γραμμές τροφοδοσίας (Vcc και GND), ενώ διαθέτει και τη λεγόμενη λειτουργία παρασιτικής τροφοδοσίας, στην οποία η γραμμή μεταφοράς δεδομένων χρησιμοποιείται και για την τροφοδοσία των συνδεδεμένων στο δίαυλο συσκευών, περιορίζοντας έτσι τον αριθμό αγωγών του διαύλου σε δύο: γραμμή μεταφοράς και γείωση. Και για τους δύο τύπους τροφοδοσίας μια pull up αντίσταση των 4,7 kΩ πρέπει να συνδεθεί στο δίαυλο 1-Wire (Εικόνα 38). Μια 1-Wire κύρια (master) συσκευή ξεκινάει και ελέγχει την επικοινωνία με μία ή περισσότερες υποτελείς (slave) συσκευές στο δίαυλο [9].




Εικόνα 38: Ο δίαυλος 1-Wire

Κάθε 1-Wire υποτελής (slave) συσκευή διαθέτει μια 64-bit μνήμη ROM, στην οποία έχουν ερροστασιακά εγγραφεί:

- Ένας 8-bit “οικογενειακός” κωδικός που χρησιμοποιείται για την αναγνώριση του τύπου της συσκευής.
- Ένας 48-bit αναγνωριστικός αριθμός, που χρησιμεύει ως η διεύθυνση της συσκευής στο δίαυλο 1-Wire.
- Ένας 8-bit κωδικός CRC που χρησιμοποιείται για την επιβεβαίωση της ακεραιότητας των δεδομένων της μνήμης.


Για να εκμεταλλευτούμε στον Arduino τις δυνατότητες του διαύλου 1-Wire θα χρησιμοποιήσουμε τη βιβλιοθήκη *OneWire* του Paul Stoffregen. Όμως αντίθετα από το δίαυλο I<sup>2</sup>C, που η σχετική βιβλιοθήκη εγκαθίσταται εξ ορισμού με το λογισμικό του Arduino, η βιβλιοθήκη *OneWire* για την υλοποίηση του πρωτοκόλλου 1-Wire απαιτεί εγκατάσταση από το χρήστη, η οποία μπορεί να πραγματοποιηθεί στο Arduino IDE μέσω του μενού “Εργαλεία/Διαχείριση βιβλιοθηκών...” ή του “Σχέδιο\Συμπερίληψη βιβλιοθήκης\Διαχείριση βιβλιοθηκών...”.




Εικόνα 39: Ο αισθητήρας DS18B20 (αδιάβροχη μορφή και μορφή ολοκληρωμένου 3 ακιδών)

Η πλέον συνηθισμένη συσκευή που μπορεί να συνδεθεί στο δίαυλο 1-Wire είναι το ψηφιακό θερμόμετρο DS18B20 που κατασκευάζεται από τη Dallas Semiconductors. Μετράει θερμοκρασίες από  $-55^{\circ}\text{C}$  μέχρι  $125^{\circ}\text{C}$  με ακρίβεια  $\pm 0,5^{\circ}\text{C}$  και διακριτική ικανότητα από 9bit μέχρι 12bit. Διατίθεται με τη μορφή ενός μικρού ολοκληρωμένου κυκλώματος τριών ακίδων, ενώ ως εμπορικό προϊόν υπάρχει και με την αδιάβροχη μορφή του (Εικόνα 39).

Οι αισθητήρες DS18B20 μπορούν να συνδεθούν στον Arduino είτε σε κανονικό (σύνδεση με τρεις αγωγούς) είτε σε παρασιτικό τρόπο (σύνδεση με δύο αγωγούς) λειτουργίας, όπως φαίνεται στα επόμενα σχήματα [10]:




Εικόνα 40: Αισθητήρες DS18B20 σε κανονικό τρόπο λειτουργίας




Εικόνα 41: Αισθητήρες DS18B20 σε παρασιτικό τρόπο λειτουργίας

Ο αισθητήρας μπορεί να συνδεθεί (κανονικός τρόπος λειτουργίας) στον Arduino όπως φαίνεται και στο επόμενο σχήμα:



Εικόνα 42: Σύνδεση του αισθητήρα DS18B20 στον Arduino



Εικόνα 43: Σύνδεση της αδιάβροχης εκδοχής του DS18B20 στον Arduino

Για να χρησιμοποιήσουμε τον αισθητήρα DS18B20 στα sketch μας, εκτός από τη βιβλιοθήκη *OneWire* για την υλοποίηση του διαύλου 1-Wire, θα χρειαστούμε επιπλέον και τη βιβλιοθήκη *Dallas Temperature* του Miles Burton, η οποία υποστηρίζει τους αισθητήρες της συγκεκριμένης οικογένειας. Και πάλι η εγκατάσταση της βιβλιοθήκης γίνεται από το Arduino IDE μέσω του “*Διαχειριστή βιβλιοθήκης*”.

Το επόμενο sketch αναζητά συσκευές στο δίαυλο 1-Wire και, όταν βρει μια, τυπώνει τη διεύθυνσή της στη σειριακή κονσόλα:

```

/*
 *
 *   Ex.20 : Αναζήτηση συσκευών στο δίαυλο 1-Wire
 *
#include <OneWire.h>
#define W1PIN      7      // Η ψηφιακή ακίδα στην οποία συνδέεται ο αγωγός δεδομένων

/*
 */
OneWire owbus(W1PIN);    // Δημιουργία αντικειμένου για τον έλεγχο του διαύλου 1-Wire

void setup(void)
{
    Serial.begin(9600);    // Ενεργοποίηση σειριακής επικοινωνίας
    searchBus();          // Αναζήτηση για 1-Wire συσκευές στο δίαυλο
}

void searchBus(void)      // Αναζήτηση της επόμενης συσκευής στο δίαυλο
{
    byte addr[8];         // Πίνακας 8 bytes για αποθήκευση της διεύθυνσης της συσκευής
    byte nDevices = 0;    // Αριθμός των συσκευών στο δίαυλο

    Serial.println("Searching for 1-Wire devices...\n\r");
    while(owbus.search(addr))    // Βρόχος αναζήτησης συσκευών στο δίαυλο
    {
        Serial.print("Address = ");
    }

```

```

For (byte i = 0; i < 7; i++) // Τυπώνει τα 7 bytes της διεύθυνσης της συσκευής
{
    if (addr[i] < 16)
    {
        Serial.print('0');
    }
    Serial.print(addr[i], HEX);
    Serial.print(":");
}
Serial.println(addr[7], HEX); // Τυπώνει και το τελευταίο byte της διεύθυνσης
nDevices++; // Αυξάνει τον αριθμό των συσκευών κατά 1
}

owbus.reset_search(); // Προετοιμασία για την επόμενη αναζήτηση

if (nDevices > 0) // Τυπώνει το τελικό μήνυμα ανάλογα αν βρέθηκαν ή όχι συσκευές
{
    Serial.println();
    Serial.print(nDevices);
    Serial.println(" device(s) found.");
}
else
{
    Serial.println("No devices found.");
}
}

void loop(void)
{
    // Τίποτα δε γίνεται εδώ. Όλα έγιναν στη συνάρτηση setup()
}

```

Για να χρησιμοποιήσουμε τον αισθητήρα για μέτρηση της θερμοκρασίας, απαιτείται κατ' αρχάς να συμπεριλάβουμε στο sketch και τη βιβλιοθήκη *Dallas Temperature*, με τη δήλωση:

```
#include <DallasTemperature.h>
```

Έπειτα πρέπει να ορίσουμε ένα αντικείμενο τύπου *DallasTemperature* για να αποκτήσουμε πρόσβαση στον κώδικα που αφορά τη διαχείριση του αισθητήρα DS18B20. Για τη δημιουργία του αντικειμένου *DallasTemperature* απαιτείται ως παράμετρος η διεύθυνση ενός αντικειμένου τύπου *OneWire*, που πρέπει να οριστεί νωρίτερα για τη δημιουργία και τον έλεγχο του διαύλου 1-Wire:

```

#define WIBUS 7
OneWire owbus(WIBUS);
DallasTemperature devices(&owbus);

```

Στη συνάρτηση *setup()* του sketch αρχικοποιείται ο δίαυλος 1-Wire και αναγνωρίζονται οι συνδεδεμένες συσκευές μέσω της εντολής:

```
devices.begin();
```

Για τη μέτρηση της θερμοκρασίας πρέπει να στείλουμε μια γενική αίτηση στο δίαυλο για επιστροφή των θερμοκρασιών από όλους τους συνδεδεμένους αισθητήρες με την εντολή:

```
devices.requestTemperatures();
```

Στη συνέχεια η ανάγνωση των επιστρεφόμενων τιμών θερμοκρασίας μπορεί να γίνει με δύο τρόπους:

1. Αν είναι γνωστή η διεύθυνση του αισθητήρα, π.χ. ως πίνακας τύπου *DeviceAddress*:

```
DeviceAddress sensor1 = { 0x28, 0xFF 0x0B, 0xB0, 0x67, 0x14, 0x03, 0x43 };
```

τότε μπορούμε άμεσα να καλέσουμε τη συνάρτηση *getTempC()* της βιβλιοθήκης *Dallas Temperature*, ως εξής:

```
float tempC = devices.getTempC(sensor1);
```

2. Αν δεν είναι γνωστή η διεύθυνση του αισθητήρα, μπορούμε να διαβάσουμε τη θερμοκρασία από τον πρώτο για παράδειγμα αισθητήρα, που αναγνωρίστηκε στο δίαυλο, με την εντολή:

```
devices.getTempCByIndex(0);
```

και ανάλογα για του υπόλοιπους (αν υπάρχουν) αισθητήρες στο δίαυλο.

Τα επόμενα δύο sketches υλοποιούν τους δύο αυτούς τρόπους που περιγράψαμε:

```
/*
```

```
Ex.21 : Πρώτο παράδειγμα χρήσης αισθητήρα DS18B20
```

```
*/
```

```
#include <OneWire.h>
```

```
#include <DallasTemperature.h>
```

```
#define W1BUS 7
```

```
// Η γραμμή μεταφοράς δεδομένων του διαύλου 1-Wire
```

```
// συνδέεται στην ακίδα 7 του Arduino
```

```
OneWire owbus(W1BUS);
```

```
// Ορισμός αντικειμένου διαχείρισης διαύλου 1-Wire
```

```
DallasTemperature devices(&owbus);
```

```
// Ορισμός αντικειμένου DallasTemperature
```

```
// Ορισμός της διεύθυνσης του συνδεδεμένου αισθητήρα.
```

```
// Πρέπει να αντικατασταθεί με τη διεύθυνση του δικού σας αισθητήρα
```

```
DeviceAddress sensors1 = { 0x28, 0xFF, 0x0B, 0xB0, 0x67, 0x14, 0x03, 0x43 };
```

```
void setup(void)
```

```
{
```

```
Serial.begin(9600);
```

```
// Ενεργοποίηση σειριακής επικοινωνίας
```

```
devices.begin();
```

```
// Ενεργοποίηση διαύλου 1-Wire
```

```
// και αναγνώριση των συνδεδεμένων συσκευών
```

```
devices.setResolution(sensors1, 10);
```

```
// Ορισμός διακριτικής ικανότητας αισθητήρα
```

```
}
```

```
void loop(void)
```

```
{
```

```
devices.requestTemperatures();
```

```
// Γενική αίτηση για λήψη μετρήσεων
```

```
float tempC = devices.getTempC(sensors1);
```

```
// Ανάγνωση θερμοκρασίας
```

```
Serial.print("Temperature: ");
```

```
// Τυπώνει τη θερμοκρασία
```

```
Serial.print(tempC);
```

```
Serial.println("°C");
```

```
delay(1000);
```

```
// Αναμονή 1 sec
```

```
}
```

```

/*
   Ex.22 : Δεύτερο παράδειγμα χρήσης αισθητήρα DS18B20
*/

#include <OneWire.h>
#include <DallasTemperature.h>

#define W1BUS 7 // Η γραμμή μεταφοράς δεδομένων συνδέεται
                // στην ακίδα 7 του Arduino

OneWire owbus(W1BUS); // Ορισμός αντικειμένου διαχείρισης διαύλου I-Wire
DallasTemperature devices(&owbus); // Ορισμός αντικειμένου DallasTemperature

void setup(void)
{
    Serial.begin(9600); // Ενεργοποίηση σειριακής επικοινωνίας

    // Ενεργοποίηση διαύλου I-Wire και αναγνώριση των συνδεδεμένων συσκευών
    devices.begin();
}

void loop(void)
{
    devices.requestTemperatures(); // Γενική αίτηση για λήψη μετρήσεων

    float tempC = devices.getTempCByIndex(0); // Ανάγνωση θερμοκρασίας από τον
                                                // πρώτο αισθητήρα του διαύλου
    Serial.print("Temperature: "); // Τυπώνει τη θερμοκρασία
    Serial.print(tempC);
    Serial.println("°C");
    delay(1000); // Αναμονή 1 sec
}

```

## Χρήση του διαύλου SPI


Το Serial Peripheral Interface (SPI) είναι ένα σειριακό πρωτόκολλο που χρησιμοποιείται συνήθως για την επικοινωνία (αποστολή και λήψη δεδομένων) μεταξύ μικροελεγκτών και περιφερειακών συσκευών, όπως αισθητήρες, κάρτες SD, κ.ά.. Πρόκειται για ένα γρήγορο (εξ ορισμού ρυθμός επικοινωνίας στον Arduino Uno 4 MHz), εύκολο και απλό στην υλοποίηση πρωτόκολλο, που υποστηρίζει την επικοινωνία μεταξύ μιας κύριας (*master*) και πολλών υποτελών (*slaves*) συσκευών. Το πρωτόκολλο υποστηρίζει σύγχρονη επικοινωνία, δηλ. επικοινωνία ελεγχόμενη από κατάλληλα χρονισμένους παλμούς για τον τέλει συγχρονισμό master - slave συσκευών.

Τυπικά για την υλοποίηση του πρωτοκόλλου SPI, και εκτός από τους αγωγούς τροφοδοσίας, απαιτούνται τρεις γραμμές (αγωγοί) κοινές για όλες τις συσκευές που πρόκειται να συνδεθούν στο δίαυλο [11]:

- MISO (Master In Slave Out): Η γραμμή μεταφοράς δεδομένων από τις υποτελείς συσκευές προς την κύρια.
- MOSI (Master Out Slave In): Η γραμμή μεταφοράς δεδομένων από την κύρια προς τις υποτελείς συσκευές.
- SCK ή SCLK (Serial Clock): Τα σήματα στη γραμμή αυτή συγχρονίζουν τη μεταφορά των δεδομένων, και δημιουργούνται από την κύρια συσκευή του διαύλου.

Επιπλέον για κάθε υποτελή συσκευή στο δίαυλο απαιτείται και μια ξεχωριστή γραμμή:

- SS (Slave Select) ή CS (Chip Select), μέσω της οποίας η κύρια συσκευή ενεργοποιεί ή απενεργοποιεί την υποτελή. Χαμηλή στάθμη στην ακίδα CS μιας υποτελούς συσκευής οδηγεί σε ενεργοποίηση της επικοινωνίας της με την κύρια συσκευή και αντίστροφα.



Εικόνα 44: Σύνδεση στο δίαυλο SPI


Στον Arduino Uno οι γραμμές του διαύλου SPI διανέμονται ως εξής [4]:

- MOSI στην ψηφιακή ακίδα 11
- MISO στην ψηφιακή ακίδα 12
- SCK στην ψηφιακή ακίδα 13
- SS (ή CS) στην ψηφιακή ακίδα 10

Η βιβλιοθήκη *SPI*, που επιτρέπει την επικοινωνία του Arduino με SPI συσκευές, εγκαθίσταται εξ ορισμού με την εγκατάσταση του Arduino IDE. Όπως ήδη αναφέραμε χαμηλή στάθμη στην ακίδα 10 (CS) του Arduino Uno θα τον ενεργοποιούσε ως υποτελή. Αυτό θα καθιστούσε τη βιβλιοθήκη


κη *SPI* αδρανής, αφού υποστηρίζει μόνο λειτουργία του Arduino ως κύρια (master) συσκευή. Για το λόγο αυτό η ψηφιακή ακίδα 10 του Arduino πρέπει πάντα, όταν χρησιμοποιούμε τη βιβλιοθήκη *SPI*, να χαρακτηρίζεται ως έξοδος (OUTPUT). Υπάρχει βέβαια η δυνατότητα να χρησιμοποιήσουμε και οποιαδήποτε άλλη ακίδα για την ενεργοποίηση μιας υποτελούς συσκευής.

Εξ ορισμού επίσης εγκαθίσταται και η βιβλιοθήκη *SD*, η οποία με χρήση του πρωτοκόλλου *SPI*, χρησιμοποιείται για την επικοινωνία του Arduino με μια μονάδα κάρτας μνήμης SD ή micro SD. Η βιβλιοθήκη *SD* αποτελεί ένα ανωτέρου επιπέδου κέλυφος για τη βιβλιοθήκη *SDFat*, η οποία παρέχει πρόσβαση ανάγνωσης/εγγραφής σε συστήματα αρχείων FAT16/FAT32 για κάρτες μνήμης SD/SDHC. Η κάρτα που θα χρησιμοποιήσουμε θα πρέπει προηγουμένως να έχει διαμορφωθεί, κατά προτίμηση για σύστημα αρχείων FAT16. Η διαμόρφωση της κάρτας SD μπορεί να γίνει για παράδειγμα με ένα ηλεκτρονικό υπολογιστή που είναι εφοδιασμένος με συσκευή ανάγνωσης/εγγραφής καρτών SD.




Εικόνα 45 : Αριστερά η μονάδα SD Card και δεξιά η μονάδα micro SD Card

Οι συνδέσεις που απαιτούνται για την επικοινωνία της μονάδας micro SD ή SD Card με τον Arduino φαίνονται στα επόμενα δύο σχήματα:



Εικόνα 46: Σύνδεση μονάδας κάρτας micro SD στον Arduino



Εικόνα 47 : Σύνδεση της μονάδας SD Card στον Arduino

Ένα πρώτο δοκιμαστικό sketch με τη μονάδα κάρτας SD ή micro SD, που απλά ελέγχει αν υπάρχει κάρτα SD στη μονάδα, έχει ως εξής:

```

/*
  Ex.23 : Πρώτο δοκιμαστικό sketch με τη μονάδα κάρτας SD ή micro SD
*/
#include <SD.h>           // Συμπερίληψη βιβλιοθήκης SD
#include <SPI.h>          // Συμπερίληψη βιβλιοθήκης SPI

void setup()
{
  Serial.begin(9600);     // Ενεργοποίηση σειριακής επικοινωνίας
  Serial.println("Initializing SD card...");
  Serial.println();

  pinMode(SS, OUTPUT);   // Σύμφωνα με την τεκμηρίωση η ακίδα 10 (SS) πρέπει
                          // να οριστεί ως έξοδος, ακόμη κι αν δε χρησιμοποιείται

  if (SD.begin())        // Έλεγχος αν υπάρχει SD κάρτα που μπορεί να αρχικοποιηθεί
  {
    Serial.println("SD initilized.");
  }
  else
  {
    Serial.println("SD did not initialize.");
  }
}

```

```
void loop()
{
    // Τίποτα δε γίνεται εδώ. Όλα έγιναν στη συνάρτηση setup()
}
```

Η συνάρτηση *SD.begin()* προετοιμάζει για χρήση τόσο το δίαυλο SPI όσο και την κάρτα SD, και επιστρέφει λογικό 1 (*true*) αν η προετοιμασία ήταν επιτυχής, και λογικό 0 (*false*) σε περίπτωση αποτυχίας. Στην περίπτωση που χρησιμοποιείται άλλη ψηφιακή ακίδα του Arduino (π.χ. η 4) ως ακίδα επιλογής (CS) μιας συσκευής SPI, αυτό πρέπει να δηλωθεί ως εξής:

```
SD.begin(4);
```

**Παρατήρηση :** Η μονάδα κάρτας SD μέσω της αντίστοιχης ακίδας που διαθέτει μπορεί να τροφοδοτηθεί είτε από τάση 5V, είτε από 3,3V.

Για να ανοίξουμε ένα αρχείο κειμένου στην κάρτα SD για εγγραφή, πρέπει πρώτα να δημιουργήσουμε ένα αντικείμενο κλάσης File, ως εξής:

```
File myfile;
```

και στη συνέχεια να δημιουργήσουμε το αρχείο στην κάρτα SD, ως εξής:

```
myFile = SD.open("datafile.txt", FILE_WRITE);
```

Η πρώτη παράμετρος ("*datafile.txt*") είναι το όνομα του αρχείου, ενώ με τη δεύτερη παράμετρο (FILE\_WRITE) δηλώνουμε ότι έχουμε τις δυνατότητες ανάγνωσης και εγγραφής στο αρχείο. Υπάρχει επιπλέον και η δυνατότητα (με την παράμετρο FILE\_READ) να ανοιχθεί το αρχείο μόνο για ανάγνωση. Στην περίπτωση που το αρχείο υπάρχει ήδη στην κάρτα, τότε με την παραπάνω εντολή το αρχείο ανοίγει για εγγραφή, αντί να δημιουργηθεί ένα νέο.

Την ύπαρξη ενός αρχείου (π.χ. με το όνομα *datafile.txt*) στην κάρτα SD μπορούμε να την ελέγξουμε με την εντολή:

```
SD.exists("datafile.txt");
```

Όταν όλες οι ενέργειες εγγραφής/ανάγνωσης στο αρχείο ολοκληρωθούν, πρέπει να το κλείσουμε με την εντολή:

```
SD.close("datafile.txt");
```

Για την εγγραφή δεδομένων στο αρχείο (*myfile*) που έχουμε ήδη ανοίξει για εγγραφή, μπορούμε να χρησιμοποιήσουμε τις εντολές:

- *myfile.write(data);* // *data* : *char* ή *byte* ή *string*
- *myfile.write(buf, len);* // *buf* : *array of char*, *len* : *αριθμός χαρακτήρων για εγγραφή*
- *myfile.print(data);* // *data* : *char*, *byte*, *string* ή *αριθμός*
- *myfile.println(data);* // *data* : *char*, *byte*, *string* ή *αριθμός*

Για την ανάγνωση δεδομένων από το αρχείο που έχουμε ήδη ανοίξει, μπορούμε να χρησιμοποιήσουμε τις συναρτήσεις:

- *data = myfile.read();* // *data* : *byte* ή *char*
- *line = myfile.readStringUntil(terminator)* // *line* : *string*, *terminator* : *char*

Η συνάρτηση *readStringUntil()* προέρχεται από την κλάση *Stream*, και -στην περίπτωσή μας- επιστρέφει το αλφαριθμητικό που διαβάζει από το αρχείο μέχρι να ανιχνευθεί ο χαρακτήρας σήμανσης τέλους (*terminator*). Συνηθισμένος χαρακτήρας σήμανσης τέλους ο χαρακτήρας *newline*, που δηλώνεται ως '\n'.

Αν θέλουμε να ελέγξουμε αν υπάρχουν και άλλοι χαρακτήρες διαθέσιμοι για διάβασμα από κάποιο αρχείο που έχουμε ήδη ανοίξει, χρησιμοποιούμε τη συνάρτηση *available()*. Για παράδειγμα η εντολή:

```
int count = myfile.available();
```

αποθηκεύει στην αέραια μεταβλητή *count* τον αριθμό των χαρακτήρων στο αρχείο *myfile* που είναι διαθέσιμοι για διάβασμα.

Στο επόμενο sketch, αφού δημιουργήσουμε ένα αρχείο κειμένου στην κάρτα SD, θα εγγράψουμε μερικά δεδομένα σ' αυτό, και τελικά θα διαβάσουμε τα δεδομένα αυτά και θα τα τυπώσουμε στη σειριακή κονσόλα:

```
/*
   Ex.24 : Δημιουργία αρχείου στη μονάδα κάρτας SD
*/
#include <SD.h>           // Συμπερίληψη βιβλιοθήκης SD
#include <SPI.h>          // Συμπερίληψη βιβλιοθήκης SPI

char myfile_name[] = "datafile.txt"; // Το όνομα του αρχείου

void setup()
{
  Serial.begin(9600);      // Ενεργοποίηση σειριακής επικοινωνίας
  pinMode(SS, OUTPUT);    // Η ακίδα 10 (SS) πρέπει να οριστεί ως έξοδος
  Serial.print("Initializing SD card...");
  if (SD.begin())         // Έλεγχος αν υπάρχει SD κάρτα που μπορεί να αρχικοποιηθεί
  {
    Serial.println("SD initiliazed.");

    writeToFile(myfile_name); // Δημιουργία αρχείου και εγγραφή δεδομένων
    Serial.println();

    readFromFile(myfile_name); // Ανάγνωση δεδομένων από το αρχείο
    Serial.println("All done!!!");
  }
  else
  {
    Serial.println("SD did not initialize.");
  }
}

void loop()
{
  // Τίποτα δε γίνεται εδώ. Όλα έγιναν στη συνάρτηση setup()
}

// Δημιουργία αρχείου στην κάρτα SD, εγγραφή δεδομένων και αποθήκευση
void writeToFile(char * filename)
{
  File myfile;           // Δημιουργία αντικειμένου τύπου File
  if (SD.exists(filename)) // Αν ήδη υπάρχει το αρχείο στην κάρτα
```

```

{
    SD.remove(filename);          // διάγραψέ το
    Serial.println("Existing file removed !");
}
// Δημιουργία αρχείου για εγγραφή
Serial.println("Creating a new file...");
myfile = SD.open(filename, FILE_WRITE);
if (myfile)                      // Αν το αρχείο δημιουργήθηκε επιτυχώς
{
    Serial.println("Writing some data to file...");
    for (int i = 1; i < 11; i++)  // Εγγραφή δεδομένων στο αρχείο
    {
        myfile.print("This is Data Line number ");
        myfile.println(i);

        // Εκτύπωση στη σειριακή κονσόλα
        Serial.print("Writing : ");
        Serial.print("This is Data Line number ");
        Serial.println(i);
    }
}
myfile.close();                  // Κλείσιμο του αρχείου
}

// Άνοιγμα αρχείου για διάβαση δεδομένων
void readFromFile(char * filename)
{
    File myfile;                  // Δημιουργία αντικειμένου τύπου File
    String line = "";            // Για την ανάγνωση δεδομένων από το αρχείο
    // Άνοιγμα αρχείου για ανάγνωση
    myfile = SD.open(filename, FILE_READ);
    Serial.println("Reading from file...");
    while (myfile.available() != 0) // Όσο υπάρχουν δεδομένα διαθέσιμα για διάβαση
    {
        line = myfile.readStringUntil('\n'); // διάβασέ τα
        Serial.println(line);                // και τύπωσέ τα στη σειριακή κονσόλα
    }
    myfile.close();                    // Κλείσιμο του αρχείου
    Serial.println();
}

```

## Διακοπές

Οι διακοπές είναι σήματα που διακόπτουν την τρέχουσα δραστηριότητα του μικροελεγκτή. Μπορούν να ενεργοποιηθούν είτε με βάση κάποιο εξωτερικό γεγονός (την αλλαγή για παράδειγμα της κατάστασης σε μια ψηφιακή είσοδο), είτε από κάποιο εσωτερικό γεγονός (π.χ. ένα σήμα από κάποιο χρονιστή). Με την ενεργοποίηση της διακοπής ο επεξεργαστής αφού αποθηκεύσει την κατάσταση στην οποία βρισκόταν τη στιγμή της ενεργοποίησης της διακοπής, διακόπτει την τρέχουσα δραστηριότητά του, εκτελεί μια ρουτίνα (ρουτίνα - συνάρτηση εξυπηρέτησης διακοπής: ISR) συσχετισμένη με τη συγκεκριμένη διακοπή, και αφού ολοκληρώσει την εκτέλεσή της επιστρέφει στην κανονική ροή του προγράμματος.

Το βασικό πλεονέκτημα από τη χρήση διακοπών είναι η ταχύτατη ανταπόκριση του συστήματος σε εξωτερικά ή εσωτερικά σήματα. Εξ ορισμού καμιά άλλη διακοπή δεν ενεργοποιείται όταν ήδη εκτελείται μια ρουτίνα διακοπής, συνεπώς απαιτείται οι ρουτίνες διακοπής να είναι όσο το δυνατό λιγότερο χρονοβόρες. Επίσης υπάρχουν και κάποιοι περιορισμοί σε σχέση με τις ρουτίνες διακοπής, όπως για παράδειγμα ότι δε δέχονται παραμέτρους και δεν επιστρέφουν καμία τιμή, ενώ κάθε μεταβλητή που η τιμή της μεταβάλλεται στη ρουτίνα πρέπει να χαρακτηριστεί ως **πτητική** (volatile). Ο χαρακτηρισμός αυτός αντικατοπτρίζει το γεγονός ότι η μεταβλητή αυτή μπορεί να αλλάξει από εξωτερικούς παράγοντες που δε βρίσκονται στον απόλυτο έλεγχο του προγράμματος, και αποτρέπει τον μεταγλωττιστή (compiler) να αφαιρέσει την αναφορά στη μεταβλητή, αν τυχόν δε χρησιμοποιείται άμεσα στις συναρτήσεις *setup()* και *loop()* του προγράμματος.

Οι διακοπές μπορούν να απενεργοποιηθούν με τις εντολές *noInterrupts()* ή *cli()*, και αντίστοιχα να ενεργοποιηθούν με τις εντολές *interrupts()* ή *sei()*. Εξ' ορισμού είναι ενεργοποιημένες. Όταν ο μικροελεγκτής αρχίζει την εκτέλεση μιας ρουτίνας εξυπηρέτησης διακοπής (ISR) οι διακοπές απενεργοποιούνται, και εκ νέου ενεργοποιούνται κατά την έξοδο από την ISR.

### α. Εξωτερικές διακοπές

Ο Arduino Uno διαθέτει δύο εξωτερικές διακοπές: INT0 και INT1, συσχετισμένες με τις ψηφιακές ακίδες 2 και 3 αντίστοιχα [4]. Δηλαδή κάποιο εξωτερικό σήμα στην ακίδα 2 ενεργοποιεί τη διακοπή INT0, και αντίστοιχα εξωτερικό σήμα στην ακίδα 3 ενεργοποιεί τη διακοπή INT1. Υποστηρίζονται τέσσερις διαφορετικοί τρόποι ενεργοποίησης των διακοπών:


- **LOW** : Ενεργοποίηση οποτεδήποτε η συσχετισμένη ακίδα είναι σε χαμηλή στάθμη.
- **CHANGE** : Ενεργοποίηση οποτεδήποτε η συσχετισμένη ακίδα αλλάζει κατάσταση.
- **RISING** : Ενεργοποίηση οποτεδήποτε η συσχετισμένη ακίδα αλλάζει κατάσταση από χαμηλή σε υψηλή στάθμη.
- **FALLING** : Ενεργοποίηση οποτεδήποτε η συσχετισμένη ακίδα αλλάζει κατάσταση από υψηλή σε χαμηλή στάθμη.

Στον Arduino η διαδικασία ενεργοποίησης/απενεργοποίησης μιας εξωτερικής διακοπής έχει απλοποιηθεί με τις εντολές:

- **attachInterrupt(digitalPinToInterrupt(pin), ISR, mode)**: Καθορίζει τη ρουτίνα που θα εκτελείται όταν η αντίστοιχη διακοπή ενεργοποιηθεί. Δέχεται τρεις παραμέτρους ως εξής:
  - *pin* : η ακίδα (2 ή 3 στον Arduino Uno) ενεργοποίησης της διακοπής.
  - *ISR* : Η ρουτίνα (συνάρτηση) που καλείται όταν ενεργοποιηθεί η διακοπή.
  - *mode* : ο τρόπος ενεργοποίησης της διακοπής (LOW, CHANGE, RISING, FALLING).
- **detachInterrupt(digitalPinToInterrupt(pin))**: Απενεργοποιεί τη διακοπή που έχει συσχετιστεί με την ακίδα *pin*.

## Ένα παράδειγμα χρήσης εξωτερικής διακοπής

Ένας πιεστικός διακόπτης (push button) συνδέεται στον Arduino μέσω μιας pull down αντίστασης των 10kΩ, όπως φαίνεται στο επόμενο σχήμα:



Εικόνα 48: Σύνδεση διακόπτη στην ψηφιακή είσοδο 2 του Arduino

Χρησιμοποιώντας την εξωτερική διακοπή INT0 -συσχετισμένη με την ψηφιακή είσοδο 2 στην οποία συνδέεται ο ένας πόλος του διακόπτη- το πρόγραμμά μας ανιχνεύει τις αλλαγές κατάστασης του διακόπτη και ανάλογα ενεργοποιεί ή απενεργοποιεί το LED που από κατασκευής είναι συνδεδεμένο στην ψηφιακή ακίδα 13 του Arduino.

/\*

*Ex.25 : Παράδειγμα χρήσης εξωτερικής διακοπής*

\*/

```
#define LEDPIN          13
```

```
#define INTERRUPT_PIN  2
```

```
volatile int state = LOW;    // Η πτητική μεταβλητή για καταχώρηση της κατάστασης του διακόπτη  
void setup()
```

```
{  
    pinMode(LEDPIN, OUTPUT);    // Καθορίζουμε την ψηφιακή ακίδα 13 ως έξοδο  
    // Ενεργοποίηση διακοπής στην ακίδα 2 , έλεγχος οποιασδήποτε αλλαγής κατάστασης  
    attachInterrupt(digitalPinToInterrupt(INTERRUPT_PIN), stateChange, CHANGE);  
}
```

```
void stateChange()          // Η ρουτίνα της διακοπής  
{  
    state = !state;        // Αλλαγή τιμής της μεταβλητής state  
    digitalWrite(LEDPIN, state);    // Αλλαγή κατάστασης LED  
}
```

```
void loop()  
{  
    // Δε χρειάζεται καμιά ενέργεια. Όλα γίνονται μέσω της ρουτίνας διακοπής.  
}
```

Ένα ιδιαίτερα σημαντικό πρόβλημα που αντιμετωπίζουμε όταν χρησιμοποιούμε διακόπτες σε ψηφιακά κυκλώματα (όπως είναι ο Arduino) είναι η “αναπήδηση”: όταν ο διακόπτης κλείνει οι δυο μεταλλικές του επαφές ενώνονται και αποσυνδέονται τυπικά 10 έως 100 φορές σε χρονικό διάστημα περί το 1 ms [12]. Δηλαδή κατά το κλείσιμο του διακόπτη δεν παίρνουμε ένα καθαρό τετραγωνικό παλμό (π.χ. μετάβαση από λογικό LOW σε λογικό HIGH), αλλά πολλές διαδοχικές εναλλαγές HIGH - LOW μέχρι την οριστική αποκατάσταση της λογικής μετάβασης. Η ταχύτητα λειτουργίας του Arduino του επιτρέπει να ανταποκριθεί σε αυτές τις ενδιάμεσες εναλλαγές κατάστασης, κάτι που μπορεί να οδηγήσει σε ανεπιθύμητες δυσλειτουργίες του εκτελούμενου κώδικα. Για την αποκατάσταση του προβλήματος η ρουτίνα εξυπηρέτησης διακοπής μπορεί να πάρει τη μορφή:

```
void stateChange()
{
    volatile static unsigned long last_interrupt_time = 0;
    unsigned long interrupt_time = millis();
    if (interrupt_time - last_interrupt_time > 20) //Αγνόησε τις διακοπές για 20 ms
    {
        digitalWrite(LEDPIN, !digitalRead(LEDPIN));
    }
    last_interrupt_time = interrupt_time;
}
```

Ουσιαστικά εδώ ρυθμίζουμε τον κώδικα ώστε να αγνοεί άλλες διακοπές για 20 ms μετά την πρώτη ενεργοποίηση της ρουτίνας εξυπηρέτησης διακοπής. Για τον προσδιορισμό του βέλτιστου χρόνου αγνόησης των διακοπών απαιτείται κάποιος πειραματισμός. Ας σημειώσουμε πως η τύπου *unsigned long* μεταβλητή *last\_interrupt\_time* έχει χαρακτηριστεί ως:

- **volatile** : γιατί η τιμή της αλλάζει μέσα σε ρουτίνα (συνάρτηση) εξυπηρέτησης διακοπής.
- **static** : ώστε να διατηρεί την τιμή της κατά τις διαδοχικές εκτελέσεις της συνάρτησης.

Η συνάρτηση *millis()* επιστρέφει το χρονικό διάστημα (σε milliseconds) που πέρασε από τη στιγμή που άρχισε η εκτέλεση του κώδικα στον Arduino (δηλ. από τη στιγμή που τροφοδοτήθηκε με ρεύμα ο Arduino ή έγινε επανεκκίνηση του συστήματος).

### β. Διακοπές χρονιστή (Timer Interrupts)


Οι διακοπές χρονιστή επιτρέπουν την εκτέλεση μιας συγκεκριμένης εργασίας σε επακριβώς καθορισμένα χρονικά διαστήματα, ανεξάρτητα από οτιδήποτε άλλο συμβαίνει στον κώδικά μας. Καθώς η διαδικασία ενεργοποίησης μιας διακοπής χρονιστή είναι μάλλον πολύπλοκη για τον αρχάριο χρήστη, θα δείξουμε πως μπορούμε να εκμεταλλευτούμε αυτή τη δυνατότητα του Arduino με τη χρήση της εξειδικευμένης βιβλιοθήκης **FlexiTimer2**, που κάνει χρήση του χρονιστή #2 (Timer2) του μικροελεγκτή, και μπορεί να εγκατασταθεί μέσω του “*Διαχειριστή βιβλιοθήκης*”, κατά τα γνωστά. Η βιβλιοθήκη *FlexiTimer2* έχει τρεις συναρτήσεις μέσω των οποίων υλοποιούνται οι διακοπές (Interrupts) για το χρονιστή #2:

- **set(interval, T2ISR)**: Θέτει το χρονικό διάστημα (*interval* σε ms) μεταξύ δύο διαδοχικών ενεργοποιήσεων της διακοπής, καθώς και τη ρουτίνα (*T2ISR*) που θα εκτελείται κατά την ενεργοποίηση της διακοπής.
- **start()**: Ενεργοποίηση των διακοπών του χρονιστή #2.
- **stop()**: Απενεργοποίηση των διακοπών.

### Παράδειγμα διακοπής χρονιστή

Ας δούμε πως όλα αυτά μπορούν να υλοποιηθούν στην περίπτωση ενός απλού συστήματος μέτρησης μαγνητικού πεδίου. Θα χρησιμοποιήσουμε ένα αισθητήρα μαγνητικού πεδίου (SS49E της


Honeywell) για τη μέτρηση του πεδίου ενός μαγνήτη. Η λειτουργία του αισθητήρα, που διατίθεται με τη μορφή ολοκληρωμένου κυκλώματος τριών ακίδων, βασίζεται στο φαινόμενο Hall.



Εικόνα 49: Ο αναλογικός αισθητήρας μαγνητικού πεδίου

Ο κατασκευαστής του αισθητήρα δίνει τα ακόλουθα στοιχεία [13]: Απουσία μαγνητικού πεδίου η τάση στην έξοδο (ακίδα analog OUT) είναι ίση με το μισό της τάσης τροφοδοσίας (τυπικά 2,5 V θεωρώντας την τάση τροφοδοσίας 5V). Η παρουσία νότιου πόλου κάθετα στην επιφάνεια του αισθητήρα αυξάνει την τάση στην έξοδο από την τάση ηρεμίας μέχρι περίπου τα 4V με τυπικό ρυθμό 1,4 mV/Gauss. Αντίστοιχα η παρουσία βόρειου πόλου προκαλεί αναλογική μείωση (μέχρι την τιμή 1V) της τάσης εξόδου του αισθητήρα. Για τη μετατροπή της τιμής που παίρνουμε από τον αναλογικο-ψηφικό μετατροπέα του Arduino σε τιμή μαγνητικού πεδίου (σε Gauss), πρέπει να σκεφτούμε πως: τα 1024 αναλογικά βήματα του ADC αντιστοιχούν σε τάση 5000 mV και 1 Gauss αντιστοιχεί σε 1,4 mV. Οπότε κάθε αναλογικό βήμα του ADC αντιστοιχεί σε 3,4877 Gauss.

Η σύνδεσή του αισθητήρα στον Arduino είναι εξαιρετικά απλή όπως φαίνεται και από το επόμενο σχήμα.



Εικόνα 50: Σύνδεση αισθητήρα μαγνητικού πεδίου στον Arduino

Θέλουμε να παίρνουμε μετρήσεις του μαγνητικού πεδίου σε συγκεκριμένα χρονικά διαστήματα, ας πούμε 2 μετρήσεις κάθε δευτερόλεπτο, δηλ. να καθορίσουμε ένα σταθερό ρυθμό δειγματοληψίας 2 Hz. Παρότι κάτι τέτοιο θα μπορούσε να επιτευχθεί με χρήση της κατάλληλης καθυστέρησης μεταξύ των διαδοχικών μετρήσεων, αυτή η λύση δεν είναι η ενδεδειγμένη αφού με τη συνάρτηση `delay()` ο μικροελεγκτής σταματά κάθε δραστηριότητα (εκτός των διακοπών) για το καθορισμένο χρονικό διάστημα, δηλ. έχουμε απώλεια σημαντικού χρόνου που θα μπορούσε να χρησιμοποιηθεί από το πρόγραμμά μας για άλλες εργασίες. Η χρήση διακοπών χρονιστή είναι η λύση στο

πρόβλημά μας. Το sketch που ακολουθεί δύο φορές κάθε δευτερόλεπτο επιστρέφει θετικές τιμές για την ένταση του μαγνητικού πεδίου όταν νότιος μαγνητικός πόλος προσεγγίζει την πρισματική επιφάνεια του αισθητήρα, και αρνητικές τιμές όταν αντίστοιχα προσεγγίζει βόρειος μαγνητικός πόλος:

```

/*
   Ex.26 : Παράδειγμα διακοπής που ενεργοποιείται από το χρονοστή #2
*/

#include "FlexiTimer2.h"           // Συμπερίληψη βιβλιοθήκης FlexiTimer2

// Η αναλογική είσοδος στην οποία συνδέεται ο αισθητήρας μαγνητικού πεδίου
#define HALLSENS_PIN A0

// Η πειραματική τιμή που επιστρέφει ο A/D μετατροπέας με τον αισθητήρα εκτός μαγνητικού πεδίου.
#define ADCZERO 505L

// Συντελεστής μετατροπής της τιμής του A/D μετατροπέα σε Gauss
#define ADCVTOGAUSS 3.4877

// Πτητική μεταβλητή που όταν γίνεται true δηλώνει ότι είναι η κατάλληλη στιγμή
// για αποστολή των δεδομένων στη σειριακή κονσόλα
volatile boolean intEnabled = false;

// Η ρουτίνα διαχείρισης της διακοπής του χρονοστή #2.
// Όταν εκτελείται απλά θέτει την τιμή της μεταβλητής intEnabled σε true
void runADC()
{
    intEnabled = true;
}

void setup()
{
    Serial.begin(115200);           //Ενεργοποίηση σειριακής επικοινωνίας
    FlexiTimer2::set(500, runADC); // Η ρουτίνα runADC() θα εκτελείται κάθε 500 ms
                                    // δηλ. ο ρυθμός δειγματοληψίας είναι 2 Hz.
}

void loop()
{
    Serial.print("Send a key to start...");
    while (Serial.available() == 0); // Αναμονή μέχρι τη σειριακή λήψη κάποιου χαρακτήρα
    FlexiTimer2::start();           // Ενεργοποίηση της διακοπής του χρονοστή #2
    do                               // Βρόχος do ... while
    {
        float gauss = (analogRead(HALLSENS_PIN) - ADCZERO) * ADCVTOGAUSS;
        if (intEnabled)
        {

```

```

        Serial.print(gauss, 2);
        Serial.println(" Gauss ");
        intEnabled = false;
    }
}
while (Serial.read() != 'S');           // Επανάληψη μέχρι να ληφθεί ο χαρακτήρας 'S'
FlexiTimer2::stop();                   // Απενεργοποίηση της διακοπής χρονιστή
}

```

Στη συνάρτηση *setup()* ενεργοποιείται αρχικά η σειριακή επικοινωνία με ρυθμό 115200bps, και μετά καθορίζεται η συνάρτηση που θα εκτελείται καθώς και ο ρυθμός με τον οποίο θα εκτελείται κάθε φορά που ενεργοποιείται η διακοπή του χρονιστή.

Στη συνάρτηση *loop()* αρχικά υπάρχει ένας βρόχος αναμονής μέχρι να ληφθεί σειριακά κάποιος χαρακτήρας. Όταν ο χαρακτήρας ληφθεί, ενεργοποιείται η διακοπή του χρονιστή #2 και ο κώδικας μπαίνει σε ένα *do ... while* βρόχο μέχρι να ληφθεί σειριακά ο χαρακτήρας 'S'. Στο βρόχο αυτό γίνεται διαρκώς μετατροπή του αναλογικού σήματος από τον αισθητήρα σε ψηφιακό, και εν συνεχεία μετατροπή του σε μονάδες έντασης μαγνητικού πεδίου. Όταν η πτητική μεταβλητή *intEnabled* γίνει true (δηλ. κάθε 0,5 s ή 500 ms) το αποτέλεσμα των μετατροπών τυπώνεται στη σειριακή κονσόλα, και η μεταβλητή *intEnabled* γίνεται false. Με την έξοδο από το βρόχο η διακοπή του χρονιστή #2 απενεργοποιείται, και ξεκινάει από την αρχή η εκτέλεση του κώδικα της συνάρτησης *loop()*.

Η τιμή (ADCZERO) που επιστρέφει ο αισθητήρας μαγνητικού πεδίου όταν βρεθεί εκτός μαγνητικού πεδίου (αλλά στην πραγματικότητα εντός του ασθενούς γήινου μαγνητικού πεδίου), μπορεί να προσδιοριστεί με το επόμενο sketch:

```

/*
   Ex.27 : Ρύθμιση για αισθητήρα μαγνητικού πεδίου SS49E
*/

#define HALLSENS_PIN A0

void setup()
{
    Serial.begin(9600);
}

void loop()
{
    Serial.print ("ADC Value: ");
    Serial.println(analogRead(HALLSENS_PIN));
    delay(1000);
}


```

**Παρατήρηση :** Στη θέση του αισθητήρα SS49E μπορούν να χρησιμοποιηθούν και άλλοι αναλογικοί αισθητήρες Hall, με διαφορετικούς όμως συντελεστές μετατροπής της τιμής που επιστρέφει ο αναλογικο-ψηφιακός μετατροπέας σε Gauss, όπως: ο A1302 με συντελεστή μετατροπής 3,756 Gauss/βήμα ή ο A1301 με συντελεστή μετατροπής 1,953 Gauss/βήμα.

## Πηγαίνοντας λίγο μακρύτερα...


### Ο ηλεκτρονόμος ή ρελέ (relay)

Πρόκειται για μια ηλεκτρομηχανική διάταξη που λειτουργεί ως διακόπτης ελεγχόμενος από τάση.



Εικόνα 51: Ο ηλεκτρονόμος ή ρελέ (relay)


Στη συνήθη SPDT (Single Pole-Dual Throw) μορφή του πρόκειται για ένα μεταγωγό διακόπτη (αποτελούμενο από τις επαφές POLE, NC, NO) ενεργοποιούμενο με τη βοήθεια ενός ηλεκτρομαγνήτη. Όταν ο ηλεκτρομαγνήτης δε διαρρέεται από ρεύμα είναι βραχυκυκλωμένες οι επαφές POLE (ή COMMON) και NC (Normal Closed) του διακόπτη, ενώ όταν ο ηλεκτρομαγνήτης διαρρέεται από ρεύμα διακόπτεται η επαφή POLE - NC, και βραχυκυκλώνονται οι επαφές POLE και NO (Normal Opened). Ουσιαστικά με το ρελέ η ακίδα POLE (ή COMMON) μπορεί να συνδέεται είτε στην ακίδα NC (όταν το ρελέ είναι απενεργοποιημένο), είτε στην ακίδα NO (όταν το ρελέ είναι ενεργοποιημένο). Αφήνοντας την επαφή NC χωρίς σύνδεση, μπορούμε να συνδέσουμε το ρελέ ως απλό (on-off) διακόπτη σε οποιοδήποτε ηλεκτρικό κύκλωμα, που μπορεί μάλιστα να λειτουργήσει σε μια ευρεία περιοχή τάσεων (μέχρι 250 V AC) και ρευμάτων (μέχρι και 10 A)<sup>1</sup>.



Εικόνα 52: Η μονάδα ρελέ


<sup>1</sup> Οι αναγραφόμενες τιμές αφορούν ρελέ που συνήθως χρησιμοποιούνται σε Arduino projects.

Στη μονάδα (module) ρελέ που πρόκειται να χρησιμοποιήσουμε η ενεργοποίηση του ηλεκτρομαγνήτη γίνεται μέσω μιας ακίδας του (Signal ή Input), που συνδέεται σε μια ψηφιακή έξοδο του Arduino. Λογικό 1 στην ακίδα Signal (ή Input) ενεργοποιεί τον ηλεκτρομαγνήτη, και λογικό μηδέν τον απενεργοποιεί. Στην εικόνα 53 φαίνεται το κυκλωματικό διάγραμμα της μονάδας του ρελέ:



Εικόνα 53: Το κυκλωματικό διάγραμμα της μονάδας ρελέ


Η σύνδεση του ρελέ ως διακόπτη σε ένα σύστημα που περιλαμβάνει σε σειρά μια αντίσταση (τιμής από 220-1000 Ω), ένα LED, και ένα σύστημα μπαταριών συνολικής τάσης 3V (ή μια πλακέ μπαταρία των 4,5V) φαίνεται στην Εικόνα 54. Ενεργοποίηση του ρελέ έχει ως αποτέλεσμα το άναμμα του LED, ενώ με την απενεργοποίηση του ρελέ το LED σβήνει.



fritzing

Εικόνα 54: Το ρελέ ως διακόπτης

Για την ενεργοποίηση του ρελέ μπορούμε να χρησιμοποιήσουμε τον Arduino, συνδέοντας τις ακίδες (+ ή 5v) και (- ή Gnd) της μονάδας του ρελέ με τις αντίστοιχες ακίδες του Arduino, και την ακίδα (S ή Signal ή Input) της μονάδας ρελέ σε μια ψηφιακή ακίδα (π.χ. την 7) του Arduino. Για την απλοποίηση του κυκλώματος δε θα χρησιμοποιήσουμε εξωτερική μπαταρία για την τροφοδοσία του LED, αλλά θα το τροφοδοτήσουμε από τις ακίδες 5V και GND της πλακέτας του Arduino.



fritzing

Εικόνα 55: Σύνδεση της μονάδας ρελέ στον Arduino


Το σχετικό sketch για τη δοκιμή του ρελέ μπορεί να έχει τη μορφή:

```
/*  
  Ex.28 : Αναβοσβήνουμε ένα LED με τη βοήθεια του ρελέ  
*/  
  
#define RLPIN 7  
  
void setup ()  
{  
  pinMode (RLPIN, OUTPUT);    // Η ακίδα RLPIN καθορίζεται ως ψηφιακή έξοδος  
}  
  
void loop ()  
{  
  digitalWrite (RLPIN, HIGH); // Ενεργοποίηση το relay  
  delay (1000);               // Αναμονή 1sec (1000 milliseconds)  
  digitalWrite (RLPIN, LOW);  // Απενεργοποίηση του relay  
  delay (1000);               // Αναμονή 1sec  
}
```

Κάποιος παρατηρητικός θα έχει ήδη καταλάβει πως το προηγούμενο sketch λειτουργικά είναι ακριβώς ίδιο με το δεύτερο παράδειγμα του βιβλίου (σελ. 28), με το οποίο αναβοσβήναμε ένα LED που είχε συνδεθεί σε κάποια ψηφιακή ακίδα (π.χ την ακίδα 7) του Arduino. Μπορούμε, δηλαδή, να πούμε πως η παρούσα εκδοχή κάνει κακή χρήση πόρων, αφού το ίδιο αποτέλεσμα μπορεί να επιτευχθεί με πολύ απλούστερο τρόπο, και συνεπώς μόνο διδακτική αξία μπορεί να έχει για την επίδειξη της λειτουργίας του ρελέ.

### Ένας απλός αυτοματισμός με ηλεκτρονόμο

Συνδυάζοντας το κύκλωμα του ρελέ για το αναβόσβημα του LED με μια φωτοαντίσταση, εύκολα δημιουργούμε το κύκλωμα ενός απλού αυτοματισμού, με τον οποίο το ρελέ ενεργοποιείται ή απενεργοποιείται (και συνεπώς ανάβει ή σβήνει το LED) ανάλογα με την ένταση του φωτός που προσπίπτει στη φωτοαντίσταση. Οι συνδέσεις με τον Arduino φαίνονται στο ακόλουθο σχήμα:




Εικόνα 56: Σύνδεση ρελέ, LED και φωτοαντίστασης στον Arduino

Το σχετικό sketch έχει τη μορφή:

```
/*  
  Ex.29 : Απλός αυτοματισμός  
*/  
  
void setup ()  
{  
  pinMode (7, OUTPUT);           // Η ακίδα 7 καθορίζεται ως ψηφιακή έξοδος  
}  
  
void loop ()  
{  
  int value = analogRead(A0);    // "Διαβάζουμε" την ένταση του φωτός  
  delay(500);                    // Μικρή αναμονή πριν τη λήψη απόφασης  
  if (value < 400)              // Αν είναι σκοτάδι (η τιμή 400 προέκυψε από δοκιμές)  
  {  
    digitalWrite (7, HIGH);     // Ενεργοποίηση το relay και άναμμα του led  
  }  
  else                          // Αλλιώς  
  {  
    digitalWrite (7, LOW);     // Απενεργοποίηση του relay και σβήσιμο του led  
  }  
}
```

Στο βρόχο *loop()*, πρώτα “διαβάζουμε” από την αναλογική είσοδο A0 την τρέχουσα τιμή σχετικά με τη φωτεινότητα, και μετά από 0,5 sec ανάλογα με την τιμή της ενεργοποιούμε ή απενεργοποιούμε το ρελέ, ανάβοντας ή σβήνοντας αντίστοιχα το LED. Η τιμή από την αναλογική είσοδο A0 στην οποία θα γίνεται η αλλαγή κατάστασης του ρελέ (400 στην περίπτωση του sketch) πρέπει να προσδιοριστεί πειραματικά (μικρότερες τιμές σημαίνουν μικρότερη φωτεινότητα).

Αντί για LED που αναβοσβήνει ανάλογα με την ένταση του φωτισμού στην φωτοαντίσταση, μπορούμε να χρησιμοποιήσουμε μια λάμπα των 220V, λαμβάνοντας όλες τις απαραίτητες προφυλάξεις, αφού η τάση των 220V είναι επικίνδυνη για την ανθρώπινη ζωή.



Εικόνα 57: Κύκλωμα απλού αυτοματισμού με λάμπα 220 V

Θα κάνουμε και κάποιες βελτιώσεις στον κώδικα, αφού: Αν έχει πέσει σκοτάδι και η λάμπα έχει

ήδη ανάψει (δηλ. έχει ενεργοποιηθεί το ρελέ), δε χρειάζεται να ξανα-ενεργοποιηθεί το ρελέ για όσο χρόνο είναι ακόμη σκοτάδι. Αντίστοιχα δεν υπάρχει λόγος να απενεργοποιήσουμε ένα ήδη απενεργοποιημένο ρελέ όσο είναι ακόμη ημέρα. Στην αντίθετη περίπτωση ο Arduino εκτελεί πολύ μεγάλο αριθμό περιττών ενεργειών. Για να λύσουμε αυτό το πρόβλημα, απαιτείται:

1. Να γνωρίζουμε κάθε στιγμή την κατάσταση της λάμπας. Αυτό το επιτυγχάνουμε με τον ορισμό μιας καθολικής (*global*) λογικής μεταβλητής (δηλ. τύπου *bool* ή *boolean*), που της δώσαμε το όνομα *lampState*, και παίρνει μόνο δύο τιμές: *false* αν η λάμπα είναι σβηστή, και *true* αν είναι αναμμένη. Αρχικά λοιπόν, που η λάμπα είναι σβηστή, η μεταβλητή *lampState* έχει τιμή *false*, ενώ της δίνεται η τιμή *true* (*lampState = true*) όταν το ρελέ ενεργοποιηθεί και ανάψει η λάμπα, και πάλι τη τιμή *false* (*lampState = false*) όταν το ρελέ απενεργοποιηθεί και σβήσει η λάμπα.
2. Να κάνουμε ορισμένους επιπλέον ελέγχους. Το ρελέ πρέπει να ενεργοποιείται (και να ανάβει η λάμπα) μόνο αν έχει πέσει σκοτάδι και η λάμπα είναι σβηστή. Αυτός ο έλεγχος προγραμματιστικά υλοποιείται μέσω της εντολής λήψης απόφασης:

```
if ((value < 400) && (lampState == false))
```

Η εντολή περιλαμβάνει δύο επιμέρους συνθήκες συνδεδεμένες μεταξύ τους με τον λογικό τελεστή **and (&&)**, που σημαίνει ότι για να εκτελεστεί το αντίστοιχο μπλοκ εντολών (για το άναμμα της λάμπας) πρέπει **και** οι δύο συνθήκες να είναι αληθείς (*true*). Η πρώτη συνθήκη (*value < 400*) είναι η συνθήκη που όταν είναι αληθής σημαίνει ότι έχει αρχίσει να πέφτει το σκοτάδι, και η δεύτερη (*lampState == false*) όταν είναι αληθής σημαίνει πως η λάμπα είναι σβηστή. Όταν η συνολική συνθήκη του *if* είναι ψευδής (*false*) εκτελείται το μπλοκ εντολών που ακολουθεί μετά το *else*, και το οποίο περιλαμβάνει ένα δεύτερο *if* με επίσης δύο επιμέρους συνθήκες συνδεδεμένες με λογικό **and**. Η πρώτη συνθήκη (*value >= 400*) είναι αληθής από όταν αρχίζει να ξημερώνει και μετά, και η δεύτερη (*lampState == true*) όταν η λάμπα είναι ήδη αναμμένη. Συνεπώς όταν και οι δύο συνθήκες είναι αληθείς (*true*) εκτελείται το μπλοκ εντολών με το οποίο σβήνει η λάμπα. Σε όλες τις άλλες περιπτώσεις (π.χ. όταν έχει αρχίσει να σκοτεινιάζει αλλά η λάμπα έχει ήδη ανάψει ή όταν έχει ξημερώσει αλλά η λάμπα είναι σβηστή) ο Arduino δεν εκτελεί κάποια συγκεκριμένη ενέργεια, αλλά επαναλαμβάνει τον κύκλο εντολών της συνάρτησης *loop()*. Περισσότερα για τους λογικούς τελεστές στο Παράρτημα Β.

Το σχετικό sketch διαμορφώνεται ως εξής:

```
/*
  Ex.30 : Αυτόματο άναμμα και σβήσιμο λάμπας
*/

bool lampState = false;           // Για να γνωρίζουμε την κατάσταση της λάμπας

void setup ()
{
  pinMode (7, OUTPUT);           // Η ακίδα 7 καθορίζεται ως ψηφιακή έξοδος
}

void loop ()
{
  int value = analogRead(A0);     // "Διαβάζουμε" την ένταση του φωτός

  // Αν έχει πέσει σκοτάδι και η λάμπα είναι σβηστή
  if ((value < 400) && (lampState == false))
  {
    delay(500);                  // Μικρή αναμονή πριν την ενεργοποίηση
```

```

        digitalWrite (7, HIGH);    // Ενεργοποίηση το relay και άναμμα της λάμπας
        lampState = true;        // Η λάμπα είναι αναμμένη
    }
    else if ((value >= 400) && (lampState == true))    // Αλλιώς
    {
        delay(500);                // Μικρή αναμονή πριν την απενεργοποίηση
        digitalWrite (7, LOW);    // Απενεργοποίηση του relay και σβήσιμο της λάμπας
        lampState = false;        // Η λάμπα είναι σβηστή
    }
}


```

Για την ασφαλή και ορθή λειτουργία της διάταξης πρέπει να λάβουμε ιδιαίτερες προφυλάξεις, όπως:

1. Το ρελέ να τοποθετηθεί μέσα σε πλαστικό κουτί, ώστε να αποφευχθεί η κατά λάθος επαφή με τους ακροδέκτες που φέρουν την υψηλή τάση των 220V.
2. Η φωτοαντίσταση πρέπει να τοποθετηθεί σε κατάλληλη θέση, ώστε να "αισθάνεται" τον περιβάλλοντα φωτισμό, αλλά να μην επηρεάζεται από τη φωτοβολία της λάμπας που ανάβει όταν ενεργοποιείται το ρελέ.

## Ο αισθητήρας υπερήχων HC-SR04


Ο αισθητήρας HC-SR04 ενσωματώνει σε μία μονάδα ένα πομπό και ένα δέκτη υπερήχων (συχνότητας 40 kHz), καθώς και τα κατάλληλα κυκλώματα ελέγχου.



Εικόνα 58: Ο αισθητήρας υπερήχων HC-SR04

Η μονάδα του αισθητήρα διαθέτει τέσσερις ακίδες για την τροφοδοσία και την επικοινωνία με τον Arduino:

- VCC που συνδέεται στην ακίδα 5V του Arduino,
- GND που συνδέεται στην αντίστοιχη ακίδα του Arduino,
- Trig (Trigger) και Echo (Receive) που συνδέονται σε δύο διαφορετικές ψηφιακές ακίδες στον Arduino.



Εικόνα 59: Κύκλος λειτουργίας του αισθητήρα HC-SR04

Η λειτουργία του αισθητήρα HC-SR04 συνοψίζεται ως εξής:


1. Ο αισθητήρας ξεκινάει ένα κύκλο μέτρησης αν η ακίδα του Trig τεθεί σε υψηλή λογική στάθμη για χρόνο τουλάχιστον 10 μs.
2. Ο πομπός αυτόματα εκπέμπει 8 μικρού εύρους παλμούς υπερήχων συχνότητας 40 kHz, και αμέσως μετά θέτει την ακίδα Echo σε υψηλή λογική στάθμη. Όταν αυτοί οι παλμοί προσπέσουν σε κάποιο αντικείμενο που βρίσκεται στη γειτονιά του αισθητήρα ανακλώνται και επιστρέφουν πίσω.
3. Η ακίδα Echo παραμένει σε υψηλή λογική στάθμη μέχρι να ανιχνευθεί το ανακλώμενο σήμα από τον δέκτη του αισθητήρα.

Με τη βοήθεια του Arduino μπορούμε κατ' αρχάς να υπολογίσουμε το χρονικό διάστημα που μεσολαβεί ανάμεσα στην εκπομπή και τη λήψη του σήματος (όσο δηλαδή χρόνο η ακίδα Echo βρίσκεται σε υψηλό δυναμικό). Αν γνωρίζουμε την απόσταση ανάμεσα στον αισθητήρα και στην ανακλαστική επιφάνεια μπορούμε να υπολογίσουμε την ταχύτητα διάδοσης του ήχου, ή αντίθετα θεωρώντας γνωστή την ταχύτητα διάδοσης του ήχου (343 m/s στους 20 °C στον ξηρό αέρα) μπορούμε να υπολογίσουμε την απόσταση της ανακλαστικής επιφάνειας από τον αισθητήρα. Οι υπολογισμοί γίνονται με βάση την απλή σχέση:

$$(\text{απόσταση}) = (\text{ταχύτητα}) \times (\text{χρόνος})$$

Στη σχέση αυτή ο χρόνος είναι ίσος με το μισό αυτού που υπολογίζουμε με τον Arduino, αφού ο χρόνος ανάμεσα στην εκπομπή και τη λήψη του σήματος αντιστοιχεί στο χρόνο που απαιτείται για να πάει το σήμα από τον αισθητήρα στην ανακλαστική επιφάνεια και να επιστρέψει. Ο αισθητήρας παρέχει λειτουργίες μέτρησης από 2 cm μέχρι 400 cm, ενώ η ακρίβεια των μετρήσεων μπορεί να φτάσει μέχρι τα 3 mm.

Για τις ανάγκες των δοκιμών μας θα συνδέσουμε τις ακίδες Trig και Echo του αισθητήρα στις ψηφιακές ακίδες 11 και 12 του Arduino αντίστοιχα, όπως φαίνεται στην επόμενη εικόνα.



Εικόνα 60: Σύνδεση του αισθητήρα υπερήχων στον Arduino

Θα χρησιμοποιήσουμε τον αισθητήρα υπερήχων για να προσδιορίσουμε την ταχύτητα του ήχου στον αέρα, και στις συνθήκες που επικρατούν κατά τη διάρκεια του πειράματος. Για το σκοπό αυτό θα τοποθετήσουμε μια ανακλαστική επιφάνεια μπροστά από τον αισθητήρα και σε κάποια απόσταση (50 – 100 cm), την οποία θα μετρήσουμε με ακρίβεια με τη βοήθεια μετροταινίας. Προσοχή πρέπει να δοθεί, ώστε να μην υπάρχουν άλλα αντικείμενα στη γειτονιά του αισθητήρα προς αποφυγή ανεπιθύμητων ανακλάσεων από αυτά, καθώς ο αισθητήρας παρουσιάζει ένα εύρος πεδίου περί τις 15°.

Όμως, για να εκμεταλλευτούμε με τον καλύτερο δυνατό τρόπο τις δυνατότητες του αισθητήρα στον Arduino θα πρέπει κατ' αρχάς να εγκαταστήσουμε την κατάλληλη βιβλιοθήκη λογισμικού. Με τη βοήθεια του “*Διαχειριστή βιβλιοθήκης*” (που εκτελείται μέσω του μενού “*Εργαλεία\Διαχείριση βιβλιοθηκών*” ή του “*Σχέδιο\Συμπερίληψη βιβλιοθήκης\Διαχείριση βιβλιοθηκών*”) αναζητούμε και εγκαθιστούμε τη βιβλιοθήκη “*NewPing*”, που περιλαμβάνει όλο τον απαραίτητο κώδικα για τη διαχείριση του αισθητήρα. Για παράδειγμα:

- Η συνάρτηση *ping()* επιστρέφει το χρονικό διάστημα από την εκπομπή μέχρι τη λήψη του ανακλώμενου σήματος.
- Η συνάρτηση *ping\_cm()* επιστρέφει την απόσταση (σε cm) ανάμεσα στον αισθητήρα και την ανακλαστική επιφάνεια.

Αφού δημιουργήσουμε ένα νέο sketch, πρέπει να δηλώσουμε την συμπερίληψη του κώδικα από τη βιβλιοθήκη. Αυτό μπορεί να γίνει είτε αυτόματα (μέσω του μενού “*Σχέδιο\Συμπερίληψη βιβλιοθήκης*”), είτε χειροκίνητα, εισάγοντας αμέσως μετά τα αρχικά σχόλια τη σχετική δήλωση συμπερίληψης:

```
#include <NewPing.h>
```

Στη συνέχεια πρέπει να δημιουργήσουμε ένα (προγραμματιστικό) αντικείμενο κλάσης *NewPing*, μέσω του οποίου θα αποκτήσουμε πρόσβαση στον κώδικα και τα δεδομένα της βιβλιοθήκης, ενώ για τη ρύθμιση των αρχικών παραμέτρων του αντικειμένου απαιτείται να δηλώσουμε τις ψηφιακές ακίδες του Arduino που χρησιμοποιούνται για τη σύνδεση του αισθητήρα:

```
#define TRIGGER_PIN 11 // Ακίδα του Arduino που συνδέεται στην ακίδα Trigger του αισθητήρα
#define ECHO_PIN 12 // Ακίδα του Arduino που συνδέεται στην ακίδα Echo του αισθητήρα

NewPing us_sensor(TRIGGER_PIN, ECHO_PIN); // Δημιουργία-ρύθμιση αντικειμένου NewPing
```

Το sketch τελικά αποκτά τη μορφή:

```
/*
   Ex.31 : Παράδειγμα χρήσης του αισθητήρα υπερήχων
*/

#include <NewPing.h> // Συμπερίληψη βιβλιοθήκης NewPing

#define TRIGGER_PIN 11 // Ακίδα του Arduino που συνδέεται στην ακίδα Trigger του αισθητήρα
#define ECHO_PIN 12 // Ακίδα του Arduino που συνδέεται στην ακίδα Echo του αισθητήρα

NewPing us_sensor(TRIGGER_PIN, ECHO_PIN); // Δημιουργία-ρύθμιση αντικειμένου NewPing

void setup()
{
    Serial.begin(9600); // Ενεργοποίηση σειριακής επικοινωνίας στα 9600 bps
}

void loop()
{
    Serial.print("Time: ");
    Serial.print(us_sensor.ping()); // Χρόνος σε μs
    Serial.println(" us");

    delay(1000); // Αναμονή 1s
}
```

Για να προσδιορίσουμε την ταχύτητα του ήχου με βάση τις τιμές χρόνου που επιστρέφει ο Arduino, μπορούμε να εργαστούμε ως εξής:

1. Υπολογίζουμε το μέσο όρο από πέντε τουλάχιστον διαδοχικές μετρήσεις που επιστρέφει στη σειριακή κονσόλα ο Arduino, και διαιρούμε αυτή την τιμή διά 2, αφού όπως έχουμε ήδη πει ο χρόνος που επιστρέφει ο Arduino είναι αυτός που απαιτείται για να πάει ο ήχος (υπέρηχος για να είμαστε ακριβείς) από τον αισθητήρα στην ανακλαστική επιφάνεια και να επιστρέψει.
2. Με το χρόνο που υπολογίσαμε στο προηγούμενο βήμα, και τη μετρημένη απόσταση ανάμεσα στον αισθητήρα και την ανακλαστική επιφάνεια, υπολογίζουμε την ταχύτητα του ήχου με την εξίσωση:

$$\text{ταχύτητα} = \frac{\text{απόσταση}}{\text{χρόνος}}$$

Βέβαια ο αισθητήρας μπορεί να χρησιμοποιηθεί και για τον υπολογισμό της απόστασης ανάμεσα στον αισθητήρα και κάποιο αντικείμενο. Το σχετικό sketch, μπορεί να έχει τη μορφή:

```
/*
   Ex.32 : Μέτρηση απόστασης με τον αισθητήρα υπερήχων
*/

#include <NewPing.h>      // Συμπερίληψη βιβλιοθήκης NewPing

#define TRIGGER_PIN 11    // Ακίδα του Arduino συνδεδεμένη στην ακίδα Trigger του αισθητήρα
#define ECHO_PIN 12      // Ακίδα του Arduino συνδεδεμένη στην ακίδα Echo του αισθητήρα

NewPing us_sensor(TRIGGER_PIN, ECHO_PIN); // Δημιουργία-ρύθμιση αντικειμένου NewPing

void setup()
{
    Serial.begin(9600);    // Ενεργοποίηση σειριακής επικοινωνίας στα 9600 bps
}

void loop()
{
    Serial.print("Distance: ");
    Serial.print(us_sensor.ping_cm()); // Απόσταση σε cm
    Serial.println(" cm");

    delay(1000);          // Αναμονή 1s
}
```

Ένας σημαντικός παράγοντας σφαλμάτων κατά τη μέτρηση αποστάσεων με τον αισθητήρα υπερήχων είναι το γεγονός πως η ταχύτητα του ήχου στον αέρα δεν είναι σταθερή, αλλά εξαρτάται και από τη θερμοκρασία, σύμφωνα με την εξίσωση [14]:

$$v_{\eta\chi} = 20,05\sqrt{\theta + 273,15} \text{ m/s } (\theta \text{ σε } ^\circ\text{C})$$

Η μέτρηση της θερμοκρασίας με κατάλληλο αισθητήρα (π.χ. τον αναλογικού τύπου LM35), μπορεί να οδηγήσει σε ακριβέστερους υπολογισμούς για την ταχύτητα του ήχου, και συνεπώς σε ακριβέστερες τιμές για τις αποστάσεις που μετράμε.



```

    {
        myFile.println("Time (s)\tTemperature(°C) "); // Εγγραφή επικεφαλίδας

        Serial.println("Data Logger started...\n");
        Serial.println("Send a key to stop...\n");

        // Η μεταβλητή startTime παίρνει τιμή ίση με το χρόνο (σε msecs )
        // που πέρασε από τη στιγμή που άρχισε η εκτέλεση του sketch
        startTime = millis();
    }
}
else // Αν η κάρτα SD δεν υπάρχει ή δε μπορεί να αρχικοποιηθεί
{
    Serial.println("No SD Card found!!!");
    while(1); // Ατέρμων βρόχος
}
}

void loop()
{
    long tmval = (millis() - startTime)/1000; // Η χρονική στιγμή λήψης της μέτρησης

    //Λήψη της θερμοκρασίας από τον αισθητήρα LM35
    float val = ((5.0 * analogRead(A0)) / 1024.0) * 100.0;

    // Με το επόμενο μπλοκ εντολών τα αποτελέσματα (χρόνος και θερμοκρασία)
    // εγγράφονται σε μια γραμμή στο αρχείο στην κάρτα SD
    myFile.print(tmval); // Εγγράφει τη χρονική στιγμή
    myFile.print("\t"); // Εγγράφει το χαρακτήρα ελέγχου tab
    myFile.println(val); // Εγγράφει τη θερμοκρασία

    if (Serial.available() > 0) // Αν έχει αποσταλεί ένας χαρακτήρας σειριακά
    {
        myFile.close(); // Κλείσιμο αρχείου
        readFromFile(filename); // Εκτύπωση των δεδομένων από το αρχείο
        while(1); // Ατέρμων βρόχος
    }
    delay(1000); // Αναμονή 1sec για τη λήψη της επόμενης μέτρησης
}

void readFromFile(char * filename) // Άνοιγμα αρχείου για διάβασμα δεδομένων
{
    File myfile; // Δημιουργία αντικειμένου τύπου File
    String line = ""; // Για την αποθήκευση δεδομένων από το αρχείο

    // Άνοιγμα αρχείου για ανάγνωση
    myfile = SD.open(filename, FILE_READ);
    Serial.println("Reading from file...");
    while (myfile.available() != 0) // Αν υπάρχουν διαθέσιμα δεδομένα
    {
        line = myfile.readStringUntil('\n'); // διάβασέ τα
    }
}

```

```

        Serial.println(line);                // και τύπωσέ τα στη σειριακή κονσόλα
    }
    myfile.close();                          // Κλείσιμο του αρχείου
    Serial.println();
}

```

Σύμφωνα με το φυλλάδιο δεδομένων του αισθητήρα LM35, η τάση στην ακίδα εξόδου του είναι ευθέως ανάλογη της θερμοκρασίας [15], ισχύει δηλαδή:

$$V_{out} = \lambda \cdot \theta$$

όπου  $V_{out}$  η τάση στην έξοδο σε mV,  $\theta$  η θερμοκρασία σε °C, και  $\lambda$  η σταθερά αναλογίας με τιμή 10 mV/°C. Συνεπώς για να υπολογίσουμε τη θερμοκρασία πρέπει:

- Να μετατρέψουμε την τιμή που επιστρέφει ο μετατροπέας αναλογικού σε ψηφιακό του Arduino σε τάση (mV), ως εξής:  $Voltage = \left( \frac{5.0}{1024.0} \right) \times (analogRead(A0)) \times 1000.0$ .
- Στη συνέχεια να διαιρέσουμε την τάση εξόδου με τη σταθερά αναλογίας  $\lambda$ , ώστε να υπολογίσουμε τη θερμοκρασία:  $\theta(^{\circ}C) = \frac{Voltage}{\lambda} = \left( \frac{5.0}{1024.0} \right) \times (analogRead(A0)) \times 100.0$ .

Έτσι εξηγείται η εντολή:  $float val = ((5.0 * analogRead(A0)) / 1024.0) * 100.0$  που χρησιμοποιήσαμε στο sketch για τον υπολογισμό της τιμής της θερμοκρασίας. Επιπλέον γίνεται φανερό και ένα μειονέκτημα του αισθητήρα LM35: Επειδή η τάση στην έξοδό του δε μπορεί να πάρει αρνητικές τιμές, δε μπορεί να χρησιμοποιηθεί για μέτρηση θερμοκρασιών κάτω του μηδενός. Αν είναι επιθυμητή η μέτρηση τέτοιων θερμοκρασιών πρέπει να χρησιμοποιηθεί είτε ένας ψηφιακός αισθητήρας π.χ. ο DS18B20 για το δίαυλο 1-Wire, είτε κάποιος πιο ευέλικτος αναλογικός αισθητήρας, π.χ. ο TMP36.


**Παρατήρηση :** Η συνάρτηση *millis()* επιστρέφει το χρόνο (σε msec) που πέρασε από τότε που άρχισε η εκτέλεση του κώδικα στον Arduino. Η τιμή του ίδιου χρόνου σε μικροδευτερόλεπτα ( $\mu s$ ) υπολογίζεται -με θεωρητική ακρίβεια 4 $\mu s$ - μέσω της συνάρτησης *micros()* [4].

## Ένα ρολόι πραγματικού χρόνου με οθόνη υγρών κρυστάλλων

Θα χρησιμοποιήσουμε τη μονάδα Tiny RTC σε συνδυασμό με μια οθόνη υγρών κρυστάλλων για σύνδεση μέσω του διαύλου I<sup>2</sup>C. Για το πρότζεκτ αυτό θα θεωρήσουμε πως:

- η μονάδα Tiny RTC αναγνωρίζεται στη διεύθυνση 0x68 στο διάλυο I<sup>2</sup>C, και
- η μονάδα LCD στη διεύθυνση 0x27.

Οι σχετικές συνδέσεις φαίνονται στην επόμενη εικόνα:



Εικόνα 62: Ρολόι πραγματικού χρόνου με οθόνη υγρών κρυστάλλων

Το αντίστοιχο sketch μπορεί να έχει τη μορφή:

```
/*  
  Ex.34 : Ρολόι πραγματικού χρόνου με οθόνη υγρών κρυστάλλων  
*/  
  
#include <Wire.h> // Συμπερίληψη της βιβλιοθήκης Wire  
#include <RTClib.h> // Συμπερίληψη της βιβλιοθήκης RTClib  
#include <LiquidCrystal_I2C.h> // Συμπερίληψη βιβλιοθήκης LiquidCrystal_I2C  
  
LiquidCrystal_I2C lcd(0x27,16,2); // Δημιουργία αντικειμένου LiquidCrystal_I2C  
RTC_DS1307 rtc; // Δημιουργία αντικειμένου τύπου RTC_DS1307  
  
void setup ()  
{  
  Serial.begin(9600); // Ενεργοποίηση σειριακής επικοινωνίας  
  rtc.begin(); // Εκκίνηση επικοινωνίας με RTC μέσω I2C  
  lcd.init(); // Ενεργοποίηση οθόνης  
  lcd.backlight(); // Ενεργοποίηση οπίσθιου φωτισμού της οθόνης  
  lcd.clear(); // Εκκαθάριση οθόνης  
  lcd.print("Real Time Clock"); // Εισαγωγικό μήνυμα  
}
```

```

void loop()           // Τυπώνει ημερομηνία και ώρα στην οθόνη υγρών κρυστάλλων
{
    DateTime now = rtc.now();           // Λήψη τρέχουσας ημερομηνίας και ώρας

    lcd.clear();
    lcd.print("Date: ");
    lcd.print(now.year(), DEC);
    lcd.print("/");
    lcd.print(now.month(), DEC);
    lcd.print("/");
    lcd.print(now.day(), DEC);

    lcd.setCursor(0,1);
    lcd.print("Time: ");
    lcd.print(now.hour(), DEC);
    lcd.print(":");
    lcd.print(now.minute(), DEC);
    lcd.print(":");
    lcd.print(now.second(), DEC);


    delay(1000);           // Αναμονή 1 sec
}

```

Εννοείται πως πρέπει να έχει προηγηθεί ο συγχρονισμός του RTC με την τρέχουσα ημερομηνία και ώρα, όπως περιγράφηκε στη σχετική παράγραφο (σελ.65) της δεύτερης ενότητας.

## Ηλεκτρικές μετρήσεις με τον Arduino και το INA219 – Νόμος του Ohm

Το ολοκληρωμένο INA219 της Texas Instruments μετράει την τάση κατά μήκος μιας αντίστασης  $0,1\Omega$  (ανοχής 1%), η οποία παρεμβάλλεται στη διαδρομή του ρεύματος, και από την τάση αυτή προσδιορίζει την ένταση του ρεύματος που διαρρέει την αντίσταση. Στο εμπόριο διατίθενται προσυναρμολογημένες μονάδες που περιέχουν το INA219 μαζί με τα απαραίτητα ηλεκτρονικά εξαρτήματα για τη σύνδεση μέσω του διαύλου I<sup>2</sup>C σε μικροϋπολογιστικά συστήματα.



Εικόνα 63: Προσυναρμολογημένη μονάδα INA219

Το ολοκληρωμένο INA219 περιλαμβάνει στην είσοδό του ένα τελεστικό ενισχυτή ακριβείας, που η μέγιστη επιτρεπτή διαφορά δυναμικού στις εισόδους του είναι  $\pm 320mV$ . Με βάση το νόμο του Ohm ( $V = I \cdot R$ ) και με  $V = \pm 320mV$  και  $R = 100m\Omega$  συμπεραίνουμε πως μπορούν να μετρηθούν ρεύματα στην περιοχή  $\pm 3,2 A$ .

Το INA219 ενσωματώνει ένα αναλογικο-ψηφιακό μετατροπέα μέγιστης ακρίβειας  $12bit$  που σημαίνει πως στην περιοχή των  $\pm 3,2 A$  έχει διακριτική ικανότητα  $0,8mA$ . Μεταβάλλοντας το κέρδος του τελεστικού ενισχυτή μπορούμε να μεταβάλλουμε την περιοχή μέτρησης αλλά και την αντίστοιχη διακριτική ικανότητα. Για παράδειγμα με το ελάχιστο προκαθορισμένο κέρδος η περιοχή μέτρησης περιορίζεται στα  $\pm 400mA$  με διακριτική ικανότητα  $0,1mA$ . Κάθε αναλογικοψηφιακή μετατροπή σε ακρίβεια  $12bit$  διαρκεί  $532\mu s$ . Το ολοκληρωμένο μπορεί να προγραμματιστεί για συνεχή τρόπο λειτουργίας, κατά τον οποίο με τη λήξη μιας μετατροπής ξεκινάει αυτόματα μια άλλη μετατροπή, ενώ το αποτέλεσμα διατηρείται στο σχετικό καταχωρητή του ολοκληρωμένου. Ένα επιπλέον σημαντικό χαρακτηριστικό του INA219 είναι η δυνατότητα λήψης  $2^N$  ( $N = 1..7$ ) διαδοχικών μετατροπών και η επιστροφή του μέσου όρου τους, με αποτέλεσμα τη μείωση του ψηφιακού θορύβου, αλλά με σημαντική αύξηση του απαιτούμενου χρόνου για τη μέτρηση [16].


Η επικοινωνία της μονάδας INA219 με τον Arduino υλοποιείται μέσω του δισύρματου διαύλου I<sup>2</sup>C, στον οποίο εξ ορισμού αναγνωρίζεται στη δεκαεξαδική διεύθυνση  $0x40$ , κάτι που όμως μπορεί να αλλάξει βραχυκυκλώνοντας τις κατάλληλες επαφές στην πλακέτα της μονάδας INA219.

Υπάρχουν διάφορες βιβλιοθήκες που απλοποιούν τον προγραμματισμό του Arduino για τη μέτρηση της έντασης ρεύματος με τη μονάδα INA219:

- Adafruit INA219 Library
- INA219 Library του John De Cristofaro
- INA219 Library του Korneliusz Jarzebski

Στα επόμενα παραδείγματα χρησιμοποιούμε τη βιβλιοθήκη του Korneliusz Jarzebski, η οποία μπορεί να μεταφορτωθεί από την ιστοσελίδα <https://github.com/jarzebski/Arduino-INA219> ως ένα συμπιεσμένο (τύπου zip) αρχείο, και να εγκατασταθεί μέσω της σχετικής εντολής (από το μενού “Σχέδιο\Συμπερίληψη βιβλιοθήκης\Προσθήκη βιβλιοθήκης ZIP”) του Arduino IDE.

Στο πρώτο παράδειγμα με χρήση της μονάδας INA219 μετράμε την ένταση του ρεύματος που διαρρέει ένα απλό κύκλωμα, που περιλαμβάνει δύο αντιστάσεις συνδεδεμένες σε σειρά και τροφοδοτούμενες από μπαταρία των 9V.



Εικόνα 64: Σύνδεση μονάδας INA219 στον Arduino για μέτρηση ρεύματος

/\*

*Ex.35 : INA219 Bi-directional Current/Power Monitor*

\*/

```
#include <Wire.h>
#include <INA219.h>
```

```
INA219 ina219;
```

```
void setup()
```

```
{
```

```
  Serial.begin(9600);
  Serial.println("INA219 Current monitor");
  Serial.println("-----");
```

```
  // Default INA219 address is 0x40
  ina219.begin();
```

```
  // Configure INA219
```

```
  // INA219_RANGE_32V: Max input Voltage = 32V
```

```
  // INA219_GAIN_320MV: Max ShuntVoltage = 320mV - Current range ±3,2 A
```

```
  // INA219_BUS_RES_12BIT: ADC precision 12bit
```

```
  // INA219_SHUNT_RES_12BIT_1S: 1 measurement = 1 Conversion
```

```
  ina219.configure( INA219_RANGE_32V, INA219_GAIN_320MV,
                   INA219_BUS_RES_12BIT, INA219_SHUNT_RES_12BIT_1S);
```

```
  // Calibrate INA219: Rshunt = 0.1 ohm, Max excepted current = 2A
```

```
  ina219.calibrate(0.1, 2);
```

```
}
```

```
void loop()
```

```
{
```

```
  Serial.print("Current (A) : ");
  Serial.println(ina219.readShuntCurrent(), 3);
  delay(1000);
```

```
}
```



```


// Configure INA219
ina219.configure( INA219_RANGE_32V, INA219_GAIN_320MV,
                  INA219_BUS_RES_12BIT, INA219_SHUNT_RES_12BIT_16S);

// Calibrate INA219 - Rshunt = 0.1 ohm, Max expected current = 2A
ina219.calibrate(0.1, 2);
}

void loop()
{
  Serial.print("Current (A) : ");
  Serial.print(ina219.readShuntCurrent(), 3);
  Serial.print(" Bus Voltage (V) : ");
  Serial.print(ina219.readBusVoltage(), 4);
  Serial.print(" Total Voltage (V) : ");
  Serial.print(ina219.readBusVoltage()+ina219.readShuntVoltage(), 4);
  Serial.print(" Bus Power (W): ");
  Serial.println(ina219.readBusPower(), 4);
  delay(1000);
}

```

Με τη βοήθεια του αισθητήρα INA219, εύκολα μπορούμε να σχεδιάσουμε ένα πείραμα για την επιβεβαίωση του νόμου του Ohm. Το σχετικό κύκλωμα φαίνεται στην επόμενη εικόνα:



Εικόνα 66: Συνδεσμολογία για το πείραμα του νόμου του Ohm

Χρησιμοποιήσαμε μια αντίσταση ονομαστικής τιμής 330Ω και με τη βοήθεια του ρυθμιζόμενου σταθεροποιημένου τροφοδοτικού περιορίσαμε την τάση στα άκρα της μέχρι τα 9V, ώστε η μέγιστη

τιμή της έντασης του ρεύματος στο κύκλωμα να μη ξεπερνάει τα 30mA περίπου.

Το σχετικό sketch για τον Arduino μπορεί να έχει τη μορφή:

```
/*
   Ex.37 : Επιβεβαίωση του Νόμου του Ohm με το INA219
*/

#include <Wire.h>
#include <INA219.h>
INA219 ina219;

void setup()
{
  Serial.begin(9600);
  Serial.println("INA219 Current/Power monitor");
  Serial.println("-----");

  // Default INA219 address is 0x40
  ina219.begin();

  // Configure INA219
  ina219.configure( INA219_RANGE_32V,
                   INA219_GAIN_320MV,
                   INA219_BUS_RES_12BIT,
                   INA219_SHUNT_RES_12BIT_16S);


  // Calibrate INA219 - Rshunt = 0.1 ohm, Max excepted current = 2A
  ina219.calibrate(0.1, 2);
}

void loop()
{
  Serial.println("Send a key to start measure...");
  while (Serial.available() == 0);


  Serial.print("Current (A) : ");
  Serial.print(ina219.readShuntCurrent(), 3);
  Serial.print(" - Voltage (V) : ");
  Serial.println(ina219.readBusVoltage(), 2);

  // Clear Serial Buffer
  while (Serial.available() != 0)
  {
    int c = Serial.read();
  }
}
```

Το λογισμικό με τη συνάρτηση *readShuntCurrent()* επιστρέφει την τιμή *I* του ρεύματος που διαρρέει την αντίσταση, και με τη συνάρτηση *readBusVoltage()* την τάση *V* στα άκρα της (Εικόνα 67). Η επιβεβαίωση του νόμου του Ohm μπορεί να γίνει π.χ. μέσω της γραφικής παράστασης  $V = f(I)$ , που μπορούμε να σχεδιάσουμε με τη βοήθεια του λογισμικού Excel (Εικόνα 68).



Εικόνα 67 : Πειραματικά δεδομένα για την επιβεβαίωση του Νόμου του Ohm




Εικόνα 68 : Η γραφική παράσταση  $V = f(I)$

Από την κλίση της γραφικής παράστασης  $V = f(I)$ , μπορούμε να υπολογίσουμε την πειραματική τιμή της αντίστασης που χρησιμοποιήσαμε:  $R = 325\Omega$ .


## Δικτύωση Bluetooth

Τα κλασικά συστήματα Arduino (Uno, Leonardo, Nano, Micro, Due και οι διάφορες παραλλαγές τους) δε διαθέτουν εξ ορισμού δυνατότητες ασύρματης επικοινωνίας μέσω Bluetooth. Μπορούν όμως να αποκτήσουν τέτοιες δυνατότητες μέσω της προσυναρμολογημένης μονάδας HC-05 ή HC-06 (Εικόνα 69).




Εικόνα 69: Οι μονάδες HC-05 (αριστερά) και HC-06 (δεξιά)

Οι μονάδες αυτές επικοινωνούν με τον Arduino σειριακά μέσω των ακίδων τους RX και TX. Λαμβάνουν σειριακά πληροφορίες από τον Arduino και τις εκπέμπουν ασύρματα, ώστε να ληφθούν από κάποια συζευγμένη και απομακρυσμένη συσκευή ή λαμβάνουν ασύρματα πληροφορίες από την απομακρυσμένη συσκευή και τις αποστέλλουν σειριακά στον Arduino. Τροφοδοτούνται με τάση από 3,6 V μέχρι 6 V, αλλά ακολουθούν λογική 3,3 V. Αυτό σημαίνει πως όταν στέλνουν δεδομένα στον Arduino μέσω της ακίδας TX, η υψηλή λογική στάθμη (HIGH) αντιστοιχεί σε 3,3V, η οποία είναι αρκετή ώστε ο Arduino να την αναγνωρίσει ως υψηλή (HIGH), παρότι ο Arduino ακολουθεί λογική 5V. Από την άλλη όταν ο Arduino στέλνει δεδομένα στην ακίδα RX της μονάδας Bluetooth, η υψηλή λογική στάθμη (HIGH) αντιστοιχεί στη λογική του Arduino, είναι δηλαδή 5 V, τάση που δεν είναι ανεκτή από τις μονάδες HC-05 και HC-06.



Εικόνα 70: Σύνδεση της μονάδας HC-05 στον Arduino (αντίστοιχες είναι και οι συνδέσεις της μονάδας HC-06)

Η απλούστερη λύση για την ορθή λειτουργία, χωρίς κίνδυνο καταστροφής των μονάδων Bluetooth είναι η χρήση ενός απλού διαιρέτη τάσης στην ακίδα τους RX (Εικόνα 70). Θα πρέπει να ληφθεί υπόψη πως η ακίδα RX της μονάδας Bluetooth συνδέεται στην ακίδα TX του Arduino, και η ακίδα της TX στην ακίδα RX του Arduino. Αν και γενικά δεν απαιτείται η τροφοδοσία της μονάδας Bluetooth από τον Arduino, απαιτείται όμως η κοινή τους γείωση (Εικόνα 71).



Εικόνα 71: Σειριακή σύνδεση δύο συσκευών

Η μέσω υλικού σειριακή επικοινωνία του Arduino υλοποιείται με τις ακίδες 0 (RX - λήψη) και 1 (TX - εκπομπή). Καθώς όμως οι ίδιες ακίδες χρησιμοποιούνται για την επικοινωνία του Arduino μέσω USB, δε μπορούμε να τις χρησιμοποιήσουμε ταυτόχρονα και για την επικοινωνία με τη μονάδα Bluetooth. Το πρόβλημα αυτό λύνεται με τη χρήση της βιβλιοθήκης SoftwareSerial, με την οποία μπορεί να υλοποιηθεί η σειριακή επικοινωνία σε οποιεσδήποτε άλλες δύο ακίδες του Arduino. Για τη χρήση της βιβλιοθήκης σε κάποιο sketch μας, θα πρέπει πρώτα να δηλωθεί η χρήση της, καθώς και οι ακίδες του Arduino που θα χρησιμεύσουν για τη σειριακή εκπομπή (TX) και λήψη (RX) ως:

```
#include <SoftwareSerial.h>
#define TX_pin 2
#define RX_pin 3
```

και μετά να δημιουργηθεί ένα αντικείμενο κλάσης SoftwareSerial ως εξής:

```
SoftwareSerial mySerial(RX_pin, TX_pin);
```

Στη συνέχεια η σειριακή επικοινωνία μεταξύ Arduino και μονάδας Bluetooth μπορεί να υλοποιηθεί με τον κλασσικό τρόπο. Για παράδειγμα με την εντολή:

```
mySerial.begin(9600);
```

ενεργοποιείται η μέσω λογισμικού σειριακή επικοινωνία με ρυθμό 9600bps, που είναι και ο προκαθορισμένος ρυθμός επικοινωνίας της μονάδας Bluetooth. Με την εντολή:

```
mySerial.print("something");
```

ο Arduino στέλνει το αλφαριθμητικό "something" στη μονάδα Bluetooth, ενώ με την εντολή:

```
char inc = mySerial.read();
```

ο Arduino διαβάζει ένα χαρακτήρα που έχει αποσταλεί από τη μονάδα Bluetooth.

Αν για οποιοδήποτε λόγο δεν είναι επιθυμητή η μέσω λογισμικού σειριακή επικοινωνία, μπορεί η μονάδα Bluetooth να συνδεθεί στις ακίδες 0 και 1 (hardware serial) του Arduino, αλλά:

1. Για τον προγραμματισμό του Arduino θα πρέπει να αποσυνδέεται η μονάδα Bluetooth και μετά να συνδέεται εκ νέου.
2. Η σειριακή μέσω USB επικοινωνία με το συνδεδεμένο υπολογιστή δεν είναι πλέον εφικτή.

Ας δούμε τώρα ένα απλό sketch επικοινωνίας μέσω Bluetooth μεταξύ του Arduino και μιας απομακρυσμένης συσκευής, π.χ του υπολογιστή μας ή του κινητού μας τηλεφώνου.

```

/*
   Ex.38 : Απλή επικοινωνία μέσω Bluetooth
*/

#include <SoftwareSerial.h>
#define TX_pin 2
#define RX_pin 3

SoftwareSerial BT(RX_pin, TX_pin);

void setup()
{
  Serial.begin(9600);
  BT.begin(9600);
}

void loop()
{
  if (BT.available())
  {
    Serial.write(BT.read());
  }
  if (Serial.available())
  {
    BT.write(Serial.read());
  }
}

```

Στο sketch αρχικά με την εντολή:

```
Serial.begin(9600)
```

ενεργοποιείται η σειριακή επικοινωνία μεταξύ του Arduino και του υπολογιστή στον οποίο συνδέεται μέσω USB με ρυθμό 9600bps.

Μετά με την εντολή:

```
BT.begin(9600)
```

ενεργοποιείται η σειριακή επικοινωνία μεταξύ του Arduino και της μονάδας Bluetooth και πάλι στα 9600bps.

Στο συνέχεια στο βρόχο *loop()* και στο μπλοκ εντολών:

```

if (BT.available())
{
  Serial.write(BT.read());
}

```

αν η μονάδα Bluetooth έχει λάβει δεδομένα από κάποια απομακρυσμένη συσκευή αυτά διαβάζονται από τον Arduino και αποτυπώνονται στη σειριακή οθόνη του Arduino IDE.

Και με τις εντολές:


```

if (Serial.available())
{
  BT.write(Serial.read());
}

```

δίνεται η δυνατότητα αποστολής στη μονάδα Bluetooth των δεδομένων που καταγράφονται στη σειριακή οθόνη του Arduino IDE. Τα δεδομένα αυτά, χωρίς περαιτέρω δική μας παρέμβαση, εκπέμπονται από τη μονάδα Bluetooth και αφού ληφθούν από την απομακρυσμένη συσκευή μπορούν να αποτυπωθούν σε αντίστοιχη σειριακή οθόνη.

Αφού μεταγλωττίσετε και “ανεβάσετε” τον κώδικα του sketch στον Arduino, θα χρειαστεί να ανοίξετε τη σειριακή οθόνη του Arduino IDE έχοντας επιλεγμένη τη σειριακή θύρα στην οποία ο υπολογιστής σας «ακούει» τον Arduino (είναι η ίδια σειριακή θύρα που χρησιμοποιήσατε κατά τον προγραμματισμό του Arduino). Στη σειριακή οθόνη θα πρέπει να ρυθμίσετε το ρυθμό επικοινωνίας στα 9600bps και να ενεργοποιήσετε την επιλογή “Δεν υπάρχει τέλος γραμμής” (Εικόνα 72).





Εικόνα 72: Η σειριακή οθόνη του Arduino IDE



Ως απομακρυσμένη συσκευή, με την οποία θα ανταλλάσσει μηνύματα ο Arduino, θα χρησιμοποιήσετε το κινητό σας τηλέφωνο. Θα πρέπει:

1. Να εγκαταστήσετε στο κινητό σας τηλέφωνο την εφαρμογή “*Serial Bluetooth Terminal*” από το Google Play Store.
2. Να ενεργοποιήσετε την επικοινωνία Bluetooth και να κάνετε τη σύζευξη του κινητού με τη μονάδα Bluetooth, η οποία στη λίστα με τις ανιχνευμένες συσκευές θα εμφανίζεται με το εξ ορισμού όνομα “*HC-05*” ή “*HC-06*” ή κάτι άλλο ανάλογα με τον κατασκευαστή (Εικόνα 73). Ο εξ ορισμού κωδικός για τη σύζευξη είναι “1234”.


Μετά είστε έτοιμοι να “τρέξετε” την εφαρμογή “*Serial Bluetooth Terminal*”, να συνδεθείτε στη μονάδα Bluetooth του Arduino και να στείλετε ή να λάβετε μηνύματα. Η σύνδεση στη μονάδα Bluetooth γίνεται μέσω του μενού “*Devices / Bluetooth Classic*” και επιλέγοντας τη συσκευή με το όνομα “*HC-05*” ή “*HC-06*” (Εικόνα 74). Τα δεδομένα που αποστέλλονται με το κινητό, λαμβάνονται από τη μονάδα Bluetooth και εμφανίζονται στη σειριακή οθόνη του Arduino IDE (Εικόνα 75) και αντίστοιχα τα δεδομένα που από λαμβάνει η μονάδα Bluetooth από τον Arduino, εμφανίζονται στην οθόνη του κινητού μας (στο παράθυρο του “*Serial Bluetooth Terminal*”, Εικόνα 74).



Εικόνα 73: Ανίχνευση και σύζευξη της μονάδας Bluetooth από το κινητό τηλέφωνο




Εικόνα 74: Η εφαρμογή Serial Bluetooth Terminal



Εικόνα 75: Ανταλλαγή μηνυμάτων μέσω Bluetooth

Με το ίδιο περίπου όνομα “*Bluetooth Serial Terminal*” υπάρχει αντίστοιχη εφαρμογή για υπολογιστές με λειτουργικό σύστημα Windows (Εικόνα 76). Θα το βρείτε δωρεάν στο Microsoft Store.




Εικόνα 76: Bluetooth Serial Terminal για Windows

Θα πρέπει βέβαια να έχει προηγηθεί η σύζευξη της μονάδας Bluetooth με τον υπολογιστή. Μπορείτε να το κάνετε κάνοντας “κλικ” στο εικονίδιο “Συσκευές Bluetooth” στη γραμμή εργασιών και επιλέγοντας “Προσθήκη μιας συσκευής Bluetooth” ή από τις “Ρυθμίσεις” και επιλέγοντας “Bluetooth και συσκευές / Προσθήκη συσκευής”.


## Προσθήκη συσκευής

Επιλέξτε το είδος συσκευής που θέλετε να προσθέσετε.



## Προσθήκη συσκευής

Βεβαιωθείτε ότι η συσκευή σας είναι ενεργοποιημένη και ανιχνεύσιμη. Επιλέξτε μια συσκευή παρακάτω για να συνδεθείτε.




Άκυρο

Άκυρο

Εικόνα 77: Προσθήκη και σύζευξη συσκευής Bluetooth στα Windows

Το λειτουργικό σύστημα αναγνωρίζει τη συζευγμένη μονάδα Bluetooth ως μία ή δύο εικονικές σειριακές θύρες. Αν και μόνο μία από αυτές τις θύρες λειτουργεί, η οποία θα ανακαλυφθεί με τη μέθοδο δοκιμής και λάθους, μια νέα δυνατότητα για την επικοινωνία Arduino και υπολογιστή προκύπτει. Μπορεί να χρησιμοποιηθεί οποιαδήποτε εφαρμογή εξομοίωσης τερματικού, όπως για παράδειγμα το ελεύθερα διανεμόμενο “Termite” που μπορείτε να μεταφορτώσετε από την ιστοσελίδα: [https://www.compuphase.com/software\\_termite.htm](https://www.compuphase.com/software_termite.htm).




Εικόνα 78: Ρυθμίσεις στο Termite

Αφού γίνουν οι κατάλληλες ρυθμίσεις στον εξομοιωτή τερματικού:

- Port: η εικονική σειριακή πόρτα που αντιστοιχεί στο μονάδα Bluetooth, και
- Baud rate: ο ρυθμός επικοινωνίας της μονάδας Bluetooth,

ο υπολογιστής με τον εξομοιωτή τερματικού μπορεί να ανταλλάσσει πληροφορίες με τον Arduino μέσω Bluetooth (Εικόνα 79).



Εικόνα 79: Σειριακή επικοινωνία Arduino - PC μέσω Bluetooth


Το ίδιο sketch που “ανεβάσατε” στον Arduino μπορεί να χρησιμοποιηθεί και για τη ρύθμιση των ιδιοτήτων της μονάδας Bluetooth, στέλνοντας μέσω της σειριακής οθόνης του Arduino IDE τις κατάλληλες “AT” εντολές. Για παράδειγμα για τη μονάδα HC-06:

- Πληκτρολογώντας “AT” στη σειριακή οθόνη και πατώντας “ENTER”, αν η μονάδα Bluetooth ανταποκριθεί με “OK”, σημαίνει πως είναι έτοιμη να δεχθεί εντολές.
- Με την εντολή “AT+NAME” ακολουθούμενη από κάποιο αλφαριθμητικό μπορεί να αλλάξει το όνομα με το οποίο η μονάδα θα είναι ορατή στον εξωτερικό κόσμο.
- Με την εντολή “AT+BAUD8” ρυθμίζεται ο ρυθμός επικοινωνίας στα 115200bps, ενώ με την εντολή “AT+BAUD4” ρυθμίζεται στα 9600bps.

Προσοχή να δοθεί πως για τη μονάδα HC-06 πρέπει να ενεργοποιήσετε την επιλογή “Δεν υπάρχει τέλος γραμμής” στη σειριακή οθόνη του Arduino IDE. Αναλυτικά οι “AT” εντολές για τις μονάδες HC-05 και HC-06 δίνονται στο Παράρτημα Γ.

### ***Το πρότυπο Bluetooth Low Energy (BLE)***

Η τεχνολογία Bluetooth σχεδιάστηκε αρχικά για τη μετάδοση μεγάλου όγκου δεδομένων σε μικρή απόσταση. Βασικό μειονέκτημά της ήταν η περιορισμένη εφαρμογή της σε φορητές συσκευές, αφού οι αντίστοιχες μονάδες κατανάλωναν πολύ γρήγορα την ενέργεια των μπαταριών. Το 2011 παρουσιάστηκε το πρότυπο Bluetooth 4.0 ή αλλιώς Bluetooth Low Energy με βασικό πλεονέκτημα τη χαμηλή κατανάλωση ενέργειας, ώστε οι συσκευές να λειτουργούν για μεγαλύτερα χρονικά διαστήματα και με μικρότερες μπαταρίες. Για το σκοπό αυτό το πρότυπο BLE χρησιμοποιεί υψηλότερους ρυθμούς μεταφοράς δεδομένων με αποτέλεσμα τη μείωση της χρονικής διάρκειας της μετάδοσης. Μειονέκτημα του προτύπου είναι ότι μπορεί να χρησιμοποιηθεί ικανοποιητικά μόνο στις περιπτώσεις που απαιτείται περιοδική μετάδοση μικρού όγκου δεδομένων.




Εικόνα 80: Οι μονάδες BLE HM-10 (αριστερά) και AT-09 (δεξιά)

Στο εμπόριο κυκλοφορούν δύο μονάδες BLE για χρήση σε κυκλώματα μικροελεγκτών οι: HM-10 και AT-09 (Εικόνα 80), με τη δεύτερη, που αποτελεί μια περιορισμένων δυνατοτήτων εκδοχή της πρώτης, να παρουσιάζει προβλήματα σύνδεσης στα Windows 11 (αν και αναγνωρίζεται σωστά). Οι δυο μονάδες συνδέονται στον Arduino με τον ίδιο ακριβώς τρόπο όπως και οι κλασσικές Bluetooth μονάδες HC-05 και HC-06, και μάλιστα το ίδιο ακριβώς sketch λειτουργεί εξίσου καλά σε όλες τις μονάδες. Για τη δοκιμή επικοινωνίας μπορεί να χρησιμοποιηθεί το κινητό τηλέφωνο και η εφαρμογή “Serial Bluetooth Terminal”. Η σύνδεση στη μονάδα γίνεται μέσω του μενού “Devices / Bluetooth LE” και επιλέγοντας τη συσκευή με το όνομα “BT05” (για τη μονάδα AT-09 που διαθέταμε), χωρίς να απαιτείται η σύζευξη της μονάδας με το κινητό.

## Ένας απλός αυτοματισμός μέσω Bluetooth

Θα δείξουμε τώρα πως μπορείτε να κατασκευάσετε ένα απλό αυτοματισμό που θα σας δίνει τη δυνατότητα να αναβοσβήνετε μια λάμπα φωτισμού εξ αποστάσεως, όση τουλάχιστον σας επιτρέπει η δικτύωση Bluetooth με τη μονάδα HC-06.



Εικόνα 81: Το κύκλωμα του αυτοματισμού

Αφού συναρμολογήσετε το κύκλωμα όπως φαίνεται στην Εικόνα 81, μεταγλωττίστε και “ανεβάστε” το επόμενο sketch στον Arduino.

/\*

*Ex.39 : Αυτοματισμός μέσω Bluetooth*

\*/

```
#include <SoftwareSerial.h>
```

```
#define TX_pin 2
```

```
#define RX_pin 3
```

```
#define bulb_pin 7
```

```
SoftwareSerial BT(RX_pin, TX_pin);
```

```
void setup() {  
  pinMode(bulb_pin, OUTPUT);  
  BT.begin(9600);  
}
```

```
void loop() {  
  if (BT.available()) {  
    char btInChar = BT.read();  
    if (btInChar == '1') {  
      digitalWrite(bulb_pin, 1);  
    }  
    else if (btInChar == '0') {  
      digitalWrite(bulb_pin, 0);  
    }  
  }  
}
```

```

else if (btInChar == 'h') {
  if (digitalRead(bulb_pin)) {
    BT.println("Bulb is ON");
  }
  else {
    BT.println("Bulb is OFF");
  }
}
}
}
}
}

```

Για να δοκιμάσετε τη λειτουργία της διάταξης, ανεβάσετε πρώτα το sketch στον Arduino και ολοκληρώσετε τη σύζευξη της μονάδας Bluetooth με τον υπολογιστή σας. Μετά “τρέξτε” την εφαρμογή “*Bluetooth Serial Terminal*” και συνδεθείτε στη μονάδα Bluetooth “HC-06” (ή οποιοδήποτε άλλο είναι το όνομα της μονάδας που έχει συνδεθεί στον Arduino). Βεβαιωθείτε πως στο πεδίο “*Transmit Line End*” έχετε επιλέξει “*No Line Ending*” και στείλτε ‘I’ στη μονάδα Bluetooth του Arduino για να ανάψετε τη λάμπα ή ‘O’ για να τη σβήσετε. Στέλνοντας το χαρακτήρα “h” ο Arduino ανταποκρίνεται αναφέροντας την κατάσταση της λάμπας.


Βέβαια η λειτουργικότητα της διάταξης αυξάνεται χρησιμοποιώντας την αντίστοιχη εφαρμογή για κινητό τηλέφωνο (“*Serial Bluetooth Terminal*”), στις ρυθμίσεις της οποίας πρέπει -για ορθή λειτουργία- να επιλεγεί “*Clear Input on Send*”.

## Μια απλή φωτοπύλη

Οι φωτοπύλες είναι διατάξεις που λειτουργούν ως ψηφιακοί διακόπτες, έχουν δηλαδή δύο καταστάσεις λειτουργίας: ON και OFF. Αποτελούνται από ένα πομπό και ένα δέκτη φωτός τοποθετημένους σε μικρή απόσταση μεταξύ τους, ώστε η δέσμη φωτός του πομπού, που κατευθύνεται προς το δέκτη, να μπορεί να διακόπτεται λόγω της διέλευσης ανάμεσά τους κάποιου αδιαφανούς αντικειμένου. Οι δύο καταστάσεις (ON και OFF) της φωτοπύλης καθορίζονται ανάλογα αν η δέσμη φωτός από τον πομπό φτάνει ή διακόπτεται και δε φτάνει στο δέκτη.

Συνδυαζόμενες οι φωτοπύλες με ένα ηλεκτρονικό χρονομετρητή επιτρέπουν την εξαιρετικά ακριβή χρονομέτρηση γεγονότων που σχετίζονται με την αλλαγή κατάστασης της φωτοπύλης, δηλαδή τον προσδιορισμό των χρονικών στιγμών που αρχίζει η διακοπή ή η αποκατάσταση της δέσμης της και συνεπώς και των χρονικών διαστημάτων που η δέσμη διακόπτεται ή έχει αποκατασταθεί.

Στην απλή εκδοχή που εδώ παρουσιάζουμε, ως πομπό στη φωτοπύλη θα χρησιμοποιήσουμε ένα κόκκινο ή άλλου χρώματος LED και ως δέκτη μια φωτοαντίσταση (LDR), ενώ ρόλο του ηλεκτρονικού χρονομετρητή θα αναλάβει ο Arduino με το κατάλληλο λογισμικό. Το σχετικό κυκλωματικό διάγραμμα φαίνεται στην Εικόνα 82.




Εικόνα 82: Το κυκλωματικό διάγραμμα μιας απλής φωτοπύλης για χρήση με τον Arduino


Το κόκκινο LED ( $D_1$ ) μέσω του αντιστάτη  $R_1 = 220 \Omega$  συνδέεται στην τάση τροφοδοσίας (5 V) του Arduino, ενώ η φωτοαντίσταση (LDR) και ο αντιστάτης  $R_2 = 1 \text{ k}\Omega$  σχηματίζουν ένα διαιρέτη τάσης. Σε καταστάσεις χαμηλού φωτισμού η φωτοαντίσταση παρουσιάζει πολύ υψηλή αντίσταση, με αποτέλεσμα η τάση στα άκρα της  $R_2$  να είναι αρκετά μικρή (κατάσταση OFF). Αν όμως ο φωτισμός αυξηθεί η αντίσταση της φωτοαντίστασης μειώνεται και αυτό οδηγεί σε αύξηση της τάσης στα άκρα της  $R_2$  (κατάσταση ON).

Χρειάζεται πειραματισμός για την επιλογή της αντίστασης  $R_2$ , ώστε οι καταστάσεις ON και OFF της φωτοπύλης να αντιστοιχούν σε καταστάσεις HIGH και LOW της ψηφιακής εισόδου D7 του Arduino, στην οποία οδηγείται η τάση στα άκρα της  $R_2$ . Στο πρωτότυπό μας πήραμε ικανοποιητικά αποτελέσματα επιλέγοντας  $R_2 = 10 \text{ k}\Omega$ . Ας σημειωθεί πως η σύνδεση του κυκλώματος της φωτοαντίστασης σε ψηφιακή είσοδο του Arduino είναι αντίθετη από τη συνήθη πρακτική, που η τάση στα άκρα της  $R_2$  συνδέεται σε κάποια από τις αναλογικές εισόδους του Arduino. Και αυτό γιατί εδώ μας ενδιαφέρει μόνο η αλλαγή κατάστασης (από HIGH σε LOW ή αντίστροφα) στην αντίσταση  $R_2$  και όχι η ακριβής τιμή της τάσης στα άκρα της.

Το κύκλωμα της φωτοπύλης εύκολα υλοποιείται σε ένα breadboard, όπως φαίνεται στην Εικόνα 83. Δε φαίνεται στην Εικόνα 83, αλλά τόσο το LED όσο και η φωτοαντίσταση είναι υπερυψωμένα σε σχέση με την επιφάνεια του breadboard και με τις κεφαλές τους να στρέφονται η μία προς την άλλη, όπως φαίνεται στην Εικόνα 84.



Εικόνα 83: Η φωτοπύλη σε breadboard



Εικόνα 84: Κατασκευή του πρωτοτύπου της φωτοπύλης

Το λογισμικό που «τρέχει» στον Arduino:

1. Καταγράφει συνεχώς την τιμή της ψηφιακής εισόδου D7, στην οποία συνδέεται η φωτοπύλη.
2. Όταν ανιχνεύσει χαμηλή λογική στάθμη, που σημαίνει έναρξη της διακοπής της φωτεινής δέσμης στη φωτοπύλη, αποθηκεύει την αντίστοιχη χρονική στιγμή (σε μs) στη μεταβλητή “*t\_in*”.
3. Μετά τίθεται σε κατάσταση αναμονής μέχρι να ανιχνευθεί υψηλή λογική στάθμη στην είσοδο D7, που σημαίνει έναρξη της αποκατάστασης του φωτός στη φωτοπύλη, και αποθηκεύει την αντίστοιχη χρονική στιγμή στη μεταβλητή “*t\_out*”.
4. Τέλος τυπώνει στη σειριακή κονσόλα τις δύο χρονικές στιγμές “*t\_in*” και “*t\_out*”.

Αν και αυτή η προγραμματιστική προσέγγιση (polling) δουλεύει πολύ καλά όσον αφορά τη μέτρηση και καταγραφή στη σειριακή κονσόλα των δύο χρονικών στιγμών (έναρξη διακοπής και έναρξη αποκατάστασης της φωτεινής δέσμης), παρουσιάζει ένα σημαντικό μειονέκτημα: Δεσμεύει όλη την υπολογιστική δύναμη του Arduino σε μια εξαιρετικά απλή και μικρών απαιτήσεων διαδικασία. Υπάρχουν δυνατότητες αντιμετώπισης αυτού του προβλήματος, αλλά σε αυτή τη φάση δε θα ασχοληθούμε με αυτές. Ο κώδικας όπως τον περιγράψαμε έχει τη μορφή:

```

/*
   Ex.40 : Απλή φωτοπόλη
*/

#define PHOTO_PIN 7


void setup() {
  Serial.begin(115200);
  pinMode(PHOTO_PIN, INPUT);
  Serial.println("In Time\tOut Time");           // Εκτύπωση τίτλου στηλών
}

void loop()
{
  if (digitalRead(PHOTO_PIN) == LOW) {          // Αν έχει αρχίσει η σκίαση της φωτοπόλης
    long t_in = micros();                       // Αποθήκευσε τη χρ. στιγμή έναρξης σκίασης

    while (digitalRead(PHOTO_PIN) == LOW);      // Αναμονή, όσο η φωτοπόλη σκιάζεται

    long t_out = micros();                      // Αποθήκευσε τη χρ. στιγμή αποκατάστασης
                                              // του φωτός στη φωτοπόλη
    Serial.println(String(t_in) + '\t' + String(t_out)); // Εκτύπωση μετρήσεων
  }
}


```



Εικόνα 85: Λεπτομέρεια της πειραματικής διάταξης για τη μέτρηση της περιόδου εκκρεμούς

Μπορούμε με τη φωτοπύλη που κατασκευάσαμε να υπολογίσουμε την περίοδο ενός απλού εκκρεμούς συγκεκριμένου μήκους, και με βάση αυτή την τιμή θα υπολογίσουμε την επιτάχυνση της βαρύτητας στον τόπο του πειράματος. Όπως φαίνεται και στη φωτογραφία της διάταξης (Εικόνα 85), στην κατακόρυφη θέση ισορροπίας του ο κύλινδρος του εκκρεμούς διακόπτει τη δέσμη της φωτοπύλης.

Θέτοντας σε κίνηση το εκκρεμές πήραμε 11 μετρήσεις ( $t_{in}$  και  $t_{out}$ ) καθώς ο μικρός κύλινδρος του εκκρεμούς περνούσε διακόπτοντας τη δέσμη της φωτοπύλης (Εικόνα 86).



Εικόνα 86: Λήψη μετρήσεων στη σειριακή κονσόλα του Arduino

Η περίοδος του εκκρεμούς υπολογίζεται ως το χρονικό διάστημα που απαιτείται για τρεις διαδοχικές διελεύσεις του κυλίνδρου του από τη φωτοπύλη. Για τους υπολογισμούς μπορούμε να χρησιμοποιήσουμε είτε τις χρονικές στιγμές εισόδου ( $t_{in}$ ) του κυλίνδρου στη φωτοπύλη, είτε τους αντίστοιχους χρόνους εξόδου ( $t_{out}$ ).

Πίνακας 3: Επεξεργασία πειραματικών δεδομένων

$\alpha/\alpha$	Διελεύσεις	$\Delta t_{in}$ ( $\mu s$ )	T (s)
1	1 $\rightarrow$ 3	1394636	1,394636
2	3 $\rightarrow$ 5	1394224	1,394224
3	5 $\rightarrow$ 7	1394196	1,394196
4	7 $\rightarrow$ 9	1394260	1,394260
5	9 $\rightarrow$ 11	1394452	1,394452

Βρίσκουμε την περίοδο του εκκρεμούς ως τη μέση τιμή των πέντε επιμέρους τιμών που υπολογίσαμε. Με ακρίβεια τεσσάρων (4) δεκαδικών ψηφίων είναι:

$$\bar{T} = 1,3944 \text{ s}$$

Για τις μικρού πλάτους αιωρήσεις η περίοδος του μήκους  $L = 0,486 \text{ m}$  εκκρεμούς δίνεται από την εξίσωση:

$$T = 2\pi\sqrt{\frac{L}{g}}, \text{ από όπου παίρνουμε: } g = \frac{4\pi^2 L}{T^2} \text{ και τελικά } g = 9,87 \text{ m/s}^2$$

**Παρατήρηση:** Αν και με τους δύο χρόνους “ $t_{in}$ ” και “ $t_{out}$ ” είμαστε σε θέση να προσδιορίσουμε το χρονικό διάστημα “ $\Delta t = t_{out} - t_{in}$ ” (χρήσιμο όταν θέλουμε να υπολογίσουμε την ταχύτητα διέλευσης ενός σώματος από τη φωτοπύλη) για λόγους που σχετίζονται με ασυμμετρίες στην καμπύλη απόκρισης της φωτοαντιστάσης δε θα χρησιμοποιήσουμε εδώ αυτή τη δυνατότητα.


## RGB LEDs – Μίξη χρωμάτων

Τα RGB LEDs συνδυάζουν σε ένα πακέτο τρία διαφορετικά LEDs: ένα κόκκινου (R), ένα πράσινου (G) και ένα μπλε (B) χρώματος και υπάρχουν σε δύο τύπους: κοινής ανόδου και κοινής καθόδου (Εικόνα 87).




Εικόνα 87: Τύποι RGB LEDs: αριστερά κοινής ανόδου, δεξιά κοινής καθόδου

Για τη σύνδεση ενός RGB LED κοινής καθόδου στον Arduino χρειάζονται τρεις αντιστάσεις περιορισμού του ρεύματος, μέσω των οποίων οι ακίδες R, G, B του LED συνδέονται π.χ. στις ακίδες 3, 5, 6 του Arduino, ενώ η κοινή κάθοδος των τριών LED (η πιο μακριά ακίδα του LED) συνδέεται στη γείωση του Arduino. (Εικόνα 88).



Εικόνα 88: Σύνδεση ενός RGB LED κοινής καθόδου στον Arduino

Καθώς κάθε ψηφιακή ακίδα μπορεί να παίρνει μόνο δύο τιμές (HIGH και LOW) που αντιστοιχούν σε υψηλή (5V) και χαμηλή (0 V) τάση, ο συνδυασμός των τριών ακίδων μπορεί να δώσει  $2^3=8$  διαφορετικούς συνδυασμούς τάσεων και συνεπώς μόνο 7 διαφορετικά χρώματα στο LED (Εικόνα 89). Ο συνδυασμός “RGB=000” διατηρεί όλα τα LED σβηστά και προφανώς δεν αντιστοιχίζεται στην εκπομπή κάποιου χρώματος.



Εικόνα 89: Επτά συνδυασμοί χρωμάτων με το RGB LED


Αν όμως μεταβάλλουμε την ένταση καθενός από τα τρία βασικά χρώματα, τότε μπορούμε να πάρουμε μέχρι 16,7 εκατομμύρια διαφορετικές αποχρώσεις από ένα RGB LED. Ο απλούστερος τρό-

πος για να το πετύχουμε είναι η μεταβολή της τάσης στα άκρα του αντίστοιχου LED. Καθώς ο Arduino δε διαθέτει μετατροπέα ψηφιακού σε αναλογικό (DAC), μέσω του οποίου θα ήταν δυνατή η παραγωγή αναλογικής (μεταβλητής) τάσης σε κάποια ακίδα του μικροελεγκτή, θα εκμεταλλευτούμε τη δυνατότητα παραγωγής “ψευδο-αναλογικής τάσης” μέσω των κατά πλάτος διαμορφωμένων παλμών (PWM), μια δυνατότητα που μπορεί να υλοποιηθεί στις ψηφιακές ακίδες 3, 5, 6, 9, 10 και 11 του Arduino.

### **Τι είναι η διαμόρφωση παλμών κατά πλάτος (PWM);**

Η διαμόρφωση παλμών κατά πλάτος είναι μια τεχνική με την οποία παίρνουμε ψευδο-αναλογικά αποτελέσματα με ψηφιακά μέσα. Δημιουργώντας σε μια από τις διαθέσιμες για το σκοπό αυτό ψηφιακές ακίδες του Arduino ένα ψηφιακό σήμα (τετραγωνικούς παλμούς) συγκεκριμένης συχνότητας και μεταβάλλοντας το χρόνο που το ψηφιακό σήμα βρίσκεται σε υψηλή λογική στάθμη σε σχέση με το χρόνο που βρίσκεται στη χαμηλή λογική στάθμη, προσομοιώνονται τάσεις ανάμεσα στα 0 V και την τάση τροφοδοσίας  $V_{cc}$  του μικροελεγκτή (5 V στην περίπτωση του Arduino). Εξ ορισμού η συχνότητα των τετραγωνικών παλμών στις ακίδες 3, 9, 10, 11 του Arduino είναι 490 Hz, ενώ για τις ακίδες 5 και 6 είναι 980 Hz, αλλά αυτό μπορεί να αλλάξει μέσω κατάλληλων εντολών.

Χαρακτηριστικό μέγεθος των κατά πλάτος διαμορφωμένων παλμών είναι ο κύκλος εργασίας (duty cycle), δηλ. το πηλίκο του χρόνου ( $T_{ON}$ ) που ο παλμός βρίσκεται στην υψηλή λογική στάθμη προς την περίοδο ( $T$ ) των παλμών. Έτσι παλμοί με κύκλο εργασίας 50% σημαίνει πως για τη μισή περίοδο βρίσκονται σε υψηλή λογική στάθμη και σε χαμηλή για το άλλο μισό της περιόδου.



Εικόνα 90: Παλμοί διαφορετικού κύκλου εργασίας

Αποδεικνύεται πως η μέση τιμή της τάσης ενός ψηφιακού παλμού είναι ανάλογη του κύκλου εργασίας του, γεγονός που μας επιτρέπει εν τέλει να δημιουργούμε ψευδο-αναλογικές τάσεις (στην ουσία μέσες τιμές τάσης) με τιμές μεταξύ 0 V και τάσης τροφοδοσίας του μικροελεγκτή, μεταβάλλοντας απλά τον κύκλο εργασίας (duty cycle) ενός τετραγωνικού παλμού.

Ο Arduino διαθέτει την εντολή “`analogWrite()`” με την οποία σε μια -από τις διαθέσιμες για το σκοπό αυτό- ψηφιακή ακίδα μπορούμε να δημιουργήσουμε τετραγωνικούς παλμούς με συγκεκριμένο κύκλο εργασίας. Παίρνει δύο παραμέτρους: Η πρώτη αντιστοιχεί στην ακίδα που θα δημιουργηθεί ο παλμός και η δεύτερη στον κύκλο εργασίας. Για παράδειγμα η εντολή “`analogWrite(3, 255)`” δημιουργεί τετραγωνικό παλμό στην ακίδα 3 με κύκλο εργασίας 100%, ενώ η εντολή “`analogWrite(5, 127)`” δημιουργεί τετραγωνικό παλμό στην ακίδα 5 με κύκλο εργασίας 50%.

### **Μίξη χρωμάτων**

Για τη δημιουργία κάποιου χρώματος από το RGB LED πρέπει στις “ψευδο-αναλογικές” ακίδες του Arduino στις οποίες συνδέονται οι αντίστοιχες ακίδες του RGB LED να γράψουμε μέσω της “`analogWrite`” τις κατάλληλες τιμές. Για παράδειγμα, αν -όπως φαίνεται στην Εικόνα 88- έ-

χουμε συνδέσει στις ακίδες 3, 5 και 6 του Arduino τις ακίδες του RGB LED που αντιστοιχούν στο κόκκινο, πράσινο και μπλε χρώμα αντίστοιχα και θέλουμε να πάρουμε κόκκινο χρώμα, θα χρειαστεί ο κώδικας:

```
analogWrite(3, 255);  
analogWrite(5, 0);  
analogWrite(6, 0);
```

ενώ αν θέλουμε να πάρουμε πορφυρό (magenta) χρώμα, το οποίο είναι αποτέλεσμα μίξης κόκκινου και μπλε χρώματος πλήρους έντασης, ο κώδικας θα γίνει:

```
analogWrite(3, 255);  
analogWrite(5, 255);  
analogWrite(6, 0);
```

Είναι φανερό πως για κάθε χρώμα που θέλουμε να δημιουργήσουμε πρέπει να χρησιμοποιηθούν τρεις εντολές για να ανάψουν τα τρία βασικά χρώματα του RGB LED στην κατάλληλη ένταση το καθένα. Σε τέτοιες περιπτώσεις η αναγνωσιμότητα του κώδικα βελτιώνεται σημαντικά με τη χρήση μιας συνάρτησης ορισμένης από το χρήστη, η οποία θα αναλάβει να εγγράψει στα τρία LEDs τις αντίστοιχες εντάσεις, οι οποίες θα δίνονται ως παράμετροι της συνάρτησης. Έχοντας ορίσει:

```
#define RPin 3  
#define GPin 5  
#define BPin 6
```

ο κώδικας για τη συνάρτηση θα έχει τη μορφή:

```
void setColor(int r, int g, int b)  
{  
  analogWrite(RPin, r);  
  analogWrite(GPin, g);  
  analogWrite(BPin, b);  
}
```

Για ένα απλό sketch, που κυκλικά θα δημιουργεί λευκό φως ακολουθούμενο από τα επτά χρώματα της ίριδας από τα οποία αποτελείται, ο κώδικας θα έχει τη μορφή:

```
/*  
  Ex.41 : RGB LED – Μίξη χρωμάτων  
*/
```

```
#define RPin 3  
#define GPin 5  
#define BPin 6
```

```
void setup()  
{  
  pinMode(RPin, OUTPUT);  
  pinMode(GPin, OUTPUT);  
  pinMode(BPin, OUTPUT);  
}
```

```
void loop()  
{  
  setColor(255,255,255);           // Λευκό
```

```

delay(1000);
setColor(255,0,0);           // Κόκκινο
delay(1000);
setColor(255,60,0);         // Πορτοκαλί
delay(1000);
setColor(255,160,0);        // Κίτρινο
delay(1000);
setColor(0,255,0);          // Πράσινο
delay(1000);
setColor(0,255,255);        // Κυανό
delay(1000);
setColor(0,0,255);          // Μπλε
delay(1000);
setColor(127,0,255);        // Ιώδες
delay(1000);
}

void setColor(int r, int g, int b)
{
  analogWrite(RPin, r);
  analogWrite(GPin, g);
  analogWrite(BPin, b);
}

```

### Παρατηρήσεις:

1. Θα πάρετε καλύτερα αποτελέσματα χρησιμοποιώντας LED διάχυσης (diffused) και κοιτάζοντας το φως τους όχι άμεσα αλλά μετά από ανάκλαση σε κάποια επιφάνεια (π.χ. ένα κομμάτι λευκό χαρτόνι).
2. Θα πρέπει επίσης να αποφύγετε την ιδέα να χρησιμοποιήσετε μια αντίσταση κοινή για τα τρία LEDs της διάταξης, γιατί αυτά δεν “ανάβουν” με την ίδια τάση ορθής πόλωσης, οπότε με την ίδια τάση τροφοδοσίας (π.χ. 5V) τα LEDs δε θα φωτοβολούν με την ίδια ένταση, με αποτέλεσμα π.χ. σε πλήρη φωτοβολία να μη παράγεται ακριβώς λευκό φως αλλά μια λίγο διαφορετική απόχρωση. Το πρόβλημα μπορεί να λυθεί χρησιμοποιώντας τρεις αντιστάσεις (μια για κάθε LED) με διαφορετικές, κατάλληλα επιλεγμένες τιμές. Ένα σημαντικότερο πρόβλημα είναι το εξής: Σκεφτείτε ότι θέλετε όταν είναι πλήρως αναμμένα και τα τρία LEDs, καθένα να διαρρέεται από ρεύμα περίπου 20 mA (που είναι μια συνηθισμένη περίπτωση). Τότε η κατάλληλη τιμή της αντίστασης επιλέγεται ώστε με τάση τροφοδοσίας π.χ. 5 V να διαρρέεται από συνολικό ρεύμα  $3 \times 20 \text{ mA} = 60 \text{ mA}$ . Αν όμως μόνο το ένα από τα τρία LEDs είναι αναμμένο, τότε με την ίδια τιμή αντίστασης τα 60 mA ρεύματος θα διέρχονται από αυτό το ένα LED. Όμως με ρεύμα τέτοιας έντασης το LED κινδυνεύει με καταστροφή και το ίδιο συμβαίνει και με τον μικροελεγκτή, ο οποίος για παράδειγμα στην περίπτωση του Arduino έχει δυνατότητα παροχής ρεύματος μέχρι 40 mA σε κάποια ψηφιακή του ακίδα. Δηλαδή η χρήση μιας αντίστασης και για τα τρία LEDs αποφεύγεται για λόγους ευελιξίας ρυθμίσεων αλλά κυρίως για λόγους προστασίας της διάταξης.


## Οι μικροελεγκτές ESP8266 και ESP32 της Espressif

Η κινέζικη Espressif παρουσίασε δύο οικογένειες μικροελεγκτών:

- Την παλιότερη και περιορισμένων δυνατοτήτων σειρά ESP8266.
- Την αυξημένων δυνατοτήτων σειρά ESP32.

Το σημαντικό πλεονέκτημα αυτών των μικροελεγκτών είναι η ενσωμάτωση δυνατοτήτων δικτύωσης WiFi και Bluetooth (με τη δεύτερη μόνο στην οικογένεια ESP32), αλλά και η δυνατότητα προγραμματισμού τους μέσω του ολοκληρωμένου περιβάλλοντος ανάπτυξης (IDE) του Arduino (αν και όχι μόνο).

Ο ESP8266 διαθέτει μόνο ένα κανάλι αναλογικοψηφιακής μετατροπής (ADC) των 10 bits και 17 ψηφιακές εισόδους/εξόδους γενικής χρήσης (GPIOs), από τις οποίες μόνο οι πέντε (5) είναι ασφαλές να χρησιμοποιηθούν τα έργα μας<sup>2</sup>. Αυτοί οι περιορισμοί είχαν ως αποτέλεσμα την εμφάνιση λίγων σχετικά πλακετών γενικής χρήσης με αυτόν τον μικροελεγκτή (Εικόνα 91). Περισσότερη προσοχή δόθηκε σε μονάδες (π.χ. ESP-01) που χρησιμοποιήθηκαν σε συνδυασμό με άλλα συστήματα μικροελεγκτών (π.χ. Arduino) για να τους προσφέρουν δυνατότητες WiFi δικτύωσης.



Εικόνα 91: Συστήματα ESP8266.  
Από αριστερά: NodeMCU, Wemos D1 mini και η μονάδα ESP-01


Οι μικροελεγκτές ESP32 (θα αναφερόμαστε στις μονάδες ESP-WROOM-32 στο εξής) διαθέτουν 34 γενικής χρήσης ψηφιακές εισόδους/εξόδους (GPIO0 - GPIO19, GPIO21 - GPIO23, GPIO25 - GPIO27, και GPIO32 - GPIO39). Αν και στον ESP32, όπως και στον ESP8266, αρκετές ακίδες είναι δεσμευμένες, υπάρχουν αρκετές ακόμη διαθέσιμες για τα έργα μας. Έτσι έχει αναπτυχθεί πλήθος από αυτόνομα και πλήρη δυνατοτήτων συστήματα βασισμένα στον ESP32. Ας δούμε τώρα αναλυτικότερα τα πιο σημαντικά χαρακτηριστικά του ESP32:

- Περιλαμβάνει δύο 32-bit μικροελεγκτές Tensilica Xtensa LX6, με μέγιστη συχνότητα χρονισμού τα 240MHz.
- Διαθέτει 448 kB μνήμης ROM, 520 kB μνήμης SRAM και 4 MB μνήμης flash. Η μνήμη flash μπορεί να διαμορφωθεί με διάφορους τρόπους. Εξ' ορισμού χρησιμοποιούνται 1,2 MB για τα προγράμματά μας, 1,2 MB για την απομακρυσμένη ενημέρωση (Over The Air update - OTA) και 1,5 MB για τη δημιουργία ενός συστήματος διαχείρισης αρχείων (SPIFFS), δη-

<sup>2</sup> Για παράδειγμα οι ακίδες GPIO6 μέχρι GPIO11 είναι συνδεδεμένες με τη μνήμη flash, κάποιες άλλες πρέπει να διατηρούνται σε υψηλή λογική στάθμη κατά την εκκίνηση του μικροελεγκτή, αλλιώς η εκκίνηση αποτυγχάνει, κ.ά.


λαδή λειτουργία ως μια μίνι κάρτα μνήμης για τη διαχείριση των απαραίτητων ανάλογα με την εφαρμογή αρχείων. Στην περίπτωση που δε χρειαζόμαστε τις δυνατότητες απομακρυσμένης ενημέρωσης μπορεί να επιλεγεί (μέσω του Arduino IDE) ένα “NO OTA” σχήμα διαμερισμού της μνήμης flash, ώστε περισσότερη μνήμη να διατεθεί για τα προγράμματά μας. Αντίστοιχα αν δε χρειαζόμαστε δυνατότητες διαχείρισης αρχείων μπορούμε να επιλέξουμε το “Minimal SPIFFS” σχήμα διαμερισμού.

- Ακολουθεί λογική 3,3V.
- Διαθέτει, όπως είπαμε, 34 ψηφιακές ακίδες, από τις οποίες οι GPIO34, 35, 36 και 39 δεν έχουν εσωτερικές αντιστάσεις πρόσδεσης στην τροφοδοσία ή τη γείωση και έτσι δε μπορούν να χρησιμοποιηθούν ως έξοδοι, αλλά μόνο ως εισοδοι.
- Δέκα (10) από τις ψηφιακές ακίδες (GPIO0, 2, 4, 12, 13, 14, 15, 27, 32 και 33) διαθέτουν εσωτερικούς χωρητικούς αισθητήρες αφής. Μπορούν δηλαδή να ανιχνεύσουν μεταβολές που συμβαίνουν όταν ακουμπήσουμε τη συγκεκριμένη ακίδα και συνεπώς μπορούν να αντικαταστήσουν μηχανικούς διακόπτες.
- Διαθέτει δεκαοκτώ (18) κανάλια αναλογικο-ψηφιακής μετατροπής (ADC) των 12 bits, τα οποία κατανέμονται σε δύο ομάδες (ADC1 και ADC2). Τα κανάλια ADC2 δεν είναι διαθέσιμα όταν χρησιμοποιείται το WiFi. Οι ακίδες του ESP32 που μπορούν να χρησιμοποιηθούν ως κανάλια ADC καταγράφονται στο **Παράρτημα Δ**. Ας σημειωθεί πως ένα βασικό πρόβλημα με τον αναλογικοψηφιακό μετατροπέα του ESP32 είναι η έλλειψη γραμμικότητας<sup>3</sup> κατά τη μετατροπή (για παράδειγμα δε μπορεί να διακρίνει μεταξύ 3,2V και 3,3 V).
- Διαθέτει τρία (3) κανάλια ασύγχρονης σειριακής επικοινωνίας (UART), δύο (2) I<sup>2</sup>C, τέσσερις (4) SPI και δύο (2) I<sup>2</sup>S διεπαφές, 16 ακίδες PWM και δύο μετατροπείς ψηφιακού σε αναλογικό (DAC).
- Διαθέτει ενσωματωμένο ένα ρολόι πραγματικού χρόνου (RTC), ώστε να είναι δυνατή η παρακολούθηση του χρόνου ακόμη και όταν ο ESP32 έχει τεθεί σε κατάσταση αναστολής λειτουργίας (deep sleep).
- Ενσωματώνει ένα αισθητήρα Hall για τη μέτρηση του περιβάλλοντος μαγνητικού πεδίου, ο οποίος στις πιο πρόσφατες εκδόσεις αντικαταστάθηκε από ένα αισθητήρα θερμοκρασίας.



Εικόνα 92: ESP32 DEVKIT v1

<sup>3</sup> Περισσότερα για το θέμα στο: <https://randomnerdtutorials.com/esp32-adc-analog-read-arduino-ide/>



Εικόνα 93: Wemos D1 R32 v1.0.0

Υπάρχουν πάρα πολλές πλακέτες που ενσωματώνουν τον ESP32. Ενδεικτικά αναφέρουμε:

- Την πλακέτα ESP32 DEVKIT, που εκθέτει σε δύο σειρές όλες τις ακίδες του ESP32 (Εικόνα 92).
- Την πλακέτα Wemos D1 R32, η οποία ακολουθεί τη φόρμα του Arduino (Εικόνα 93) και εκθέτει ένα περιορισμένο πλήθος ακίδων του ESP32 (16 ψηφιακές και 7 αναλογικές, που βεβαίως μπορούν και αυτές να χρησιμοποιηθούν ως ψηφιακές).


Και οι δυο προαναφερθείσες πλακέτες ενσωματώνουν την χαμηλού κόστους και χαμηλής ενεργειακής κατανάλωσης μονάδα ESP-WROOM-32, για την οποία ας τονίσουμε ακόμη μια φορά:

- Τις **δυνατότητες WiFi**, που επιτρέπουν τη σύνδεση του μικροελεγκτή σε κάποιο δίκτυο WiFi, ώστε να αποκτήσει πρόσβαση στο διαδίκτυο (λειτουργία σταθμού) ή τη δημιουργία ασύρματου δικτύου WiFi με τον ίδιο τον μικροελεγκτή (λειτουργία σημείου πρόσβασης).
- Τις **δυνατότητες Bluetooth** ασύρματης επικοινωνίας, τόσο με το κλασσικό πρωτόκολλο, όσο και με το πρωτόκολλο BLE (Bluetooth Low Energy). Η υποστήριξη του κλασσικού πρωτοκόλλου Bluetooth έχει καταργηθεί σε νεότερες εκδόσεις του ESP32 (ESP32-C3, ESP32-C6, ESP32-S3), ενώ η έκδοση ESP32-S2 δεν υποστηρίζει Bluetooth επικοινωνία.
- Σε σχέση με τον ESP8266 προσφέρουν μεγαλύτερες WiFi ταχύτητες, καλύτερη συνδεσιμότητα και πιο ασφαλείς εφαρμογές Internet of Things (IoT).


Δεν πρέπει να παραλείψουμε να πούμε πως εκτός από τον προγραμματισμό μέσω Arduino IDE οι μονάδες ESP32 μπορούν να προγραμματιστούν και σε MicroPython (που αποτελεί και τη νέα τάση στον προγραμματισμό των μικροελεγκτών).

Με τις πιο πρόσφατες προτάσεις της Espressif (ESP32-C3, ESP32-S2) έχει γίνει προσπάθεια να καλυφθεί το κενό μεταξύ ESP8266 και ESP32. Οι μονοπύρηνες υλοποιήσεις, η έλλειψη υποστήριξης Bluetooth στον ESP32-S2, οι λιγότερες γενικής χρήσης είσοδοι/έξοδοι στον ESP32-C3 και κυρίως η ενσωματωμένη υποστήριξη USB έχουν οδηγήσει σε πλακέτες πολύ μικρών διαστάσεων, αλλά εξαιρετικών δυνατοτήτων. Στην ίδια κατεύθυνση (της χαμηλότερης επεξεργαστικής


ισχύος) κινούνται και υλοποιήσεις με τον ESP32-C6 (που υποστηρίζει επικοινωνία WiFi 6) ή τον ESP32-H2, ενώ ο ESP32-S3 αποτελεί αναβάθμιση του κλασσικού ESP32, τόσο σε επίπεδο μικροεπεξεργαστή (διπύρηνος Tensilica Xtensa LX7), όσο και σε επίπεδο διαθέσιμων εισόδων/εξόδων και υποστηριζόμενων διεπαφών.



Εικόνα 94: Πλακέτα με τον ESP32-S2



Εικόνα 95: Πλακέτα με τον ESP32-C3



Εικόνα 96: Πλακέτα με τον ESP32-C6


## Εγκατάσταση του ESP32 στο Arduino IDE

Για να εγκαταστήσετε [17] το λογισμικό που είναι απαραίτητο, ώστε να προγραμματίσετε τις διάφορες πλακέτες ESP32 με το Arduino IDE, ακολουθήστε την εξής διαδικασία:

1. Ανοίξτε το παράθυρο “Προτιμήσεις” και στο πεδίο “Επιπλέον URLs διαχειριστή πλακετών” εισάγετε την τιμή:


[https://raw.githubusercontent.com/espressif/arduino-esp32/gh-pages/package\\_esp32\\_index.json](https://raw.githubusercontent.com/espressif/arduino-esp32/gh-pages/package_esp32_index.json)

Μπορείτε να εισάγετε περισσότερα URLs, διαχωρίζοντάς τα με κόμμα.



Εικόνα 97: Το παράθυρο "Προτιμήσεις"

2. Μέσω του μενού “Εργαλεία/Πλακέτα” ανοίξτε το “Διαχειριστή πλακετών” και κάντε αναζήτηση με τον όρο “ESP32”.



Εικόνα 98: Διαχειριστής πλακετών

3. Αφού επιλέξετε το πακέτο “esp32 by Espressif Systems” πιάστε το πλήκτρο “Εγκατάσταση” και μόλις ολοκληρωθεί η εγκατάσταση επανεκκινήστε το Arduino IDE.

Με τον ίδιο τρόπο μπορεί να γίνει και η εγκατάσταση του ESP8266 στο Arduino IDE. Προσθέστε στο πεδίο “Επιπλέον URLs διαχειριστή πακετών” στις “Προτιμήσεις” του IDE την τιμή:


[http://arduino.esp8266.com/stable/package\\_esp8266com\\_index.json](http://arduino.esp8266.com/stable/package_esp8266com_index.json)

και με τον “Διαχειριστή πακετών” εγκαταστήστε το πακέτο “esp8266 by ESP8266 Community”.

### Ένα παράδειγμα αξιοποίησης του ESP32: Μίνι μετεωρολογικός σταθμός

Θα δούμε τώρα πως, αξιοποιώντας την πλακέτα Wemos D1 R32 και τη μονάδα BME280 της Bosch, μπορούμε να κατασκευάσουμε ένα μίνι μετεωρολογικό σταθμό. Η μονάδα BME280 αποτελεί αναβάθμιση της BMP280, αφού εκτός από μετρήσεις ατμοσφαιρικής πίεσης και θερμοκρασίας μπορεί επιπλέον να μετράει και τη σχετική υγρασία. Η επικοινωνία με τον μικροελεγκτή γίνεται μέσω του διαύλου I<sup>2</sup>C, ενώ μπορεί να τροφοδοτηθεί από τάση 3,3V μέχρι 5 V και συνεπώς συνδέζεται ιδανικά με συστήματα λογικής 3,3V, όπως ο ESP32.

Για τη σύνδεση της μονάδας BME280 στον μικροελεγκτή απαιτούνται μόνο τέσσερα (4) καλώδια, δύο για την τροφοδοσία και δύο για τους αγωγούς SDA, SCL του διαύλου I<sup>2</sup>C (Εικόνα 99).



Εικόνα 99: Σύνδεση μονάδας BME280 στον Wemos D1 R32

Τις δυνατότητες της μονάδας BME280 μπορούμε να εκμεταλλευτούμε με χρήση της βιβλιοθήκης BME280 του Tyler Glenn, η οποία μπορεί να εγκατασταθεί μέσω του “Διαχειριστή βιβλιοθήκης” του Arduino IDE. Την ίδια βιβλιοθήκη έχουμε χρησιμοποιήσει και στο [παράδειγμα](#) (σελ. 61) με τη μονάδα BMP280 της ίδιας εταιρείας. Μπορούμε ακόμη να χρησιμοποιήσουμε και στην περίπτωση του BME280 το ίδιο λογισμικό που δημιουργήσαμε στο παράδειγμα με τον BMP280. Απαραίτητες είναι μόνο τρεις αλλαγές:

- Η κλήση της συνάρτησης *Wire.begin* για την ενεργοποίηση του διαύλου I<sup>2</sup>C πρέπει να γίνει ως: *Wire.begin(SDA, SCL)*, όπου SDA και SCL είναι οι προκαθορισμένες ακίδες (21 και 22 αντίστοιχα) του ESP32 στις οποίες υλοποιείται ο διάυλος I<sup>2</sup>C.
- Πρέπει να γίνει αλλαγή στο μήνυμα λάθους, ώστε να ανταποκρίνεται στην περίπτωση που δε βρεθεί ο αισθητήρας BME 280 (προηγουμένως αναφερόταν στον αισθητήρα BMP280).

- Στη συνάρτηση `prindBME280Data()` προστίθεται ο απαραίτητος κώδικας για την εκτύπωση στη σειριακή κονσόλα της τιμής της σχετικής υγρασίας.

```

/*
  Ex.42 : Mini meteo station v.1
*/
#include <Wire.h>           // Συμπερίληψη της βιβλιοθήκης Wire
#include <BME280I2C.h>     // Συμπερίληψη της βιβλιοθήκης BME280I2C

BME280I2C bme;           // Εξ ορισμού ρυθμίσεις : forced mode, standby time = 1000 ms
                        // Oversampling = pressure ×1, temperature ×1, humidity ×1, filter off

float tempC(NAN), pres(NAN), hum(NAN);

void setup()
{
  Serial.begin(9600);      // Ενεργοποίηση σειριακής επικοινωνίας
  Wire.begin(SDA, SCL);   // Ενεργοποίηση διαύλου I2C

  while(!bme.begin())     // Αναμονή για ανίχνευση συμβατής συσκευής
  {
    Serial.println("BME280 not found!");
    delay(1000);
  }
}

void loop()
{
  bme.read(pres, tempC, hum);
  printBME280Data();      // Λήψη και τύπωση των αποτελεσμάτων της μέτρησης
  delay(1000);
}

void printBME280Data()
{
  Serial.print("Pressure: ");
  Serial.print(pres);
  Serial.print(" Pa");
  Serial.print("\tTemperature: ");
  Serial.print(tempC);
  Serial.print("°C");
  Serial.print("\tHumidity: ");
  Serial.print(hum);
  Serial.println(" %");
}


```

Αν και δεν είναι απαραίτητες, ενσωματώθηκαν ακόμη δύο αλλαγές, ώστε να διευκολυνθούν περαιτέρω αλλαγές με τις οποίες θα εκμεταλλευτούμε τις δυνατότητες δικτύωσης του ESP32:

- Οι μεταβλητές `pres`, `tempC` και `hum` που χρησιμεύουν για την αποθήκευση των μετρήσεων πίεσης, θερμοκρασίας και σχετικής υγρασίας ορίστηκαν ως καθολικές.

- Η λήψη των μετρήσεων με τη συνάρτηση `bme.read(pres, tempC, hum)` μεταφέρθηκε ως ανεξάρτητη εντολή στο σώμα της συνάρτησης `loop()`.

Για το ανέβασμα του λογισμικού στον μικροελεγκτή πρώτα επιλέγουμε μέσω του μενού “Εργαλεία/Πλακέτα/ESP32 Arduino” την πλακέτα “Wemos D1 R32” (ή όποια άλλη διαθέτουμε) και μετά τη θύρα στην οποία το Arduino IDE “βλέπει” την πλακέτα του ESP32. Τέλος, κάνουμε κλικ στο κουμπί “Ανέβασμα” (ή “Upload to I/O Board”) στη γραμμή εργαλείων του IDE του Arduino και το ολοκληρωμένο περιβάλλον αναλαμβάνει να μεταγλωττίσει και να ανεβάσει τον κώδικα στη μνήμη flash του ESP32. Αφού ολοκληρωθεί το ανέβασμα του κώδικα στον ESP32, κάθε δευτερόλεπτο στη σειριακή κονσόλα εμφανίζονται οι τρέχουσες μετρήσεις πίεσης, θερμοκρασίας και σχετικής υγρασίας (Εικόνα 100).



Εικόνα 100: Πίεση, θερμοκρασία και σχετική υγρασία στη σειριακή κονσόλα

Ας πούμε στο σημείο αυτό πως οι μετρήσεις θερμοκρασίας που επιστρέφει ο αισθητήρας αφορούν τη θερμοκρασία στο εσωτερικό του και όχι τη θερμοκρασία περιβάλλοντος. Γι' αυτό άλλωστε και είναι πάντα κατά 1-2°C μεγαλύτερες από τις αντίστοιχες τιμές της θερμοκρασίας περιβάλλοντος. Για χρήση σε ένα μετεωρολογικό σταθμό προτείνεται ένας ανεξάρτητος αισθητήρας μέτρησης θερμοκρασίας, όπως ο DS18B20 της Dallas Semiconductors.

### **Ασύρματη μετάδοση δεδομένων μέσω Bluetooth**

Ο απλούστερος τρόπος για να εκμεταλλευτούμε τις δυνατότητες ασύρματης δικτύωσης του ESP32 είναι χρησιμοποιώντας το κλασικό Bluetooth πρωτόκολλο επικοινωνίας, μέσω του οποίου μπορεί να αποκατασταθεί ασύρματη σειριακή επικοινωνία μεταξύ του ESP32 και μιας άλλης συζευγμένης συσκευής, π.χ. ένα κινητό τηλέφωνο. Χρησιμοποιώντας τη βιβλιοθήκη `BluetoothSerial` της Espressif, η διαδικασία γίνεται σχεδόν εξίσου εύκολη με την εργασία με σειριακή σύνδεση. Ας το δούμε αναλυτικότερα τι αλλαγές απαιτούνται στον κώδικα:

1. Συμπερίληψη της βιβλιοθήκης *BluetoothSerial*:  
`#include <BluetoothSerial.h>`
2. Δημιουργία αντικειμένου κλάσης *BluetoothSerial*:  
`BluetoothSerial SerialBT;`
3. Έναρξη της επικοινωνίας Bluetooth και ορισμός του ονόματος με το οποίο ο ESP32 θα είναι ορατός σε άλλες Bluetooth συσκευές:  
`SerialBT.begin("ESP32MeteoStation");`
4. Τέλος, στη συνάρτηση *loop()* γίνεται κλήση της συνάρτησης *BTPrint()*, η οποία ορίζεται αμέσως μετά και στην οποία γίνεται η αποστολή των δεδομένων σε κάθε Bluetooth συνδεδεμένη συσκευή.

Με την ενσωμάτωση των αλλαγών αυτών, ο κώδικας παίρνει τη μορφή:

```

/*
   Ex.43 : Mini meteo station v.2
*/

#include <Wire.h>           // Συμπερίληψη της βιβλιοθήκης Wire
#include <BME280I2C.h>     // Συμπερίληψη της βιβλιοθήκης BME280I2C
#include <BluetoothSerial.h> // Συμπερίληψη της βιβλιοθήκης BluetoothSerial

BluetoothSerial SerialBT; // Δημιουργία αντικειμένου BluetoothSerial
BME280I2C bme;           // Εξ ορισμού ρυθμίσεις : forced mode, standby time = 1000 ms
                        // Oversampling = pressure ×1, temperature ×1, humidity ×1, filter off

float tempC(NAN), pres(NAN), hum(NAN);

void setup()
{
  Serial.begin(9600);      // Ενεργοποίηση σειριακής επικοινωνίας
  Wire.begin(SDA, SCL);   // Ενεργοποίηση διαύλου I2C

  SerialBT.begin("ESP32MeteoStation"); // Ρύθμιση BluetoothSerial

  while(!bme.begin())     // Αναμονή για ανίχνευση συμβατής συσκευής
  {
    Serial.println("BME280 not found!");
    delay(1000);
  }
}

void loop()
{
  bme.read(pres, tempC, hum);
  printBME280Data();      // Λήψη και τύπωση των αποτελεσμάτων της μέτρησης
  printBT();              // Αποστολή δεδομένων μέσω Bluetooth
  delay(1000);
}

```

```

void printBME280Data()
{
    Serial.print("Pressure: ");
    Serial.print(pres);
    Serial.print(" Pa");
    Serial.print("\tTemperature: ");
    Serial.print(tempC);
    Serial.print("°C");
    Serial.print("\tHumidity: ");
    Serial.print(hum);
    Serial.println(" %");
}


```

```

void printBT()
{
    SerialBT.print(pres);
    SerialBT.print(" Pa\t");
    SerialBT.print(tempC);
    SerialBT.print("°C\t");
    SerialBT.print(hum);
    SerialBT.println(" %");
}

```

Με την ολοκλήρωση του ανεβάσματος του κώδικα στον ESP32, μπορείτε να ενεργοποιήσετε την επικοινωνία Bluetooth στο κινητό σας και να πραγματοποιήσετε τη σύζευξη με τον ESP32, που στις Bluetooth συσκευές στο κινητό σας αναγνωρίζεται με το όνομα “ESP32MeteoStation”. Μετά είστε έτοιμοι να “τρέξετε” στο κινητό κάποια εφαρμογή σειριακού τερματικού, όπως το “Serial Bluetooth Terminal”, να συνδεθείτε στη μονάδα Bluetooth του ESP32 και να στείλετε ή να λάβετε μηνύματα. Η σύνδεση στη μονάδα Bluetooth γίνεται μέσω του μενού “Devices / Bluetooth Classic” και επιλέγοντας τη συσκευή με το όνομα “ESP32MeteoStation” (Εικόνα 101).



Εικόνα 101: Τα δεδομένα από τον BME280 στην οθόνη του κινητού

## Ο μετεωρολογικός σταθμός στην IoT πλατφόρμα ThingSpeak

Αλλά, όπως έχουμε πει, το σημαντικότερο πλεονέκτημα του ESP32 είναι η δυνατότητα WiFi δικτύωσης που διαθέτει. Ας δούμε λοιπόν τώρα πως, μπορούμε να αξιοποιήσουμε τις δυνατότητες WiFi του ESP32 για την online απεικόνιση των μετρήσεων του σταθμού μας στην IoT πλατφόρμα ThingSpeak. Στην ελεύθερη χρεώσεων έκδοσή της η πλατφόρμα αυτή μπορεί να δεχτεί και να απεικονίσει με διάφορους τρόπους (π.χ. σε γραφήματα ή σε πεδία τιμών ή σε εικονικά όργανα) περιορισμένο αριθμό μετρήσεων από αισθητήρες. Στην έκδοσή του μετεωρολογικού σταθμού που παρουσιάζουμε εδώ ο ESP32 στέλνει μια σειρά δεδομένων (πίεση, θερμοκρασία και σχετική υγρασία) κάθε 20 δευτερόλεπτα. Αυτό δεν αποτελεί πρόβλημα, αφού αυτά τα μεγέθη δεν παρουσιάζουν πολύ απότομες διακυμάνσεις. Προαπαιτούμενα για τον IoT μετεωρολογικό σταθμό μας είναι:

1. Η βιβλιοθήκη ThingSpeak της Mathworks, που θα την εγκαταστήσετε μέσω του “Διαχειριστή βιβλιοθηκών” (από το μενού “Εργαλεία/Διαχειριστής βιβλιοθηκών” του Arduino IDE και αναζήτηση με τον όρο “ThingSpeak”).
2. Η βιβλιοθήκη SimpleTimer, που δε θα τη βρείτε στον “Διαχειριστή βιβλιοθηκών”, αλλά θα τη μεταφορτώσετε, ως ένα συμπιεσμένο αρχείο τύπου “zip”, από την ιστοσελίδα της στο Github “<https://github.com/marcelloromani/Arduino-SimpleTimer/tree/master>”. Μετά θα αποσυμπιέσετε το φάκελο “SimpleTimer” στο φάκελο του υπολογιστή σας που αποθηκεύονται οι βιβλιοθήκες του Arduino (“Εγγραφα/Arduino/Libraries” σε συστήματα Windows).
3. Δημιουργία ελεύθερου χρεώσεων λογαριασμού στο ThingSpeak. Ξεκινάμε με την επιλογή “Get Started For Free” στην ιστοσελίδα “<https://thingspeak.mathworks.com>” της πλατφόρμας. Στην επόμενη σελίδα επιλέγουμε “Create one!” για να δημιουργήσουμε το λογαριασμό μας. Μετά σημειώνουμε τα στοιχεία που μας ζητάει (email, όνομα κλπ) και πατάμε το πλήκτρο “Continue”. Αν στην επόμενη σελίδα σας ενημερώσει ότι ανιχνεύτηκε προσωπικό email (“Personal Email Detected”), τσεκάρετε την επιλογή “Use this email for my MathWorks Account” και πιάστε “Continue”. Επιβεβαιώστε το email σας και πιάστε “Continue”. Δώστε τον κωδικό (password) του λογαριασμού σας, τσεκάρετε την επιλογή “I accept the Online Services Agreement” και συνεχίστε (“Continue”). Αυτό ολοκληρώνει τη δημιουργία του λογαριασμού σας, οπότε πιάστε “Ok” και στο αναδυόμενο παράθυρο “ThingSpeak Usage Intent” επιλέξτε “Personal, non-commercial projects”. Κλείστε το αναδυόμενο παράθυρο πιέζοντας “Ok”.

Το ThingSpeak είναι μια IoT πλατφόρμα, η οποία χρησιμοποιεί κανάλια για να αποθηκεύσει τα δεδομένα που στέλνει μια εφαρμογή ή μια συσκευή. Θα πρέπει λοιπόν να δημιουργήσετε ένα κανάλι για την αποθήκευση των δεδομένων πίεσης, θερμοκρασίας και σχετικής υγρασίας που κάθε 20 δευτερόλεπτα θα στέλνει ο ESP32. Ακολουθήστε τη διαδικασία:

1. Αν έχετε αποσυνδεθεί από το λογαριασμό σας στο ThingSpeak, συνδεθείτε και θα μεταφερθείτε αυτόματα στη σελίδα “MyChannels”, όπου μπορείτε να δείτε τα κανάλια που έχετε δημιουργήσει ή να δημιουργήσετε ένα νέο. Θα δημιουργήσετε ένα νέο κανάλι πιέζοντας το πλήκτρο “New Channel”. Στη σελίδα “New Channel” δώστε ένα όνομα και μια σύντομη περιγραφή για το κανάλι σας. Μετά τσεκάρετε τις επιλογές “Field 1”, “Field 2” και “Field 3” και δώστε τους τις τιμές “Pressure”, “Temperature” και “Rel. Humidity” αντίστοιχα. Με αυτό τον τρόπο ενημερώνετε το ThingSpeak για τα μεγέθη (πίεση, θερμοκρασία και σχετική υγρασία) που θα πρέπει να ανα-

### New Channel

Name	My meteo station
Description	A BME280 meteo station
Field 1	Pressure <input checked="" type="checkbox"/>
Field 2	Temperature <input checked="" type="checkbox"/>
Field 3	Rel. Humidity <input checked="" type="checkbox"/>
Field 4	<input type="text"/> <input type="checkbox"/>
Field 5	<input type="text"/> <input type="checkbox"/>
Field 6	<input type="text"/> <input type="checkbox"/>
Field 7	<input type="text"/> <input type="checkbox"/>
Field 8	<input type="text"/> <input type="checkbox"/>


Εικόνα 102: Δημιουργία καναλιού στο ThingSpeak

μένει και για τη σειρά με την οποία θα τα λαμβάνει από τον ESP32. Κανένα άλλο στοιχείο δεν απαιτείται για τη λειτουργία του σταθμού μας, οπότε μπορούμε στο τέλος της σελίδας να πιέσουμε το πλήκτρο “Save Channel”.

2. Στην επόμενη σελίδα (με τίτλο “My meteo station”) και στην καρτέλα “Private View” ρυθμίζεται ο τρόπος εμφάνισης των δεδομένων στο ιδιωτικό κανάλι μας στο ThingSpeak. Εξ ορισμού οι μετρήσεις που λαμβάνει εμφανίζονται σε χρονικές γραφικές παραστάσεις, αλλά αυτό μπορεί να αλλάξει. Οι δυνατότητες που έχουμε είναι:

- Μέσω της επιλογής “Add Widgets” μπορούμε να ζητήσουμε την εμφάνιση κάποιου πεδίου σε εικονικό όργανο μέτρησης (*gauge*) ή αριθμητική οθόνη (*numeric display*).
- Μέσω της επιλογής “Add Visualizations” τα δεδομένα μπορούν να εμφανιστούν σε χρονική γραφική παράσταση.

Στην περίπτωση μας χρησιμοποιήσαμε εικονικά όργανα για την πίεση και τη σχετική υγρασία και γραφική παράσταση για την θερμοκρασία (Εικόνα 103). Κάθε εικονικό όργανο ή γραφική παράσταση διαθέτει πληθώρα ρυθμίσεων που καθορίζουν την αντίστοιχη εμφάνιση και για τη λειτουργία των οποίων δε χρειάζονται ιδιαίτερες εξηγήσεις (Εικόνα 104).



Εικόνα 103: Ρύθμιση εμφάνισης των μετρήσεων που στέλνει ο ESP32

3. Στην ίδια σελίδα (με τίτλο “My meteo station”) μπορούμε:

- Να βρούμε τον κωδικό αριθμό του καναλιού μας (“Channel ID”).
- Στην καρτέλα “Sharing” να καθορίσουμε αν το κανάλι θα είναι ιδιωτικό ή δημόσιο. Αν επιλέξουμε το κανάλι μας να είναι δημόσιο, τότε πρέπει να ρυθμίσουμε την εμφάνιση των δεδομένων στην καρτέλα “Public View”. Σε πρώτη φάση είναι προτιμότερο να το αφήσουμε ιδιωτικό. Άλλωστε μπορούμε οποιαδήποτε στιγμή αργότερα να το μετατρέψουμε σε δημόσιο.
- Στην καρτέλα “API Keys” θα βρούμε τους κωδικούς “Write API Key” και “Read API Key”.

Τόσο ο κωδικός “Channel ID”, όσο και ο κωδικός “Write API Key” χρειάζονται, ώστε να έχει τη δυνατότητα ο ESP32 να στείλει δεδομένα στο ThingSpeak.

Configure widget parameters

Name: Pressure

Field: Field 1

Min: 800

Max: 1000

Display Value:

Units: kPa

Tick Interval: 50

Update Interval: 15 second(s)

Create Cancel

Εικόνα 104: Παράμετροι εικονικού οργάνου μέτρησης

Ας δούμε τώρα τμηματικά το λογισμικό για τον ESP32. Στο τμήμα των συμπεριλήψεων (includes) έχουμε:

```
#include <Wire.h>
#include <BME280I2C.h>
#include <WiFi.h>
#include <ThingSpeak.h>
#include <SimpleTimer.h>
```

Αναλυτικά:

- Οι βιβλιοθήκες “Wire” και “BME280I2C” χρειάζονται για την επικοινωνία του ESP32 με τον αισθητήρα BME280.
- Η βιβλιοθήκη “WiFi” ανήκει στις βιβλιοθήκες που εξ ορισμού εγκαθίστανται όταν εγκαθιστούμε τις πλακέτες ESP32 στο Arduino IDE και είναι απαραίτητη για τη σύνδεση του ESP32 σε ένα WiFi δίκτυο.
- Η βιβλιοθήκη “ThingSpeak” αναλαμβάνει την διαδικτυακή επικοινωνία μεταξύ ESP32 και της πλατφόρμας ThingSpeak.
- Με τη βιβλιοθήκη “SimpleTimer” μπορούμε μέσω λογισμικού να δημιουργήσουμε ένα χρονόμετρο, ώστε συγκεκριμένες εντολές να εκτελούνται σε τακτά χρονικά διαστήματα.

Ακολουθεί το τμήμα ορισμών μεταβλητών και αντικειμένων:

```
char ssid[] = "myNET";
char pass[] = "myPass";
unsigned long myChannelNumber = myChannelID;
const char * myWriteAPIKey = "myWriteKey";

float tempC(NAN), pres(NAN), hum(NAN);

BME280I2C bme;
WiFiClient client;
SimpleTimer timer;
```

Στις τέσσερις παραμέτρους “ssid”, “pass”, “myChannelNumber” και “myWriteAPIKey” θα αντικαταστήσετε τις δοθείσες τιμές αντίστοιχα με: το όνομα και τον κωδικό του δικτύου σας και με τον αναγνωριστικό αριθμό του καναλιού σας στο ThingSpeak και τον κωδικό εγγραφής στο κανάλι σας. Τα δύο τελευταία στοιχεία τα έχετε σημειώσει όταν -με βάση τις οδηγίες που δώσαμε- δημιουργήσατε το κανάλι σας στο ThingSpeak.

Ακολουθούν οι τρεις μεταβλητές “tempC”, “pres” και “hum” για την αποθήκευση των τιμών θερμοκρασίας, πίεσης και σχετικής υγρασίας από τον αισθητήρα, και τέλος δημιουργούνται τρία αντικείμενα, ως εξής:

- Αντικείμενο “bme” κλάσης “BME280I2C” για τη διαχείριση και λήψη δεδομένων από τον αισθητήρα BME280.
- Αντικείμενο “client” κλάσης “WiFiClient” για τη σύνδεση στο δίκτυο WiFi.
- Αντικείμενο “timer” κλάσης “SimpleTimer” για την αποστολή των δεδομένων στην πλατφόρμα ThingSpeak κάθε 20 δευτερόλεπτα.

Ακολουθούν τέσσερις συναρτήσεις οριζόμενες από το χρήστη, ως εξής:

1. Η τύπου “float” (κινητής υποδιαστολής) συνάρτηση “round\_it”, που παίρνει δύο παραμέτρους ένα αριθμό “f” τύπου κινητής υποδιαστολής και ακόμη έναν ακέραιο “dec”. Η συνάρτηση στρογγυλοποιεί τον αριθμό “f” σε πλήθος δεκαδικών ψηφίων που καθορίζεται από τον αριθμό “dec”.

```
float round_it(float f, int dec) {  
    return roundf(f * pow(10,dec)) / pow(10,dec);  
}
```

2. Η τύπου “void” συνάρτηση “readSensors” με την οποία ο ESP32 λαμβάνει τις μετρήσεις πίεσης, θερμοκρασίας και σχ. υγρασίας από τον BME280, μετά στρογγυλοποιεί τις τιμές αυτές σε δύο δεκαδικά ψηφία για την πίεση και ένα για τη θερμοκρασία και την σχετική υγρασία. Τέλος αν έχει ενεργοποιηθεί η σειριακή επικοινωνία μεταξύ ESP32 και υπολογιστή μέσω του καλωδίου USB, οι τιμές πίεσης, θερμοκρασίας και σχετικής υγρασίας οι τιμές των μετρούμενων μεγεθών τυπώνονται στη σειριακή κονσόλα.

```
void readSensors()  
{  
    bme.read(pres, tempC, hum);  
  
    pres = round_it(pres, 2);  
    tempC = round_it(tempC, 1);  
    hum = round_it(hum, 1);  
  
    if (Serial)  
    {  
        Serial.print(pres,2);  
        Serial.print(" ");  
        Serial.print(tempC,1);  
        Serial.print(" ");  
        Serial.println(hum,1);  
    }  
}
```

3. Η τύπου “void” συνάρτηση “toThingSpeak” με την οποία ο ESP32 διαβάζει τις τιμές πίεσης, θερμοκρασία και σχ. υγρασίας από τον BME280 και τις στέλνει στο κανάλι μας στο

ThingSpeak. Λόγω κάποιου προβλήματος στη βιβλιοθήκη η εμφάνιση της πίεσης στο κανάλι μας δεν ήταν σωστή, γι' αυτό και η τιμή της αποστέλλεται ως αλφαριθμητικό (“String”). Η σειρά με την οποία στέλνονται οι μετρήσεις πρέπει να είναι ίδια με αυτή που δηλώσαμε κατά τη δημιουργία του καναλιού μας στο ThingSpeak (στο *Field1* η πίεση, στο *Field2* η θερμοκρασία και στο *Field3* η σχετική υγρασία).

```
void toThingSpeak()
{
  readSensors();

  ThingSpeak.setField(1, tempC);
  ThingSpeak.setField(2, String(pres,2));
  ThingSpeak.setField(3, hum);

  int x = ThingSpeak.writeFields(myChannelNumber, myWriteAPIKey);
}
```

4. Η τύπου “void” συνάρτηση “*WiFi\_try\_to\_connect*” αναλαμβάνει τη σύνδεση του ESP32 στο WiFi δίκτυο, αν δεν είναι ήδη συνδεδεμένος και πρέπει να εκτελείται μέσα στη συνάρτηση “*loop*”, ώστε να εξασφαλίζει την επανασύνδεση σε περίπτωση τυχαίας αποσύνδεσης.

```
void WiFi_try_to_connect()
{
  if (WiFi.status() != WL_CONNECTED)
  {
    WiFi.begin(ssid, pass);
    Serial.print("Attempting to connect to SSID: ");
    Serial.println(ssid);

    while (WiFi.status() != WL_CONNECTED)
    {
      Serial.print(".");
      delay(500);
    }

    Serial.println("Connected.");
  }
}
```

Στη συνάρτηση “*setup*”:

- Αρχικοποιείται η επικοινωνία με συσκευές συνδεδεμένες στο δίαυλο I<sup>2</sup>C (ο BME280 στην περίπτωση μας).
- Ενεργοποιείται η μέσω καλωδίου USB σειριακή επικοινωνία με ρυθμό 9600 bps.
- Ρυθμίζεται ο ESP32 ως σταθμός WiFi (WIFI\_STA).
- Αρχικοποιείται η σύνδεση με το ThingSpeak.
- Ενεργοποιείται η επικοινωνία με τον αισθητήρα BME280, αν στο δίαυλο I<sup>2</sup>C βρεθεί συμβατή συσκευή.
- Τέλος με τη συνάρτηση “*timer.setInterval(20000L, toThingSpeak)*” ενημερώνεται το μέσω λογισμικού χρονόμετρο (software timer) ώστε κάθε 20 δευτερόλεπτα (20000 ms) να εκτελεί τη συνάρτηση “*toThingSpeak*”, ώστε οι τιμές πίεσης, θερμοκρασίας και σχετικής υγρασίας να αποστέλλονται προς εγγραφή στο κανάλι μας στο ThinkSpeak.

```

void setup()
{
  Wire.begin(SDA, SCL);
  Serial.begin(9600);
  WiFi.mode(WIFI_STA);
  ThingSpeak.begin(client);

  while (!bme.begin()) {
    Serial.println("BME280 not found!");
    delay (1000);
  }
  timer.setInterval(20000L, toThingSpeak);
}

```




Και τέλος στη συνάρτηση “loop”:

- Γίνεται η σύνδεση ή επανασύνδεση στο δίκτυο WiFi.
- Εκτελείται το χρονόμετρο, το οποίο όπως ήδη εξηγήσαμε μετράει αν από την προηγούμενη εκτέλεσή του ή την αρχή εκτέλεσης του sketch έχουν περάσει 20 δευτερόλεπτα. Αν αυτό έχει συμβεί αποστέλλει τις τρέχουσες μετρήσεις στο ThingSpeak.

```

void loop()
{
  WiFi_try_to_connect();
  timer.run();
}

```



Εικόνα 105: Τα δεδομένα του μετεωρολογικού μας σταθμού στο ιδιωτικό κανάλι μας στο ThingSpeak

Ακολουθεί ο πλήρης κώδικας το λογισμικού για τον ESP32:

```

/*
    Ex.44 : ThingSpeak (IoT) meteo station
*/

#include <Wire.h>
#include <WiFi.h>
#include <BME280I2C.h>
#include <ThingSpeak.h>
#include <SimpleTimer.h>

char ssid[] = "myNET"; // Change with your SSID
char pass[] = "myPASSWORD"; // Change with your PASSWORD
unsigned long myChannelNumber = myChannel; // Change with your Channel ID
const char * myWriteAPIKey = "myWriteAPIKey"; // Change with your API Write Key

float tempC(NAN), pres(NAN), hum(NAN);

BME280I2C bme;
WiFiClient client;
SimpleTimer timer;y

float round_it(float f, int dec) {
    return roundf(f * pow(10,dec)) / pow(10,dec);
}

void readSensors() {
    bme.read(pres, tempC, hum);

    pres = round_it(pres, 2);
    tempC = round_it(tempC, 1);
    hum = round_it(hum, 1);

    if (Serial)
    {
        Serial.print(pres,2);
        Serial.print(" ");
        Serial.print(tempC,1);
        Serial.print(" ");
        Serial.println(hum,1);
    }
}

void toThingSpeak() {
    readSensors();

    // write to the ThingSpeak channel
    ThingSpeak.setField(1, String(pres,2));
    ThingSpeak.setField(2, tempC);
    ThingSpeak.setField(3, hum);

    int x = ThingSpeak.writeFields(myChannelNumber, myWriteAPIKey);
}

```

```

void WiFi_try_to_connect() {
  if (WiFi.status() != WL_CONNECTED)
  {
    WiFi.begin(ssid, pass);
    Serial.print("Attempting to connect to SSID: ");
    Serial.println(ssid);
    while (WiFi.status() != WL_CONNECTED)
    {
      Serial.print(".");
      delay(500);
    }
    Serial.println("Connected.");
  }
}

void setup() {
  Wire.begin(SDA,SCL);
  Serial.begin(9600);
  WiFi.mode(WIFI_STA);
  ThingSpeak.begin(client);           // Initialize ThingSpeak

  while (!bme.begin())
  {
    Serial.println("BME280 not found!");
    delay (1000);
  }

  timer.setInterval(20000L, toThingSpeak);
}

void loop() {
  WiFi_try_to_connect();
  timer.run();
}

```

Οι δυνατότητες WiFi δικτύωσης του ESP32 δεν περιορίζονται σε απλές, σαν κι αυτή εδώ, περιπτώσεις. Μπορούμε για παράδειγμα να στήσουμε ολόκληρο διακομιστή ιστού (web server) στον ESP32, ώστε να μην εξαρτάται το project μας από εξωτερικές ιστοσελίδες. Τα 4 MB μνήμης flash του ESP32 είναι αρκετά και για την αποθήκευση του κώδικα, αλλά και για το στήσιμο του web server και την αποθήκευση της ίδιας της ιστοσελίδας του project μας. Καθώς το βιβλίο αυτό αναφέρεται στον αρχάριο χρήση, αφήνουμε τέτοια project για κάποια άλλη εκδοτική μας προσπάθεια.



## Παράρτημα Α: Τύποι δεδομένων

Η γλώσσα προγραμματισμού του Arduino υποστηρίζει τους εξής τύπους για την αποθήκευση δεδομένων [4]:

<b>int</b>	Στον Arduino Uno (και σε όλα τα μοντέλα που βασίζονται σε μικροελεγκτές ATmega) χρησιμοποιείται για την αποθήκευση ακέραιων αριθμών 16 bits (2 bytes) από -32.768 μέχρι 32.767. Στον Arduino Due και στα μοντέλα που βασίζονται σε επεξεργαστές SAMD χρησιμοποιείται για την αποθήκευση ακέραιων αριθμών 32 bits (4 bytes) από -2.147.483.648 μέχρι 2.147.483.647.
<b>short</b>	Για την αποθήκευση ακέραιων αριθμών 16 bits (2 bytes) από -32.768 μέχρι 32.767.
<b>long</b>	Για ακέραιες μεταβλητές των 32 bits (ή 4 bytes) από -2.147.483.648 μέχρι 2.147.483.647.
<b>word</b>	Για ακέραιους αριθμούς 16 bits χωρίς πρόσημο (από 0 μέχρι 65.535).
<b>float</b>	Για δεκαδικούς αριθμούς από 3.4028235E+38 μέχρι -3.4028235E+38. Αποθηκεύονται σαν 32 bits (4 bytes) πληροφορίας.
<b>double</b>	Στον Arduino Uno (και σε όλα τα μοντέλα που βασίζονται σε μικροελεγκτές ATmega) δεν έχει καμία διαφορά με τον τύπο float. Στον Arduino Due και στα μοντέλα που βασίζονται σε επεξεργαστές SAMD χρησιμοποιείται για την αποθήκευση δεκαδικών αριθμών ακρίβειας 64 bits.
<b>byte</b>	Για την αποθήκευση ακέραιων αριθμών ακρίβειας 8 bits από 0 μέχρι 255.
<b>char</b>	Καταλαμβάνει 1 byte μνήμης και αποθηκεύει την τιμή ενός χαρακτήρα. Τα γράμματα των χαρακτήρων γράφονται σε μοναδικά εισαγωγικά, π.χ: 'A'.
<b>boolean</b>	Καταλαμβάνει 1 byte μνήμης και αποθηκεύει μόνο δύο τιμές: <b>true</b> ή <b>false</b> .
<b>array</b>	Συλλογή μεταβλητών που είναι προσβάσιμες με έναν αριθμό ευρετηρίου (πίνακας).
<b>string</b>	Συλλογή αλφαριθμητικών χαρακτήρων (συμβολοσειρά). Μπορεί να δηλωθεί με τον τύπο δεδομένων <b>String</b> , ή ως ένας πίνακας χαρακτήρων τερματισμένος με το χαρακτήρα null (ASCII code 0).
<b>unsigned int</b>	Στον Arduino Uno (και σε όλα τα μοντέλα που βασίζονται σε μικροελεγκτές ATmega) χρησιμοποιείται για την αποθήκευση ακέραιων αριθμών 16 bits (2 bytes) χωρίς πρόσημο από 0 μέχρι 65.535. Στον Arduino Due και στα μοντέλα που βασίζονται σε επεξεργαστές SAMD χρησιμοποιείται για την αποθήκευση ακέραιων αριθμών 32 bits (4 bytes) από 0 μέχρι 4.294.967.295.
<b>unsigned long</b>	Χρησιμοποιείται για την αποθήκευση ακέραιων αριθμών 32 bits (4 bytes) από 0 μέχρι 4.294.967.295.
<b>unsigned char</b>	Ίδιος με τον τύπο byte.

## Παράρτημα Β: Τελεστές

Ένας τελεστής είναι ένα σύμβολο (ή συνδυασμός συμβόλων) που πληροφορεί τον μεταγλωττιστή να πραγματοποιήσει συγκεκριμένη μαθηματική ή λογική πράξη. Οι τελεστές της γλώσσας C++ μπορεί να είναι:

- Αριθμητικοί
- Λογικοί
- Σύγκρισης
- Επιπέδου bit
- Αντικατάστασης

Στους πίνακες που ακολουθούν αναφέρονται οι τελεστές που χρησιμοποιούνται στον προγραμματισμό του Arduino καθώς και η σύντομη περιγραφή τους [18, 19, 4].

### Αριθμητικοί Τελεστές

Όνομα	Σύμβολο	Παράδειγμα	Περιγραφή
Ανάθεση	=	$A = 5$	Αποθηκεύει την τιμή 5 στη μεταβλητή A
Πρόσθεση	+	$Y = A + B$	Πρόσθεση των αριθμών A και B
Αφαίρεση	-	$Y = A - B$	Αφαίρεση του B από τον A
Πολλαπλασιασμός	*	$Y = A * B$	Πολλαπλασιασμός των αριθμών A, B
Διαίρεση	/	$Y = B / A$	Διαίρεση B διά A. Όταν και οι δύο αριθμοί είναι ακέραιοι, το αποτέλεσμα είναι επίσης ακέραιος, ενώ αν ένας από τους δύο είναι κινητής υποδιαστολής, εφαρμόζεται διαίρεση πραγματικών αριθμών.
Υπόλοιπο	%	$Y = B \% A$	Υπόλοιπο της διαίρεσης B / A. Εφαρμόζεται μόνο σε ακέραιους αριθμούς.

### Λογικοί τελεστές

Όνομα	Σύμβολο	Παράδειγμα	Περιγραφή
AND	&&	$(A \&\& B)$	Λογική σύζευξη (AND) Το αποτέλεσμα είναι true, <b>μόνο</b> αν και οι δύο τελεστέοι (A, B) είναι true.
OR		$(A    B)$	Λογικό διάζευξη (OR) Το αποτέλεσμα είναι true, <b>αν ένας τουλάχιστον</b> από τους τελεστέους είναι true.
NOT	!	$!(A)$	Λογική άρνηση (NOT) Αν ο τελεστέος είναι true η λογική του άρνηση είναι false και το αντίστροφο.

### Τελεστές σύγκρισης

Όνομα	Σύμβολο	Παράδειγμα	Περιγραφή
Ίσο με	==	(A == B)	True αν η μεταβλητή A είναι ίση με τη B
Όχι ίσο με	!=	(A != B)	True αν η μεταβλητή A δεν είναι ίση με τη B
Μικρότερο από	<	(A < B)	True αν η μεταβλητή A είναι μικρότερη από τη B
Μεγαλύτερο από	>	(A > B)	True αν η μεταβλητή A είναι μεγαλύτερη από τη B
Μικρότερο από ή ίσο με	<=	(A <= B)	True αν η μεταβλητή A είναι μικρότερη από ή ίση με τη B
Μεγαλύτερο από ή ίσο με	>=	(A >= B)	True αν η μεταβλητή A είναι μεγαλύτερη από ή ίση με τη B

### Τελεστές επιπέδου bit

Όνομα	Σύμβολο	Παράδειγμα	Περιγραφή
and	&	(A & B)	Στο αποτέλεσμα, ένα bit είναι 1, μόνο αν τα αντίστοιχα bits στους τελεστέους είναι και τα δύο 1.
or		(A   B)	Στο αποτέλεσμα, ένα bit είναι 1, αν τουλάχιστον ένα από τα αντίστοιχα bits στους τελεστέους είναι 1.
xor	^	(A ^ B)	Στο αποτέλεσμα, ένα bit είναι 1, αν ακριβώς ένα (όχι και τα δύο) από τα αντίστοιχα bits στους τελεστέους είναι 1.
not	~	(~A)	Αντιστρέφει τα bits του τελεστέου, δηλαδή αν είναι 0 τα κάνει 1 και αν είναι 1 τα κάνει 0.
shift left	<<	A << 2	Μετατοπίζει τα bits του αριστερού τελεστέου προς τα αριστερά. Ο αριθμός των θέσεων μετατόπισης καθορίζεται από το δεξιό τελεστέο (2 στην περίπτωση του παραδείγματος). Οι κενές θέσεις που προκύπτουν στα δεξιά γεμίζουν με μηδενικά.
shift right	>>	A >> 2	Μετατοπίζει τα bits του αριστερού τελεστέου προς τα δεξιά. Ο αριθμός των θέσεων μετατόπισης καθορίζεται από το δεξιό τελεστέο (2 στην περίπτωση του παραδείγματος). Οι κενές θέσεις που προκύπτουν στα αριστερά γεμίζουν με μηδενικά.

### Τελεστές αντικατάστασης

Όνομα	Σύμβολο	Παράδειγμα	Περιγραφή
Αύξηση	++	A++	Ισοδύναμο με $A = A + 1$
Μείωση	--	A--	Ισοδύναμο με $A = A - 1$
Πρόσθεση	+=	B += A	Ισοδύναμο με $B = B + A$
Αφαίρεση	-=	B -= A	Ισοδύναμο με $B = B - A$
Πολλαπλασιασμός	*=	B *= A	Ισοδύναμο με $B = B * A$
Διαίρεση	/=	B /= A	Ισοδύναμο με $B = B / A$
Υπόλοιπο	%=	B %= A	Ισοδύναμο με $B = B \% A$
OR επιπέδου bit	=	A  = 2	Ισοδύναμο με $A = A   2$
AND επιπέδου bit	&=	A &= 2	Ισοδύναμο με $A = A \& 2$

## Παράρτημα Γ: AT εντολές για τις μονάδες Bluetooth

### Μονάδα HC-05

Για την αποστολή των εντολών “AT” από τη σειριακή οθόνη του Arduino IDE στη μονάδα **HC-06** πρέπει να έχει ενεργοποιηθεί η επιλογή “Αμφότερα NL & CR”. Η μονάδα **HC-05** διαθέτει ένα πλήρες σετ εντολών, οι σημαντικότερες από τις οποίες είναι:

Εντολή	Ανταπόκριση	Σχόλια
AT	OK	Επιβεβαίωση επικοινωνίας
AT+RESET	OK	Επαναφορά τω εξ ορισμού ρυθμίσεων
AT+VERSION?	+VERSION: <Param> OK	Επιστρέφει την έκδοση λογισμικού της μονάδας
AT+NAME?	+NAME:<Param> OK ή FAIL	Επιστρέφει το όνομα της συσκευής. Εξ ορισμού “HC-05”.
AT+NAME=xyz	OK	Καθορισμός του ονόματος της μονάδας σε xyz
AT+PSWD?	+PSWD : <Param> OK	Επιστρέφει τον κωδικό σύζευξης (PIN). Εξ ορισμού “1234”.
AT+PSWD=pppp	OKsetPIN	Καθορισμός του κωδικού σύζευξης (PIN) στην τετραψήφια τιμή pppp
AT+UART?	+UART=baud rate, stop bit, parity bit  OK	Επιστρέφει τις ρυθμίσεις σειριακής επικοινωνίας της μονάδας.
AT+UART=x,y,z	OK	Καθορισμός των παραμέτρων σειριακής επικοινωνίας της μονάδας. Αποδεκτές τιμές: x : Ρυθμός επικοινωνίας 1200, 2400, 4800, 9600, 19200, 38400, 57600, 115200, 230400, 460800, 921600, 1382400 bps y : Stop bit 0 : 1 bit 1 : 2 bit z : Parity bit 0: None 1: Odd parity 2: Even parity Εξ ορισμού ρυθμίσεις: 9600, 0, 0
AT+ROLE?	+ROLE:<x> OK	x = 0 : Slave x = 1 : Master
AT+PARAM=<x>		x = 2 : Slave-Loop Εξ ορισμού: Slave

## Μονάδα HC-06

Για την αποστολή των εντολών “AT” από τη σειριακή οθόνη του Arduino IDE στη μονάδα Bluetooth HC-06 πρέπει να έχει ενεργοποιηθεί η επιλογή “Δεν υπάρχει τέλος γραμμής”. Το περιορισμένο σετ εντολών της μονάδας HC-06 έχει ως εξής:

Εντολή	Ανταπόκριση	Σχόλια
AT	OK	Επιβεβαίωση επικοινωνίας
AT+VERSION	OK<param>	Επιστρέφει την έκδοση λογισμικού της μονάδας
AT+NAMExyz	OKsetname	Καθορισμός του ονόματος της μονάδας σε xyz
AT+PINpppp	OKsetPIN	Καθορισμός του κωδικού σύζευξης (PIN) στην τετραψήφια τιμή pppp
AT+BAUDx	OKx	Καθορισμός του ρυθμού σειριακής επικοινωνίας της μονάδας. Αποδεκτές τιμές: x = 1 : 1200 bps x = 2 : 2400 bps x = 3 : 4800 bps x = 4 : 9600 bps x = 5 : 19200 bps x = 6 : 38400 bps x = 7 : 57600 bps x = 8 : 115200 bps x = 9 : 230400 bps x = A : 460800 bps x = B : 921600 bps x = C : 1382400 bps

### Παράρτημα Δ: Οι ADC ακίδες του ESP32

Ομάδα	Κανάλι	Ακίδα (GPIO)
ADC1	0	36
	1	37
	2	38
	3	39
	4	32
	5	33
	6	34
	7	35
ADC2	0	4
	1	0
	2	2
	3	15
	4	13
	5	12
	6	14
	7	27
	8	25
	9	26

## Βιβλιογραφία

- [1] H. Barragán, «The Untold History of Arduino,» [Ηλεκτρονικό]. Available: <https://arduinohistory.github.io>. [Πρόσβαση Δεκέμβριος 2018].
- [2] «Arduino,» [Ηλεκτρονικό]. Available: <https://en.wikipedia.org/wiki/Arduino>. [Πρόσβαση Δεκέμβριος 2018].
- [3] «ATmega328P Datasheet,» [Ηλεκτρονικό]. Available: <https://engineering.purdue.edu/ME588/SpecSheets/atmega328p.pdf>. [Πρόσβαση Δεκέμβριος 2018].
- [4] «Arduino Reference,» [Ηλεκτρονικό]. Available: <https://www.arduino.cc/reference/en/>.
- [5] «What does this while(true) loop do in Arduino or any other language?,» [Ηλεκτρονικό]. Available: <https://stackoverflow.com/questions/48524857/what-does-this-whiletrue-loop-do-in-arduino-or-any-other-language>. [Πρόσβαση Δεκέμβριος 2018].
- [6] «Photoresistor,» Wikipedia, [Ηλεκτρονικό]. Available: <https://en.wikipedia.org/wiki/Photoresistor>. [Πρόσβαση Δεκέμβριος 2018].
- [7] «I<sup>2</sup>C,» Wikipedia, [Ηλεκτρονικό]. Available: [https://en.wikipedia.org/wiki/I<sup>2</sup>C](https://en.wikipedia.org/wiki/I%C2). [Πρόσβαση Δεκέμβριος 2018].
- [8] «TinyRTC I2C Module,» Mantech Electronics, [Ηλεκτρονικό]. Available: <http://www.mantech.co.za/Datasheets/Products/MD0095-180521A.pdf>. [Πρόσβαση Δεκεμβριος 2018].
- [9] «1-Wire,» Wikipedia, [Ηλεκτρονικό]. Available: <https://en.wikipedia.org/wiki/1-Wire>. [Πρόσβαση Δεκέμβριος 2018].
- [10] Tweaking4all, «Arduino ds18b20 temperature sensor,» [Ηλεκτρονικό]. Available: <https://www.tweaking4all.com/hardware/arduino/arduino-ds18b20-temperature-sensor/>. [Πρόσβαση Απρίλιος 2019].
- [11] «Serial Peripheral Interfac,» Wikipedia, [Ηλεκτρονικό]. Available: [https://en.wikipedia.org/wiki/Serial\\_Peripheral\\_Interface](https://en.wikipedia.org/wiki/Serial_Peripheral_Interface). [Πρόσβαση Δεκέμβριος 2018].
- [12] P. Horowitz και W. Hill, The Art of Electronics, New York: Cambridge University Press, 1989.
- [13] Honeywell, «SS39ET/SS49E/SS59ET Series Linear Hall-effect Sensor ICs,» [Ηλεκτρονικό]. Available: [https://sensing.honeywell.com/index.php?ci\\_id=50359](https://sensing.honeywell.com/index.php?ci_id=50359). [Πρόσβαση Φεβρουάριος 2019].
- [14] «Speed of Sound,» Wikipedia, [Ηλεκτρονικό]. Available: [https://en.wikipedia.org/wiki/Speed\\_of\\_sound](https://en.wikipedia.org/wiki/Speed_of_sound). [Πρόσβαση Δεκέμβριος 2018].
- [15] «LM35 Precision Centigrade Temperature Sensors,» Texas Instruments, [Ηλεκτρονικό].

Available: <http://www.ti.com/lit/ds/symlink/lm35.pdf>.

[Πρόσβαση Δεκέμβριος 2018].

[16] «Adafruit INA219 Current Sensor Breakout,» [Ηλεκτρονικό]. Available: <https://cdn-learn.adafruit.com/downloads/pdf/adafruit-ina219-current-sensor-breakout.pdf>.

[Πρόσβαση Δεκέμβριος 2018].

[17] Random Nerd Tutorials, [Ηλεκτρονικό].

Available: <https://randomnerdtutorials.com/projects-esp32/>.

[Πρόσβαση Νοέμβριος 2024].

[18] «Arduino Operators,» [Ηλεκτρονικό].

Available: [https://www.tutorialspoint.com/arduino/arduino\\_operators.htm](https://www.tutorialspoint.com/arduino/arduino_operators.htm).

[Πρόσβαση Δεκέμβριος 2018].

[19] «Εισαγωγή στη γλώσσα C & Εφαρμογές,» [Ηλεκτρονικό].

Available: [http://www.dga.gr/guest/pluginfile.php/5773/mod\\_resource/content/4/03%20-%20Operations%2C%20Operators%20and%20Constants.pdf](http://www.dga.gr/guest/pluginfile.php/5773/mod_resource/content/4/03%20-%20Operations%2C%20Operators%20and%20Constants.pdf).

[Πρόσβαση Δεκέμβριος 2018].



